1. Loft Language Reference

A statically-typed scripting language with null safety and built-in parallel execution.

Version 2026.8.0

Every code example in this document is an executable part of the test suite.

Contents

2. Getting Started

Loft is a lightweight scripting language with null safety, built-in parallel execution, and a fast interpreter called **loft**. This page shows how to install loft and write your first program.

2.0.1. Prerequisites

Building from source requires the Rust toolchain (Rust 1.82 or later). Install it from rustup.rs if you do not already have it:

```
curl --proto '=https' --tlsv1.2 -sSf https://sh.rustup.rs | sh
```

2.0.2. Install from source

The quickest way to get the `loft` binary on your PATH:

```
cargo install --git https://github.com/loft-lang/loft --bin loft
```

This compiles loft in release mode and places the binary in `~/.cargo/bin/`, which is already on your PATH if you used rustup.

2.0.3. Build locally

Clone the repository and build with Cargo:

```
git clone https://github.com/loft-lang/loft
cd loft
cargo build --release
```

The binary is at `target/release/loft`. Copy it anywhere on your PATH, for example:

```
cp target/release/loft ~/.local/bin/
```

2.0.4. Verify the installation

```
loft --version
```

2.0.5. Hello, World

Create a file called `hello.loft`:

```
fn main() {
    println("Hello, Loft!");
}
```

Run it:

```
loft hello.loft
```

```
Hello, Loft!
```

2.0.6. A slightly bigger program

Variables, loops, and string interpolation all work without any ceremony:

```
fn main() {
    // Variables are mutable by default; types are inferred.
    total = 0;
    for n in 1..=10 {
        total += n;
    }
    println("Sum 1..10 = {total}");

    // Vectors use square-bracket literals.
    words = ["loft", "is", "fast"];
    for w in words {
        print("{w} ");
    }
    println("");
}
```

2.0.7. Command-line flags

Run `loft --help` for the full list. The most commonly used flags are:

2.0.8. Running your program as a native binary

The interpreter runs programs immediately, but for compute-heavy work you can compile to a native binary instead. This takes a few seconds the first time (it calls `rustc` in the background) but runs 10–50× faster:

```
loft --native myprogram.loft
```

To see the generated Rust code without running it:

```
loft --native-emit myprogram.rs
cat myprogram.rs
```

Native compilation requires Rust 1.85 or later on your PATH. If `rustc` is not found, loft prints a clear error message telling you how to install it.

2.0.9. Standard library

The standard library is a set of `.loft` files bundled alongside the loft binary. They are loaded automatically before your program runs — you do not need any `import` or `use` statement to call `println`, `assert`, or any other built-in function.

The full function reference is in the Standard Library section of these docs.

2.0.10. Using libraries

Place a library file (e.g. `lib/myutil.loft`) next to your program, then bring it into scope:

```
use myutil;
```

```
fn main() {
    result = myutil::compute(42);
    println("{result}");
}
```

See Libraries for the full search path and naming rules.

2.0.11. Editor setup

Loft syntax highlighting extensions are planned. In the meantime:

- **VS Code** — use Rust syntax highlighting as a close approximation.
- **Vim / Neovim** — associate `*.loft` with Rust via `autocmd BufRead *.loft set filetype=rust` in your `vimrc`.
- **Any editor** — the Web IDE (in progress) provides syntax highlighting and navigation in the browser.

2.0.12. Next steps

- Read the language overview starting with Keywords and control flow.
- See vs Rust if you are coming from Rust.
- Browse the Standard Library reference for built-in functions.
- Check the Roadmap for planned features.

3. vs Rust

Loft is inspired by Rust but designed for general-purpose scripting with a shorter learning curve and less ceremony. It trades some of Rust's safety guarantees and expressive power for a simpler mental model. This page documents the most important differences so that Rust programmers know exactly what they gain and what they give up.

3.0.1. Variables — no let or mut

```
x = 42
x += 1
const y = 42   // opt-in: locked in debug builds
```

```
let mut x = 42;
x += 1;
let y = 42;     // immutable by default; compiler-enforced
```

Upside Less boilerplate. No need to decide upfront whether a variable will be mutated. Variables are mutable by default, so refactoring is frictionless. Use const to signal that a value should not change — in debug builds the runtime locks the store immediately after initialisation, catching accidental writes early.

Downside Immutability is opt-in, not the default. Rust makes variables immutable unless you write mut, catching accidents at compile time for free. Loft's const lock is only checked at runtime and only in debug builds, so it provides less of a safety net.

3.0.2. Null instead of Option<T>

```
struct Point {
    label: text,          // nullable
    x: integer not null,  // never null
}
p = Point { x: 1 };
assert(p.label == null, "absent");
```

```
struct Point {
    label: Option<String>,  // explicit absence
    x: i32,                 // always present
}
let p = Point { label: None, x: 1 };
assert!(p.label.is_none());
```

Upside Natural for database records where fields are often absent. No Some(x)/None wrapping. Conditionals on null are concise: if v == null.

Downside Null is opt-out, not opt-in. A field is nullable unless explicitly marked not null, so forgetting the annotation leaves a potential null silently. In Rust, Option is visible in the type and forces handling at every call site.

3.0.3. Parameter mutability — const and & instead of ownership

```
// &: vector growth (append) is visible to caller
fn push_two(v: &vector<integer>) {
    v += [1, 2];  // caller sees the change
}
// const: read-only (locked in debug builds)
fn count(v: const vector<integer>) - integer {
    length(v)
}
// &: explicit write-back for primitives
fn add_to(n: &integer, delta: integer) {
    n += delta;
}
```

```
fn push_two(v: &mut Vec<i32>) {
    v.push(1);
    v.push(2);
}
fn count(v: &Vec<i32>) -> usize {
    v.len()
}
fn add_to(n: &mut i32, delta: i32) {
    *n += delta;
}
```

Upside No borrow-checker errors. No lifetime annotations. No ownership transfer to reason about. Struct field mutations through a parameter are always visible to the caller. Use & on a collection parameter to allow vector growth (append) to propagate back. const parameters signal read-only intent, and in debug builds the runtime locks the store for the duration of the call — enough to catch most accidents during development.

Downside The compile-time guarantees are much weaker. Rust's borrow checker statically proves no aliased mutations, no dangling references, and no data races — before the program ever runs. Loft's const is a debug-only runtime check. Aliasing is unchecked at compile time, and the engine's Rust runtime is what truly enforces memory safety.

3.0.4. ^ is XOR; ** or pow() for exponentiation

```
sq   = 2 ** 10          // ** exponentiation — integers: 1024
area = PI * r ** 2.0    // floats too; pow(r, 2.0) also works
bits = a | b              // bitwise OR
mask = a & b              // bitwise AND
xor  = a ^ b              // bitwise XOR
```

```
let sq   = 2i32.pow(10);   // 1024 — no ** operator in Rust
let area = PI * r.powi(2); // or r.powf(2.0)
let bits = a | b;          // bitwise OR
```

```
let mask = a & b;          // bitwise AND
let xor  = a ^ b;          // bitwise XOR
```

Upside Loft has a ** exponentiation operator that works on both integers (2 ** 10 == 1024) and floats (2.0 ** 3.0 == 8.0); pow(base, exp) is also available for float/single. ^ behaves exactly as Rust developers expect — it is bitwise XOR. Bitwise operator precedence matches Rust and C: | (loose) ⟶ ^ ⟶ & ⟶ shifts ⟶ arithmetic (tight).

Downside The ** operator is loft-specific — Rust has no exponentiation operator and reaches for methods instead (.pow(), .powi(), .powf()), so the muscle memory does not transfer. And ^ is XOR in both languages, never exponentiation — a trap for anyone reaching for it as a power operator out of habit.

3.0.5. No loop keyword — use while or for + break

```
while !done() { step(); }  // while works

// Infinite loop workaround — use a long range:
for _ in 0l..9223372036854775807l {  // long: ~9.2×10^18
    if should_exit() { break; }
    step();
}
```

```
while !done() { step(); }

loop {           // truly infinite — no bound needed
    if should_exit() { break; }
    step();
}
```

Upside while condition { } works exactly as expected. Every for loop has an iteration variable, making it easy to add index tracking or a cycle limit without restructuring.

Downside There is no loop { } keyword for an unconditionally infinite loop. The workaround — a for loop over a long range — is verbose but practically unbounded: the 64-bit maximum (9.2×10^18) exceeds any realistic event-loop iteration count.

3.0.6. Filtered loops — for … if …

```
for x in items if x > 0 {
    total += x;
}
```

```
for x in items.iter().filter(|&&x| x > 0) {
    total += x;
}
```

Upside Reads like natural language. No closure syntax, no double-reference in the filter predicate. v#remove inside a filtered loop also safely removes the current element while iterating — something that requires careful index management in Rust.

Downside The inline if filter is limited to the loop it lives in. For multi-stage pipelines, loft now offers map(v, fn f), filter(v, fn pred), and reduce(v, fn f, init) as composable higher-order functions — see section 11. Rust's lazy iterator chain (.filter().map().take_while()) avoids intermediate allocations; loft's higher-order functions allocate a new vector at each stage.

3.0.7. Loop attributes: #first, #count, #index

```
for x in 1..=5 {
    if !x#first { b += ", "; }
    b += "{x#count}:{x}";
}
```

```
for (i, x) in (1..=5).enumerate() {
    if i != 0 { b.push_str(", "); }
    b.push_str(&format!("{i}:{x}"));
}
```

Upside No tuple destructuring needed. x#first reads as "is this the first x", which is self-explanatory. Eliminates helper counter variables for the common pattern of comma-separated output.

Downside The #attribute syntax is unique to loft and unfamiliar to everyone else. Rust's .enumerate() composes freely with other adapters; loft's attributes are only available on the loop variable of the innermost loop.

3.0.8. Named loop break — x#break

```
for x in 1..5 {
    for y in 1..5 {
        if x * y >= 6 { x#break; }
    }
}
```

```
'outer: for x in 1..5 {
    for y in 1..5 {
        if x * y >= 6 { break 'outer; }
    }
}
```

Upside Reuses the existing loop variable name as the label. No separate 'label declaration at the loop head. The name x#break reads as "break out of the loop over x".

Downside Rust's lifetime-label syntax 'outer is explicit and visually separate from the loop body. x#break is easy to confuse with the loop attributes x#first and x#count, which have completely different semantics.

3.0.9. Methods by self name, not impl blocks

```
fn greet(self: Person) - text {
    "Hello, {self.name}!"
}
fn name_len(self: const Person) - integer {
    length(self.name)  // const: read-only access guaranteed
}
p.greet()     // dot syntax
greet(p)      // free-function call — identical
```

```
impl Person {
    fn greet(&self) -> String {
        format!("Hello, {}!", self.name)
    }
    fn name_len(&self) -> usize {
        self.name.len()
    }
}
p.greet();     // only dot syntax
```

Upside No impl block ceremony. Methods and free functions are the same thing — just a naming convention. Functions can be added to any type from any file without modifying the original struct definition, similar to extension methods. Mark the first parameter const to guarantee the method never modifies the receiver — analogous to Rust's &self.

Downside No consuming self. Plain (non-const) methods behave like &mut self — the caller's value is always mutably aliased. Multiple functions with the same name but different non-variant self types are a compile error — no standard overloading.

3.0.10. Polymorphic enum dispatch

```
enum Shape {
    Circle { r: float },
    Rect   { w: float, h: float }
}
fn area(self: Circle) - float { PI * pow(self.r, 2.0) }
fn area(self: Rect)   - float { self.w * self.h }

s.area()    // dispatches on the runtime variant
```

```
enum Shape {
    Circle { r: f64 },
    Rect   { w: f64, h: f64 },
}
fn area(s: &Shape) -> f64 {
    match s {
        Shape::Circle { r } => PI * r * r,
        Shape::Rect { w, h } => w * h,
```

```
    }
}
```

Upside Each variant's behaviour lives in its own small function — easy to read and extend. Adding a new type of shape only requires a new fn area(self: NewShape) without modifying the dispatch site. Feels like OOP method overriding without the inheritance.

Downside Rust's match is exhaustive: the compiler forces you to handle every variant. In loft, leaving a variant without an implementation emits a compiler warning but does not stop compilation. To suppress the warning and produce a null return, write an explicit empty-body stub: fn area(self: NewShape) -> float { }. Exhaustiveness is enforced by discipline, not the type system.

3.0.11. Closures — same-scope capture works

```
// Capture works when called in the same scope:
offset = 10;
shift = fn(x: integer) -> integer { x + offset };
assert(shift(5) == 15, "shift");

// Non-capturing lambdas work with map/filter/reduce:
doubled = map([1, 2, 3], fn(x: integer) -> integer { x * 2 });
evens   = filter([1, 2, 3, 4], |x| { x % 2 == 0 });

// Cross-scope: returning a capturing lambda works:
fn make_adder(n: integer) -> fn(integer) -> integer {
    fn(x: integer) -> integer { x + n }
}
```

```
// Closure captures context freely, any scope:
let offset = 5;
let shift = |x| x + offset;
assert_eq!(shift(5), 10);

// Works as argument to higher-order functions:
let shifted: Vec<i32> = v.iter()
    .map(|x| x + offset)
    .collect();

// Returning a capturing closure:
fn make_adder(n: i32) -> impl Fn(i32) -> i32 {
    move |x| x + n
}
```

Upside Same-scope capture works for integers, text, and mutable variables — no capture modes (move vs borrow), no Fn/FnMut/FnOnce trait bounds, no lifetime annotations. Both long-form (fn(x: integer) -> integer { x * 2 }) and short-form (|x| { x * 2 }) lambdas are supported. Named function references (fn name) are compile-checked.

Downside Capture is always by value (like Rust move). Rust's Fn/FnMut/FnOnce traits give more flexibility: closures can borrow by reference, be stored in structs, and passed freely across scopes. Loft captures at definition time only — no borrow-based captures.

3.0.12. Generic functions — pass-through only, no trait bounds

```
// Pass-through generics work:
fn identity<T>(x: T) - T { x }
identity(42)        // integer
identity("hello")  // text

// Trait-bounded generics are not supported:
// fn max<T: PartialOrd>(a: T, b: T) -> T { ... }  // not valid
// Must write a version per type:
fn max_int(a: integer, b: integer) - integer {
    if a  b { a } else { b }
}
```

```
fn identity<T>(x: T) -> T { x }

fn max<T: PartialOrd>(a: T, b: T) -> T {
    if a > b { a } else { b }
}
// One function works for all comparable types.
```

Upside Basic generic functions (identity, pick_second, wrappers) work out of the box. Collections (vector<T>, hash<T>, sorted<T>) are generic at the engine level, covering the most common need. No dyn Trait, impl Trait, or associated types to learn.

Downside Interface bounds (<T: Ordered>, <T: Addable>, <T: Printable>) are structural and static-dispatch only — a type satisfies an interface implicitly by defining the required operators, and there are no trait objects (dyn Trait), associated types, or blanket impls. Rust's nominal traits and dynamic dispatch reach cases loft's static interfaces cannot.

3.0.13. String formatting — embedded expressions

```
msg  = "Hi {name}, score: {score:6.2}"   // width.precision
rank = "seed {n:+}"                      // sign shows on integers: "seed +42"
prec = "{value:.2}"                      // precision only
hex  = "{n:#x}"
list = "{for x in 1..4 {x*2}}"          // [2,4,6]
     = "{\"hello\":3}"                    // nested string literal — works
```

```
let msg  = format!("Hi {name}, score: {:6.2}", score);
let rank = format!("seed {:+}", n);
let prec = format!("{:.2}", value);
let hex  = format!("{:#x}", n);
// no in-string iteration; needs a separate collect step
// nested string literals in format args are fine in Rust
```

Upside Concise — no format!() call, no separate variable. Full format expressions (arithmetic, slices, for comprehensions) can appear directly inside {}. Specifiers mirror Rust: :width, :.precision, :width.precision, sign (+), radix (#x, #o, b), alignment (<, >, ^), and zero-padding (0N) all work. Note that the sign (+) and zero-pad (0N) flags apply to integer output only — they are dropped for floats.

Downside Unknown radix letters in specifiers are compile-time errors (e.g. :5z or :5B are both rejected). Radix letters are case-sensitive: valid ones are b, o, x/X, e, and j/json — uppercase B or O produce an error. Applying a numeric specifier to an incompatible type (such as :x on a text value, or zero-padding on a boolean) is a compile-time error.

3.0.14. Built-in parallel for-loops — par(...)

```
fn double(r: const Score) - integer { r.value * 2 }

sum = 0;
for item in scores par(b=double(item), 4) {
    sum += b;   // b is double(item), computed in parallel
}               // results arrive in original order

// Method form:
for item in scores par(b=item.value(), 4) { ... }
```

```
use rayon::prelude::*;

fn double(r: &Score) -> i32 { r.value * 2 }

let sum: i32 = scores.par_iter()
    .map(double)
    .sum();      // order is not guaranteed without extra work

// rayon is an external crate; add to Cargo.toml first
```

Upside Built into the language — no external crate, no Cargo.toml edit. The compiler validates the worker function signature at the call site. Results are delivered in the original vector order. The thread count is set per call, making it easy to tune for the hardware at hand.

Downside Workers can return primitives (integer, long, float, boolean), text, and inline enums — but not struct references. Workers cannot capture local variables: context must be embedded as fields alongside the data in the element struct. For multi-stage transformations, use map() / filter() / reduce() sequentially — each stage allocates a new intermediate vector.

4. vs Python

Loft and Python share a similar surface syntax — assignment without type annotations, brace-free expressions, and names that just work. But loft is statically typed and compiled to bytecode, while Python is dynamically typed and interpreted. This page documents the most important differences so that Python programmers know exactly what they gain and what they give up.

4.0.1. Variables — static types inferred at first assignment

```
x = 42           // integer — fixed at first assignment
x += 1
name = "Bob"     // text — a different variable, not a rebind
x = "oops"      // compile error: cannot assign text to integer
```

```
x = <span class="nm">42</span>           <span class="cm"># int inferred; can be
rebound to any type</span>
x += <span class="nm">1</span>
name = <span class="st">"Bob"</span>
x = <span class="st">"oops"</span>       <span class="cm"># fine — x is now a
str</span>
```

Upside Type errors are caught before the program runs. Renaming or reshaping data structures surfaces mismatches immediately. There is no equivalent of Python's runtime TypeError, AttributeError, or "NoneType has no attribute X" crash — those classes of bug are detected at compile time.

Downside Types are fixed at first assignment and cannot change. Python's dynamic typing genuinely speeds up prototyping: a function that accepts anything, a list that holds mixed types, or a variable that starts as an integer and becomes a float later are all natural in Python and impossible in loft. Adding a type annotation layer (mypy, pyright) gives Python static safety without giving up its flexibility.

4.0.2. Null — structured absence, not a universal wildcard

```
struct User {
    email: text,             // nullable by default
    age:   integer not null,  // can never be null
}
u = User { age: 30 };
if u.email == null { print("no email"); }
u.age = null;  // compile error: age is not null
```

```
<span class="kw">from</span> dataclasses <span class="kw">import</span> dataclass
<span class="kw">from</span> typing <span class="kw">import</span> Optional

<span class="en">@dataclass</span>
<span class="kw">class</span> <span class="en">User</span>:
    age: <span class="bi">int</span>
    email: Optional[<span class="bi">str</span>] = <span class="kw">None</span>
<span class="cm"># explicit Optional</span>
```

```
u = User(age=30)
if u.email is None:
    print("no email")
u.age = None  # allowed at runtime; mypy catches this
```

Upside Nullability is part of the struct definition — readable at a glance. not null fields are enforced by the compiler and additionally locked at runtime in debug builds, catching accidental null writes early in development. No Optional[T] wrapping needed; the comparison v == null is natural.

Downside The default is nullable, not non-nullable — the safe default is backwards from what static analysis advocates recommend. Python with mypy and Optional[T] annotation gives static guarantees that loft's runtime checks do not. Python's None also participates in truthiness testing (if not email:), pattern matching, and or chaining in ways that loft's null does not support.

4.0.3. Structs vs classes and dicts

```
struct Point { x: float, y: float }
p = Point { x: 1.0, y: 2.0 };
p.x += 0.5;
p.z;          // compile error: no field z on Point
p.x = "hi"; // compile error: cannot assign text to float

fn distance(self: const Point) -> float {
    sqrt(self.x * self.x + self.y * self.y) ?? 0.0
}
```

```
from dataclasses import dataclass
import math

@dataclass
class Point:
    x: float
    y: float

    def distance(self) -> float:
        return math.sqrt(self.x**2 + self.y**2)

p = Point(x=1.0, y=2.0)
p.z        # AttributeError at runtime (or dict: KeyError)
p.x = "hi" # allowed at runtime; mypy catches this
```

Upside Field access is checked at compile time — typos in field names are caught before any code runs. Struct memory layout is fixed and unboxed; integer and float fields live directly in memory with no heap allocation overhead. Methods are ordinary named functions — they can be added from any file, at any time, without modifying the struct definition.

Downside No inheritance, no __repr__, no operator overloading (__add__, __eq__, etc.), no properties or descriptors. Python's @dataclass generates __init__, __repr__, and __eq__ automatically. Loft structs are data holders; all display and comparison logic must be written by hand. Python also supports plain dicts as lightweight records, which is often more convenient for ad-hoc data.

4.0.4. while loop — yes; no loop or while … else

```
while !ready() { step(); }  // works

while length(queue)  0 {
    process(queue[0]);
    queue#remove;
}

// No loop keyword — infinite loop needs a large bound:
for _ in 0..2147483647 {
    if should_exit() { break; }
    step();
}
```

```
<span class="kw">while</span> <span class="kw">not</span> <span class="fn-
call">ready</span>():
    <span class="fn-call">step</span>()

<span class="kw">while</span> queue:
    <span class="fn-call">process</span>(queue.<span class="fn-call">pop</
span>(<span class="nm">0</span>))

<span class="kw">while</span> <span class="kw">True</span>:          <span
class="cm"># truly infinite</span>
    <span class="kw">if</span> <span class="fn-call">should_exit</span>(): <span
class="kw">break</span>
    <span class="fn-call">step</span>()
```

Upside while condition { } works exactly as expected. Filtered loops (for x in v if pred(x)) and loop attributes (x#first, x#count) add expressive power to for without needing a separate loop form.

Downside No while True: equivalent — infinite loops require a bounded for range as a workaround. No while … else (executes when the condition becomes false without a break), a Python pattern with no clean equivalent in loft.

4.0.5. Polymorphic enum dispatch vs isinstance

```
enum Shape {
    Circle { r: float },
```

```
    Rect   { w: float, h: float }
}
fn area(self: Circle) - float { PI * pow(self.r, 2.0) }
fn area(self: Rect)   - float { self.w * self.h }

s.area()   // dispatches on the runtime variant
```

```
import math
from dataclasses import dataclass

@dataclass
class Circle: r: float
@dataclass
class Rect:   w: float; h: float

def area(s):
    if isinstance(s, Circle):
return math.pi * s.r ** 2
    if isinstance(s, Rect):
return s.w * s.h

# Python 3.10+ structural pattern matching:
match s:
    case Circle(r=r): return math.pi * r ** 2
    case Rect(w=w, h=h): return w * h
```

Upside Each variant's behaviour lives in its own small, named function — easy to read, easy to navigate in an editor. Adding a new shape only requires a new fn area(self: NewShape) with no changes to existing code or a central dispatch function. The compiler warns when a variant has no implementation for a called method.

Downside Unlike Rust's match, loft does not enforce exhaustiveness — a missing variant implementation produces only a warning, not a compile error. Python 3.10+ structural pattern matching (match/case) fully destructs the matched value and is exhaustive when case _: is present. Python also supports inheritance-based polymorphism (class Circle(Shape)) and abstract base classes, giving far richer dispatch options.

4.0.6. String formatting — embedded expressions

```
msg  = "Hi {name}, score: {score:8.2}"
sign = "seed {n:+}"              // sign shows on integers: "seed +42"
hex  = "{n:#x}"
list = "{for x in 1..4 {x*2}}"   // [2,4,6]
pad  = "{count:08}"               // zero-padded (integers)
```

```
msg  = <span class="st">f"Hi {name}, score: {score:8.2f}"</span>
sign = <span class="st">f"seed {n:+}"</span>
hex  = <span class="st">f"{n:#x}"</span>
<span class="cm"># no loop in f-string; use a join:</span>
lst  = <span class="st">","</span>.<span class="fn-call">join</span>(<span
class="bi">str</span>(x*<span class="nm">2</span>) <span class="kw">for</span> x
<span class="kw">in</span> <span class="bi">range</span>(<span class="nm">1</
span>, <span class="nm">4</span>))  <span class="cm"># "2,4,6"</span>
pad  = <span class="st">f"{count:08}"</span>
```

Upside All loft strings are implicitly format strings — no f prefix needed. Inline for loops inside {} produce a bracketed list ([2,4,6]) without a separate join. Format specifiers mirror Python's f-string mini-language: width, precision, sign, alignment, zero-padding, and radix (#x, #o, b) all work — though the sign (+) and zero-pad (0N) flags apply to integer output only, and are dropped for floats.

Downside Python f-strings accept arbitrary expressions: method calls ({obj.method():.2f}), ternary expressions ({"yes" if ok else "no"}), and join operations inline. Loft restricts what can appear inside {}. Python also supports conversion flags (!r for repr, !s for str, !a for ASCII) and the = debug specifier ({x=} prints x=42); loft has none of these.

4.0.7. Collections — typed and built-in

```
nums:   vector<integer> = [1, 2, 3];
lookup: hash<text>    = {};
lookup["key"] = "value";
scores: sorted<integer> = {};
scores[user] = 95;  // O(log n) keyed insert

// Element type is enforced at compile time:
nums += ["x"];  // compile error: expected integer
```

```
nums   = [<span class="nm">1</span>, <span class="nm">2</span>, <span
class="nm">3</span>]          <span class="cm"># list — any element type</span>
lookup = {}                 <span class="cm"># dict — any key/value type</span>
lookup[<span class="st">"key"</span>] = <span class="st">"value"</span>

<span class="cm"># sorted dict requires an external package:</span>
<span class="kw">from</span> sortedcontainers <span class="kw">import</span>
SortedDict
scores = <span class="fn-call">SortedDict</span>()
scores[user] = <span class="nm">95</span>

nums.<span class="fn-call">append</span>(<span class="st">"x"</span>)
<span class="cm"># allowed at runtime</span>
```

Upside All collection types — vector, hash, and sorted map — are built in; no extra import or install needed. Element types are checked at compile time. The same [] indexing syntax works on all three. for x in v if pred(x) { v#remove; } safely removes the current element while iterating — something Python requires careful index management for.

Downside Python's built-in dict and list are among the most heavily optimised data structures in any scripting runtime. Python also has set and frozenset (no loft equivalent), ordered insertion semantics on dict (Python 3.7+), and an enormously expressive comprehension syntax ([x*2 for x in v if x > 0]). Loft's map() / filter() / reduce() higher-order functions allocate a new vector at each stage, while Python's generators are lazy and allocation-free until consumed.

4.0.8. Closures — same-scope capture works

```
// Capture works when called in the same scope:
offset = 10;
shift = fn(x: integer) -> integer { x + offset };
assert(shift(5) == 15, "shift");

// Non-capturing lambdas work with map/filter/reduce:
doubled = map([1, 2, 3], fn(x: integer) -> integer { x * 2 });
evens   = filter([1, 2, 3, 4], |x| { x % 2 == 0 });

// Capturing lambda with map:
offset = 10;
shifted = map(nums, fn(x: integer) -> integer { x + offset });
```

```
offset = <span class="nm">10</span>
shift = <span class="kw">lambda</span> x: x + offset  <span class="cm"># capture
works anywhere</span>
<span class="kw">assert</span> <span class="fn-call">shift</span>(<span
class="nm">5</span>) == <span class="nm">15</span>

nums = [<span class="nm">1</span>, <span class="nm">2</span>, <span
class="nm">3</span>, <span class="nm">4</span>, <span class="nm">5</span>]
doubled = <span class="bi">list</span>(<span class="bi">map</span>(<span
class="kw">lambda</span> x: x * <span class="nm">2</span>, nums))
evens   = <span class="bi">list</span>(<span class="bi">filter</span>(<span
class="kw">lambda</span> x: x % <span class="nm">2</span> == <span class="nm">0</
span>, nums))

<span class="cm"># capturing lambda passed to map:</span>
shifted = [x + offset <span class="kw">for</span> x <span class="kw">in</span>
nums]
```

Upside Same-scope capture works for integers, text, and mutable variables — no capture modes, no scope surprises. Both long-form (fn(x: integer) -> integer { x * 2 }) and short-form (|x| { x * 2 }) lambdas are supported. Named function references (fn name) are compile-checked. Higher-order functions (map, filter, reduce) accept both lambdas and fn-refs.

Downside Capture is by value at definition time — later mutations to the original variable do not affect the lambda (and vice versa). Python's lambda and nested def close over variables by reference, so mutations are shared. Python's list comprehensions ([x + offset for x in v]) are shorter than map(v, fn(x) { x + offset }).

4.0.9. No exception handling — file errors use FileResult

```
// Logic errors: assert aborts the program
fn divide(a: float, b: float) - float {
    assert(b != 0.0, "division by zero");
    a / b ?? 0.0
}

// Reading: check f#exists before using content
f = file("data.txt");
if f#exists {
    data = f.content();
} else {
    print("file not found");
}

// Mutating ops return a FileResult enum; match on the bare variant
result = delete("old.txt");
match result {
    Ok               = print("deleted"),
    NotFound         = print("not found"),
    PermissionDenied = print("permission denied"),
    IsDirectory      = print("is a directory"),
    _                = print("other error"),
}

// Or just check success:
if !move("a.txt", "b.txt").ok() { print("rename failed"); }
```

```
<span class="cm"># Logic errors: raise an exception</span>
<span class="kw">def</span> <span class="fn-call">divide</span>(a: <span
class="bi">float</span>, b: <span class="bi">float</span>) -> <span
class="bi">float</span>:
    <span class="kw">if</span> b == <span class="nm">0</span>:
        <span class="kw">raise</span> <span class="fn-call">ValueError</
span>(<span class="st">"division by zero"</span>)
    <span class="kw">return</span> a / b

<span class="kw">try</span>:
    result = <span class="fn-call">divide</span>(x, y)
<span class="kw">except</span> ValueError <span class="kw">as</span> e:
    <span class="bi">print</span>(<span class="st">f"error: {e}"</span>)

<span class="cm"># File errors: catch specific exception types</span>
<span class="kw">try</span>:
    <span class="kw">with</span> <span class="bi">open</span>(<span
class="st">"data.txt"</span>) <span class="kw">as</span> f:
        data = f.<span class="fn-call">read</span>()
<span class="kw">except</span> FileNotFoundError:
    <span class="bi">print</span>(<span class="st">"file not found"</span>)
```

```
<span class="kw">try</span>:
    os.<span class="fn-call">remove</span>(<span class="st">"old.txt"</span>)
<span class="kw">except</span> FileNotFoundError:
    <span class="bi">print</span>(<span class="st">"not found"</span>)
<span class="kw">except</span> PermissionError:
    <span class="bi">print</span>(<span class="st">"permission denied"</span>)
<span class="kw">except</span> IsADirectoryError:
    <span class="bi">print</span>(<span class="st">"is a directory"</span>)
```

Upside File errors are represented as a typed FileResult enum — Ok, NotFound, PermissionDenied, IsDirectory, Other — so every failure case is named and exhaustively matchable. No accidental exception swallowing, no hidden exit paths, no stack-unwinding overhead. The control flow of every loft function is straightforward to read.

Downside Logic errors (assert, panic) abort the entire program — there is no try/except to catch them. Python's exception hierarchy supports retry logic, cleanup via finally, and graceful degradation from arbitrary errors anywhere in the call stack — patterns that are not expressible in loft today.

4.0.10. Built-in parallel for-loops — par(...)

```
fn score(item: const Record) - integer { item.value * 2 }

total = 0;
for item in records par(s=score(item), 4) {
    total += s;   // results arrive in original order
}                 // 4 worker threads, no GIL
```

```
<span class="kw">from</span> concurrent.futures <span class="kw">import</span>
ProcessPoolExecutor

<span class="kw">def</span> <span class="fn-call">score</span>(item): <span
class="kw">return</span> item.value * <span class="nm">2</span>

<span class="cm"># ProcessPoolExecutor: bypasses GIL via separate processes</
span>
<span class="kw">with</span> <span class="fn-call">ProcessPoolExecutor</
span>(max_workers=<span class="nm">4</span>) <span class="kw">as</span> ex:
    results = <span class="bi">list</span>(ex.<span class="fn-call">map</
span>(score, records))
total = <span class="bi">sum</span>(results)

<span class="cm"># ThreadPoolExecutor: simpler but GIL limits CPU-bound work</
span>
```

Upside Built into the language — no import, no boilerplate. The GIL does not apply; worker threads run on separate OS threads inside the same process. Results arrive in the original vector order. The compiler validates the worker function signature at the call site. The thread count is set per call, making it easy to tune for the hardware.

Downside Workers can return primitives (integer, long, float, boolean), text, and inline enums — but not struct references. Context must be embedded as fields in the element struct; workers cannot capture local variables. Python's ProcessPoolExecutor works with any picklable object and gives full control over timeouts, cancellation, and error propagation.

4.0.11. Generic functions — pass-through only, no duck typing

```
// Pass-through generics work:
fn identity<T>(x: T) - T { x }
identity(42)        // integer
identity("hello")  // text

// Algorithms requiring operations on T still need per-type versions:
fn max_int(a: integer, b: integer) - integer {
    if a  b { a } else { b }
}
```

```
<span class="cm"># Duck typing: works for any T that supports ></span>
<span class="kw">def</span> <span class="fn-call">max_val</span>(a, b):
    <span class="kw">return</span> a <span class="kw">if</span> a > b <span
class="kw">else</span> b

<span class="cm"># With type hints (Python 3.12+):</span>
<span class="kw">from</span> typing <span class="kw">import</span> TypeVar
T = <span class="fn-call">TypeVar</span>(<span class="st">"T"</span>)
<span class="kw">def</span> <span class="fn-call">max_val</span>(a: T, b: T) ->
T:
    <span class="kw">return</span> a <span class="kw">if</span> a > b <span
class="kw">else</span> b
```

Upside Basic generic functions (pass-through, wrapping, forwarding) work out of the box. Collections (vector<T>, hash<T>, sorted<T>) are generic at the engine level, covering the most common need without any user-visible type parameter syntax.

Downside Operations on T require an explicit interface bound (<T: Ordered> for comparison, <T: Addable> for arithmetic, <T: Printable> for display); unbounded T stays opaque. Python's duck typing lets one function work on any type that supports the needed operation (len, >, +) with no annotation, and Python 3.12 TypeVar/Protocol add optional static checking on top.

4.0.12. Function signatures — defaults and named args, but no *args

```
fn connect(host: text, port: integer = 80, tls: boolean = true) - text { ... }

connect("localhost");                // uses defaults
connect("localhost", tls: false);    // named arg — skips port
connect("localhost", 443, false);   // positional — all explicit
```

```
<span class="kw">def</span> <span class="fn-call">connect</span>(host: <span
class="bi">str</span>, port: <span class="bi">int</span> = <span class="nm">80</
```

```
span>, tls: <span class="bi">bool</span> = <span class="kw">True</span>):
    ...

<span class="fn-call">connect</span>(<span class="st">"localhost"</span>)
<span class="cm"># uses defaults</span>
<span class="fn-call">connect</span>(<span class="st">"localhost"</span>,
tls=<span class="kw">False</span>) <span class="cm"># keyword arg — skips port</
span>

<span class="kw">def</span> <span class="fn-call">log</span>(*args,
**kwargs): ...   <span class="cm"># variadic</span>
```

Upside Default parameter values and named arguments work similarly to Python. connect("host", tls: false) skips defaulted middle parameters. Every call site is readable without jumping to the function definition.

Downside No *args or **kwargs — variadic dispatch must use a vector parameter. Python's flexible argument syntax also enables decorator patterns and configuration-driven code that is harder to express in loft.

4.0.13. Ecosystem — minimal standard library

```
// import "mylib" loads a .loft file relative to the script.
// The full standard library ships inside the interpreter binary;
// there is no package manager or external dependency system.

// Built-in: text, math, file I/O, collections,
//           logging, threading, image, lexer/parser
```

```
<span class="kw">import</span> numpy <span class="kw">as</span> np          <span
class="cm"># numerical arrays / SIMD</span>
<span class="kw">import</span> pandas <span class="kw">as</span> pd         <span
class="cm"># dataframes</span>
<span class="kw">import</span> requests             <span class="cm"># HTTP</
span>
<span class="kw">import</span> flask                <span class="cm"># web
framework</span>
<span class="kw">import</span> sqlalchemy           <span class="cm"># database
ORM</span>
<span class="kw">import</span> scikit_learn         <span class="cm"># machine
learning</span>
<span class="cm"># 500 000+ packages on PyPI, installable with:</span>
<span class="cm">#   pip install <package></span>
```

Upside Zero external dependencies — the interpreter is a single self-contained binary. Deployment is copying one file. There is no requirements.txt, no virtual environment, no version conflict, and no supply-chain risk. The built-in library covers text manipulation, math, file I/O, typed collections, parallel execution, image handling, and a full lexer/parser framework.

Downside Python's ecosystem is its defining advantage. NumPy, pandas, scikit-learn, TensorFlow, requests, Flask, SQLAlchemy, pytest, and hundreds of thousands of other packages are not available to loft programs. Any data science, web development, database integration, or protocol implementation task will require reimplementing from scratch what Python solves with a single pip install. For these domains, loft is the wrong tool today.

4.0.14. Exponentiation with ** or pow(); ^ is XOR

```
area  = PI * r ** 2.0      // ** exponentiation (like Python)
bits  = a | b              // bitwise OR
xor   = a ^ b              // bitwise XOR — not exponentiation
cube  = x ** 3             // integer cube — exact: 2 ** 10 == 1024
```

```
<span class="kw">import</span> math
area  = math.pi * r ** <span class="nm">2</span>   <span class="cm"># ** is
exponentiation</span>
bits  = a | b              <span class="cm"># bitwise OR</span>
xor   = a ^ b              <span class="cm"># bitwise XOR</span>
cube  = x ** <span class="nm">3</span>             <span class="cm"># integer
cube — exact</span>
cube  = x ** (<span class="nm">1</span>/<span class="nm">3</span>)         <span
class="cm"># cube root as float</span>
```

Upside Loft's ** exponentiation operator matches Python's — 2 ** 10 == 1024 on integers, 2.0 ** 3.0 == 8.0 on floats (pow(base, exp) is available too). ^ behaves as bitwise XOR in both loft and Python — no mismatch. Bitwise operator precedence matches: | (loose) → ^ → & → shifts → arithmetic (tight).

Downside Python's ** additionally handles complex numbers, and its three-argument pow(base, exp, mod) form computes modular exponentiation efficiently; loft has neither. Loft's pow() function itself operates on float/single only — use the ** operator for integer exponentiation.

5. Keywords

This page covers the building blocks that shape how your program makes decisions and repeats work: conditions, loops, breaks, and a few loop helpers that make common patterns shorter. Every example is verified with an assert so you can see the exact expected result.

```
fn main() {
```

5.0.1. Conditionals

'if' runs a block only when its condition is true. If the condition is false the block is skipped entirely — nothing else happens. 'panic' stops the program immediately with a message. During development it is a clear way to mark situations that should never occur.

```
  if 2 > 5 {
    panic("Incorrect test");
  }
```

Combine conditions with 'and' / 'or' (or their symbol equivalents '&&' / '||'). Note that '&' is the bitwise AND operator — it works on the individual 1s and 0s inside a number, which is different from the logical 'and' keyword that works on true/false values. An 'if/else' chain picks exactly one branch to execute.

```
  a = 12;
  if a > 10 and a & 7 == 4 {
    a += 1;
  } else {
    a -= 1;
  }
```

Text enclosed in double quotes can contain expressions inside curly braces — the expression is evaluated and its result is inserted into the text. 'assert' checks that its first argument is true; if it is false the second argument is printed as an error message.

```
  assert(a == 13, "Incorrect value {a} != 13");
```

5.0.2. if as an expression

Every block in Loft produces a value — the last expression inside it. That means you can use 'if/else' on the right-hand side of an assignment, picking between two values based on a condition. No ternary operator needed.

```
  b = if a == 13 {
    "Correct"
  } else {
    "Wrong"
  };
  assert(b == "Correct", "Logic expression");
```

5.0.3. Iteration

‘for x in a..b’ loops over the integers starting at a and stopping before b (exclusive upper bound). Use ‘..=’ to include the upper bound. Here we sum 1 + 2 + 3 + 4 + 5 = 15.

```
t = 0;
for i in 1..6 {
  t += i;
}
assert(t == 15, "Total was {t} instead of 15");
```

‘rev(range)’ reverses the direction of any range. ‘1..=5’ is inclusive, so it visits 5, 4, 3, 2, 1 in that order. Multiplying each digit into a running total builds the number 54321.

```
t = 0;
for i in rev(1..=5) {
  t = t * 10 + i;
}
assert(t == 54321, "Result was {t} instead of 54321");
```

5.0.4. Nested loops and break

Loft has two loop statements: ‘for’, which walks a range or a collection, and ‘while’, which repeats as long as its condition holds. A ‘for’ over a range carries its own upper bound, so it is the one to reach for; a ‘while’ does not, and ‘while true { }’ runs until something stops it. ‘break’ exits the nearest enclosing loop immediately. To exit an outer loop from inside an inner one, write ‘outerVar#break’. Here x#break leaves both loops when the product x*y reaches 16.

```
b = "";
for x in 1..5 {
  for y in 1..5 {
    if y > x {
      break;
```

this breaks the inner y loop

```
      }
      if x * y >= 16 {
        x# break;
      }
      if len(b) > 0 {
        b += "; ";
      }
      b += "{x}:{y}";
    }
  }
  assert(b == "1:1; 2:1; 2:2; 3:1; 3:2; 3:3; 4:1; 4:2; 4:3", "Incorrect sequence
'{b}'");
```

5.0.5. Skipping an iteration: continue

'continue' is the sibling of 'break': instead of leaving the loop it skips the rest of the current iteration and moves on to the next value. Here we sum 1..=5 but skip 3, so the total is 1 + 2 + 4 + 5 = 12.

```
t = 0;
for i in 1..=5 {
  if i == 3 {
    continue;
  }
  t += i;
}
assert(t == 12, "continue skipped 3: total was {t} instead of 12");
```

5.0.6. Loop helpers: #first and #count

Inside a loop body you have access to some useful metadata about the current iteration. 'x#first' is true only for the very first element that passes the filter. 'x#count' is a zero-based index counting only the elements that were not filtered out. An 'if' clause directly after 'in range' filters which values enter the loop — here we skip every value where x % 3 == 1, keeping 2, 3, 5, 6, 8, 9.

```
b = "";
for x in 1..=9 if x % 3 != 1 {
  if !x#first {
    b += ", ";
  }
  b += "{x#count}:{x}";
}
assert(b == "0:2, 1:3, 2:5, 3:6, 4:8, 5:9", "Sequence '{b}'");
```

5.0.7. Formatting a loop inline

A for loop can appear inside a format string. The results are collected into a bracketed, comma-separated list. A format specifier after the closing brace is applied to every element — ':02' means at least 2 digits, zero-padded. This is handy for building compact representations on the fly.

```
  assert("a{for x in 1..7 {x*2}:02}b" == "a[02,04,06,08,10,12]b", "Format
range");
}
```

6. Texts

Text is one of the most frequently used types in any real program — for output, for reading input, for building messages. This page shows you how Loft stores and manipulates text, how to measure it, slice it, search it, and format it exactly the way you need.

The key thing to know upfront: Loft stores text as a sequence of bytes using UTF-8 encoding — the same format used on the web. Plain ASCII letters, digits, and punctuation each take exactly one byte. Accented letters, emoji, and many non-Latin scripts take 2–4 bytes. Most operations count bytes rather than visible characters. That makes them fast and simple, but you need to keep multi-byte characters in mind when you slice by position.

```
fn main() {
```

6.0.1. Joining and measuring text

Use '+' to join two pieces of text into one. The result is a new value — neither of the originals is modified. Placing a value name inside curly braces inside a format string inserts the value as text. You can also control the width: '{a:4}' pads the value on the right to at least 4 characters.

```
    a = "1" + "2";
    assert("'{a:4}'" == "'12  '", "Formatting text");
    assert(len(a + "123") == 5, "Text length");
```

'len()' counts characters (Unicode code points), the human count: an emoji like '😃' is one character, a heart symbol '♥' is one, and each ASCII letter is one. For the UTF-8 byte length (bounds checks, byte indexing) use 'size()'.

```
    assert(len("😃") == 1, "Emoji is one character");
    assert(len("♥") == 1, "Heart is one character");
    assert(len("abc") == 3, "ASCII character count");
```

6.0.2. Reading individual characters

Square brackets with a single byte index give you the character at that position. For plain ASCII text every byte is one character, so indexing by position is straightforward. Loft will return the full character even if it spans multiple bytes.

```
    s = "ABCDE";
    assert(s[0] == 'A', "Character at byte 0");
    assert(s[2] == 'C', "Character at byte 2");
```

6.0.3. Slicing text

A slice 's[start..end]' gives you the bytes from 'start' up to but not including 'end'. You can omit the start to begin at 0, or omit the end to go all the way to the last byte. Loft snaps offsets to valid character boundaries so you can never accidentally cut a multi-byte character in half.

```
assert(s[0..2] == "AB", "Explicit start slice");
assert(s[..2] == "AB", "Open-start slice");
assert(s[1..4] == "BCD", "Sub-string by byte range");
assert(s[3..] == "DE", "Open-ended sub-string");
```

A negative end index counts backwards from the end of the text. '−1' means "stop one byte before the very last byte". Combined with a start offset this lets you trim a known suffix without knowing the exact length.

```
txt = "12😊🙃45";
assert(txt[2.. - 1] == "😊🙃4", "UTF-8 sub-string by byte range");
```

6.0.4. Iterating over characters

A 'for' loop over a text value visits one character at a time, even when characters span multiple bytes. Two loop helpers give you position information without any extra code: 'c#index' is the byte offset where the current character starts. 'c#next' is the byte offset immediately after the current character ends. This makes it easy to build a byte-offset map or to collect characters starting from a specific position.

```
result = "";
positions = [];
nexts = [];
for c in "Hi 😊!" {
  positions += [c#index];
  nexts += [c#next];
  if c#index >= 3 { result += c; }
}
assert(result == "😊!", "Character iteration was {result}");
assert("{positions}" == "[0,1,2,3,7]", "Character positions was {positions}");
assert("{nexts}" == "[1,2,3,7,8]", "Character nexts was {nexts}");
```

6.0.5. Searching inside text

These built-in functions answer common "does this text contain…?" questions. 'starts_with' and 'ends_with' check the boundaries. 'find' returns the byte offset of the first match, or null if not found. 'contains' is true if the needle appears anywhere in the text. All positions are byte offsets, consistent with 'len()' and slicing.

```
assert("something".starts_with("some"), "starts_with");
assert("something".ends_with("thing"), "ends_with");
assert("something".find("th") == 4, "find returns byte offset");
assert("a longer text".contains("longer"), "contains");
```

'trim()' removes spaces and other whitespace from both ends of a text value. It is useful when reading user input or parsing text from a file.

```
assert(trim("  hello  ") == "hello", "trim");
```

6.0.6. Escaping braces in format strings

Inside a format string '{' and '}' are special: they introduce an interpolated expression. To include a literal brace in the output, double it: '{{' produces '{' and '}}' produces '}'.

```
  brace_inner = "cd";
  assert("ab{{cd}}e" == "ab{{{brace_inner}}}e", "Escaping braces");
```

6.0.7. Aligning text in a fixed-width field

The ':' specifier controls alignment and width. '<' left-aligns the value, '>' right-aligns it (the default). The width can be a constant or a small arithmetic expression — here '2+3' evaluates to 5, giving a field of width 5.

```
  vr = "abc";
  assert("1{vr:<2+3}2{vr}3{vr:6}4{vr:>7}" == "1abc  2abc3abc   4    abc", "Text
alignment");
}
```

7. Integers

Numbers are at the heart of almost every program. This page covers the integer types Loft provides, the arithmetic and bitwise operations you can perform on them, how to convert between numbers and text, and what Loft does when something goes wrong — such as dividing by zero.

The default number type is ‘integer’: a 64-bit signed whole number, the same as i64 in Rust. It can hold values from about −9 quintillion to +9 quintillion (roughly ±9.2 × 10^18), so everyday counts and very large numbers both fit in the one type — there is no separate ‘long’. For decimal (fractional) numbers, see the Float page.

```
fn main() {
```

7.0.1. Converting between numbers and text

Wrapping a value in ‘{...}’ inside a string formats it as text. Going the other way, ‘as integer’ parses a text value into a number. If the text cannot be parsed, the result is null — not a crash.

```
  v = 4;
  assert("{v}" == "4", "Format integer as text");
  assert("123" as integer? == 123, "Parse text to integer");
  assert(!("abc" as integer?), "Unparseable text gives null");
```

7.0.2. Arithmetic and operator precedence

Loft follows standard mathematical precedence: ‘*’ and ‘/’ before ‘+’ and ‘-’. Bitwise operators (<<, &, ^) have their own precedence — when mixing them with arithmetic, parentheses make your intent clear and avoid surprises. Note: ‘^’ is XOR, not exponentiation. Use pow(base, exp) for powers.

```
  assert(1 + 2 * 4 == 9, "Multiplication before addition");
  assert(1 + 2 << 2 == 12, "Shift: (1+2) << 2 = 12");
  assert(0x0a8 & 15 == 8, "Bitwise AND masks low 4 bits");
  assert(42 ^ 0b111111 == 21, "Bitwise XOR");
  assert(105 % 100 == 5, "Modulus (remainder)");
  assert(pow(2.0, 3.0) == 8.0, "pow() for exponentiation");
```

‘abs()’ returns the absolute value — the distance from zero, always positive.

```
  assert(1 + abs(-2) == 3, "abs(-2) == 2");
```

7.0.3. Division by zero — produces null and keeps running

Most languages crash on division by zero. Loft does NOT: a divide (or modulo) by zero is uncomputable, so it produces null and execution CONTINUES — the spreadsheet model, where one bad cell shows an error but the rest still recalculate. This holds the same everywhere (development, test, and production) — one bad calculation never halts the run.

- **Undefended** (bare `1 / 0`): you get null, and at this unguarded site

```
loft also reports a warning so the divide-by-zero is not invisible.
```

- **Defended** (`1 / 0 ?? fallback`, or a following `if x != null`):

```
`??` supplies a non-null fallback and the warning is suppressed —
you have explicitly handled the case.
```

```
  a = 2 * 2;
  a -= 4;
```

a is now 0

```
  assert(!(12 / a ?? null), "Division by zero gives null when defended with `??
null`");
```

7.0.4. Warning when you write a literal zero divisor

When the divisor is a literal 0 written directly in your source code, loft warns you while reading your code (before running it), because that is almost certainly a mistake:

```
n / 0   // warning: Division by constant zero
n % 0   // warning: Modulo by constant zero
```

Use a variable (like ‘a’ above) when you intentionally want null-on-zero division without a warning.

7.0.5. Embedding integers in text

```
  assert("a{12}b" == "a12b", "Integer in format string");
```

A full expression can appear inside ‘{...}’, not just a variable name.

```
  assert("a{1 + 2 * 3}b" == "a7b", "Expression in format string");
```

7.0.6. Number format specifiers

After a ‘:’ inside ‘{...}’ you can control how a number is displayed:

```
#x  — hexadecimal with '0x' prefix
o   — octal
b   — binary
+   — always show a sign (+ or -)
N   — minimum field width (space-padded on the left)
0N  — minimum field width (zero-padded on the left)
```

```
  assert("a{1+2+32:#x}b" == "a0x23b", "Hex format with 0x prefix");
  assert("{12:o}" == "14", "Octal");
  assert("{12:+4}" == " +12", "Sign and width");
  assert("{1:03}" == "001", "Zero-padded width");
  assert("{42:b}" == "101010", "Binary");
```

Hexadecimal literals in source code accept both lower and upper case digits.

```
    assert(0xff == 255, "Lowercase hex literal");
    assert(0xFF == 255, "Uppercase hex literal");
    assert(0x2A == 42, "Uppercase hex digit");
```

7.0.7. Large integers

Because ‘integer’ is 64-bit, values far past the old 2-billion limit work directly — no separate type is needed. Arithmetic, comparisons, and the format specifiers above all behave the same at these magnitudes.

```
    big = 1000000000 * 5;
    assert(big == 5000000000, "Integers well past 2 billion");
    assert("{big}" == "5000000000", "Large integer formatted as text");
```

7.0.8. Common pitfall: integer overflow

A calculation whose result is too large to represent overflows — and in loft an overflow is uncomputable, so the result is null. It never wraps to a silently-wrong number, and it never crashes; execution continues with the null, exactly like divide-by-zero above. Check the result for null (or keep intermediate values in range) when multiplying or summing very large numbers.

```
}
```

8. Boolean

A boolean value is either ‘true’ or ‘false’. Booleans appear naturally wherever you make a decision: in ‘if’ conditions, loop filters, and comparisons. Loft adds a third state called ‘null’ — meaning “no value” — which behaves like false whenever a boolean is expected.

This design means you rarely need to write a separate null-check: ‘!x’ is true both when x is false and when x is null.

```
fn main() {
```

A comparison produces a boolean result directly. ‘!’ flips a boolean: true ⟶ false, false ⟶ true.

```
  assert(!(3 > 2 + 4), "3 is not greater than 6");
  assert(true as text == "true", "Convert boolean to text");
```

8.0.1. Logical operators: and / or

‘and’ (also ‘&&’) is true only when both sides are true. ‘or’ (also ‘||’) is true when at least one side is true. Both skip the right side when the left side already determines the result. For example, ‘false and X’ is always false, so X is never evaluated. This matters when the right side could produce null or has a side effect.

```
  assert(1 > 0 and 2 > 1, "Both conditions true");
  assert(1 > 2 or 3 > 2, "Second condition true");
  assert(!(1 > 2 && 3 > 2), "First false → whole 'and' is false");
  assert(!(1 > 2 || 3 > 4), "Both false → 'or' is false");
```

8.0.2. Null in boolean context

Division by zero (and other failed operations) produce null when defended with `?? null` (or `if result != null { ... }` after the assignment). Null in a boolean context is treated as false, so ‘!’ is true. This lets you write guard clauses without a separate null-check syntax:

```
if !result { ... handle missing value ... }
```

Even without `?? null`, a bare `1 / 0` never halts: it yields null and logs a warning, behaving the same everywhere — development, test, and production. The `?? null` above just marks the case as handled, which suppresses that warning.

```
  zero = 0;
  assert(!(12 / zero ?? null), "null from division-by-zero is false-like when
defended");
  assert(12 > 0, "positive integer is true");
```

8.0.3. Bitwise operators

While ‘and’/‘or’ work on true/false values, bitwise operators work on the individual bits of an integer. They are useful for flags, masks, and low-level data manipulation.

```
'&'  — keeps only bits set in BOTH operands (AND)
'|'  — keeps bits set in EITHER operand (OR)
'<<' — shift bits left N positions  (×2^N)
'>>' — shift bits right N positions (÷2^N)
```

Tip: '&' binds less tightly than comparison operators. Use parentheses when you mix bitwise and comparison expressions in the same condition.

```
    assert((0x0f & 0xa8) == 8, "Bitwise AND: keeps only bits in both");
    assert((0xf0 | 0x0f) == 0xff, "Bitwise OR: combines bits from either");
    assert(1 << 4 == 16, "Left shift: 1 × 2^4 = 16");
    assert(256 >> 3 == 32, "Right shift: 256 ÷ 2^3 = 32");
```

8.0.4. Negation

'!' is the logical NOT operator. It flips any boolean expression.

```
    assert(!false, "not false is true");
    assert(!(1 > 2), "not false comparison is true");
```

8.0.5. Formatting booleans

Booleans can be embedded in a format string like any other value. The '^' alignment specifier centres the value in a field of given width. '<' aligns left, '>' aligns right (right-alignment is the default).

```
    assert("1{true:^7}2" == "1 true  2", "Centred boolean in field of width 7");
    assert("{false}" == "false", "Plain boolean format");
```

8.0.6. '&' vs 'and'

'&' is bitwise AND on integers; 'and' (or '&&') is logical AND on booleans. Writing 'a & b' where both sides are booleans is a compile error — the compiler requires 'and' or '&&' for boolean logic.

```
    flag_a = 3 > 1;
```

true

```
    flag_b = 4 > 2;
```

true

```
    assert(flag_a and flag_b, "Correct: logical AND on two booleans");
}
```

9. Float

A ‘float’ stores a number with a decimal point, like 3.14 or −0.001. It gives you about 15 significant digits of precision — enough for scientific work, games, and most real-world maths. When you write a number with a decimal point in Loft, it is automatically a float.

```
fn main() {
```

Write a decimal point in a literal and Loft treats the whole expression as a float. The usual arithmetic operators (+, -, *, /) all work the way you would expect.

```
  assert(1.5 + 0.5 == 2.0, "Float addition");
  assert(3.0 / 2.0 == 1.5, "Float division");
  assert(2.0 * 1.5 == 3.0, "Float multiplication");
```

9.0.1. Undefined results are null (‘float?’)

Some float operations have no real answer for some inputs: dividing or taking ‘%’ by zero, ‘sqrt’ of a negative, ‘ln’/‘log’ of zero or a negative, ‘asin’/‘acos’ outside −1..1, or an out-of-domain ‘pow’. Rather than crash, Loft yields null for these — so ‘/’, ‘%’, ‘sqrt’, ‘ln’, ‘log’, ‘pow’, ‘asin’ and ‘acos’ each produce a NULLABLE float (‘float?’). The null then propagates: any further arithmetic on a null stays null. Discharge it with ‘?? fallback’ when you need a plain float (just as ‘“1.5” as float?’ below is a float? you can default).

```
  assert((sqrt(-1.0) ?? 0.0) == 0.0, "sqrt of a negative is null");
  assert(((sqrt(-1.0) + 5.0) ?? 0.0) == 0.0, "null propagates through
arithmetic");
```

The ‘f’ suffix selects single-precision floats, which take half the memory of a regular float but have fewer decimal digits of accuracy. This trade-off is common in graphics and audio code where exact decimal values matter less than speed or memory.

```
  x = 0.1f + 2 * 1.0f;
  assert(x == 2.1f, "Single precision float");
```

‘as float’ converts an integer to a float so you can mix them in calculations. Putting a float inside ‘{...}’ converts it to text for display. You can also go the other way: ‘“1.5” as float’ parses the text back to a number.

```
  assert(3 as float == 3.0, "Integer to float");
  assert("1.5" as float? == 1.5, "Text to float");
  assert("{1.5}" == "1.5", "Float formatted as text");
```

9.0.2. Formatting Floats

Put a colon after the value inside ‘{...}’ to control how it looks. ‘{value:width.precision}’ — ‘width’ is the minimum number of characters printed (padded with spaces on the left); ‘precision’ fixes the number

of decimal places. You can use either part on its own: '{value:.2}' just fixes decimal places, '{value:5}' just sets the minimum width.

```
    assert("{1.2:4.2}" == "1.20", "Float with width and precision");
    assert("{334.1:.2}" == "334.10", "Float with precision only");
    assert("{1.4:5}" == "  1.4", "Float with width only");
```

9.0.3. Math Functions

'PI' is a built-in constant (approximately 3.14159265358979). Multiplying it by 1000 and rounding gives 3142, which confirms the value is correct.

```
    assert(round(PI * 1000.0) == 3142.0, "PI constant");
```

'pow(base, exponent)' raises a number to a power — this is exponentiation. Note: '^' in Loft is bitwise XOR, NOT exponentiation. Always use pow() for powers. 'log(value, base)' is the inverse: log(x, b) answers "b to the what power equals x?" Here: 4^5 = 1024, and log(1024, 2) = 10 because 2^10 = 1024.

```
    assert(log(pow(4.0, 5), 2) == 10.0, "log base 2 of 4^5");
```

'sin' and 'cos' work in radians, not degrees. A full circle is 2*PI radians; a half circle (180 degrees) is PI radians. sin(PI) should be exactly zero but computers use approximations for decimal numbers, so you may get something like 0.0000000001 instead. We use ceil() to round it up to 0. cos(PI) is exactly −1, so the whole expression ceil(0 + −1 * 1000) lands at −1000.

```
    assert(ceil(sin(PI) + cos(PI) * 1000) == -1000.0, "sin and cos");
```

'abs()' returns the absolute value — the distance from zero, always non-negative. Useful any time you care about magnitude but not direction.

```
    assert(abs(-2.5) == 2.5, "Absolute value of float");
```

'round()' picks the nearest whole number (0.5 rounds up). 'ceil()' always rounds up to the next whole number, even for 2.01. 'floor()' always rounds down to the previous whole number, even for 2.99. All three give back a float, not an integer — so 3.0, not 3.

```
    assert(round(2.6) == 3.0, "round up");
    assert(round(2.4) == 2.0, "round down");
    assert(ceil(2.1) == 3.0, "ceil");
    assert(floor(2.9) == 2.0, "floor");
```

Single-precision floats work inside format strings just like regular floats.

```
    assert("a{0.1f + 2 * 1.0f}b" == "a2.1b", "Single-precision format");
}
```

10. Functions

Functions let you give a reusable piece of logic a name so you can call it from multiple places without copying code. This page covers: declaring functions, default argument values, reference parameters (so a function can modify the caller's variable), early return, const parameters, and type-based dispatch.

10.0.1. Declaring Functions

The keyword 'fn' starts a function definition. List parameters as 'name: type' separated by commas. '-> type' declares what the function gives back. The last expression in the body is returned automatically — no 'return' needed there. Default values let callers skip optional arguments.

```
fn greet(name: text, greeting: text = "Hello") -> text {
  greeting + ", " + name + "!"
}
```

10.0.2. Default arguments that involve other arguments

A default value can be any expression, not just a literal — including one that references earlier parameters, which are in scope by the time their default is evaluated. So 'end' can default directly to the length of the earlier 't'.

```
fn substring(t: text, start: integer = 0, end: integer = len(t)) -> text {
  t[start..end]
}
```

10.0.3. Named Arguments

When a function has several parameters with defaults, you can skip middle ones by naming the arguments you want to provide. Positional arguments come first; once you use a name, all remaining arguments must also be named.

```
fn connect(host: text, port: integer = 8080, tls: boolean = true) -> text {
  "{host}:{port}:{tls}"
}
```

10.0.4. Reference Parameters and Defaults

A function with no '-> type' is called for its side effects, not its result. Adding '&' before a parameter type makes it a reference: the function receives a direct link to the caller's variable and can read or write it. Without '&', Loft copies the value in, so changes inside the function stay local. Default values (written '= value' after the type) are substituted when the caller omits that argument.

```
fn add(a:  & integer, b: integer, c: integer = 0) {
  a += b + c;
}
```

10.0.5. Early Return

Use 'return value;' to exit a function before reaching the end. This is handy for guard clauses: handle the special cases first, then write the normal path without extra nesting.

```
fn classify(n: integer) -> text {
  if n < 0 {
    return "negative";
  }
  if n == 0 {
    return "zero";
  }
  "positive"
}
```

10.0.6. Const and Reference Parameters

'const' on a parameter tells the compiler "this function must never change this value." The compiler enforces it — any assignment to a const parameter is a compile error. '&' is the opposite promise: "this function will modify this value." Declaring '&' without actually writing to the parameter is also a compile error. These rules make it easy to read a function signature and know what it does to its inputs.

```
fn scale(a: integer, factor: integer) -> integer {
  a * factor
}
```

10.0.7. Type-Based Dispatch

You can define two functions with the same name as integer as their parameter types differ. Loft picks the right one at compile time based on the type of the argument you pass.

```
fn describe_int(v: integer) -> text {
  "integer:{v}"
}
```

```
fn describe_text(v: text) -> text {
  "text:{v}"
}
```

10.0.8. Function References

'fn <name>' creates a reference to a named function that you can store or pass around. The compiler checks that the name exists and is a function — a typo is a compile error. The result has type 'fn(param_types) -> return_type' and can be:

```
- stored in a variable,
- called directly with 'f(args)', and
- passed as a parameter to another function.
```

```
fn double_it(x: integer) -> integer {
  x * 2
}
```

```
fn negate_it(x: integer) -> integer {
  - x
}
```

```
fn apply_fn(f:  fn(integer) -> integer, x: integer) -> integer {
  f(x)
}
```

10.0.9. Lambda Expressions

A lambda is an anonymous (unnamed) function written right where you need it. Use the full form when you want to be explicit about types:

```
fn(param: type) -> return_type { body }
```

Use the short form (pipe-bar syntax) when the call site already knows the types:

```
|param| { body }
```

Both forms work anywhere a function reference is expected — especially with map, filter, and reduce. The short form does not allow type annotations; use the full fn(...) form when you need explicit types. Lambdas that need outer variables can capture them: see the Closures page.

```
fn main() {
```

Passing both arguments explicitly overrides the default.

```
  assert(greet("World", "Hi") == "Hi, World!", "Explicit argument");
```

Leaving out the second argument causes Loft to use the default “Hello”.

```
  assert(greet("World") == "Hello, World!", "Default argument");
```

‘add’ takes ‘a’ by reference, so every call updates the original variable ‘v’. Watch how v accumulates: 1 -> 3 -> 8.

```
  v = 1;
  add(v, 2);
```

v = 1 + 2 = 3

```
  add(v, 4, 1);
```

v = 3 + 4 + 1 = 8

```
  assert(v == 8, "Reference parameter: {v}");
```

'classify' uses early returns to handle each case separately.

```
assert(classify(-5) == "negative", "Classify negative");
assert(classify(0) == "zero", "Classify zero");
assert(classify(3) == "positive", "Classify positive");
```

'scale' uses plain parameters (no 'const', no '&'): it only reads them to compute the product, so the caller's values are never touched.

```
assert(scale(3, 7) == 21, "scale(3,7)");
```

Loft selects the right function based on the type of the argument.

```
assert(describe_int(42) == "integer:42", "Integer describe");
assert(describe_text("hi") == "text:hi", "Text describe");
```

A function reference is stored in a variable with type 'fn(...) -> ...'. Calling it looks exactly like a regular function call.

```
f = double_it;
assert(f(5) == 10, "fn-ref stored and called: {f(5)}");
```

Pass function references as arguments to higher-order functions.

```
assert(apply_fn(double_it, 7) == 14, "fn-ref as arg (double)");
assert(apply_fn(negate_it, 3) == -3, "fn-ref as arg (negate)");
```

Lambdas with map, filter, and reduce. Use |...| short form — types are inferred from the call site.

```
nums = [1, 2, 3, 4, 5];
doubled = map(nums, |x| { x * 2 });
assert(doubled[0] == 2, "lambda map first element");
assert(doubled[4] == 10, "lambda map last element");
```

filter keeps only elements where the lambda returns true.

```
evens = filter(nums, |x| { x % 2 == 0 });
assert(evens[0] == 2, "filter evens first");
assert(evens[1] == 4, "filter evens second");
```

reduce collapses the whole list into one value. The lambda receives the running total (acc) and the next element (x).

```
total = reduce(nums, 0, |acc, x| { acc + x });
assert(total == 15, "reduce sum: {total}");
```

Use fn(...) long form when you need explicit types — for example when storing a lambda in a variable or passing it without call-site context.

```
  negate = fn(x: integer) -> integer { -x };
  assert(negate(7) == -7, "stored lambda: {negate(7)}");
  assert(apply_fn(|x| { x * x }, 6) == 36, "lambda as arg: 6^2");
```

— Named arguments —

```
  assert(connect("example.com") == "example.com:8080:true", "all defaults");
  assert(connect("example.com", tls: false) == "example.com:8080:false", "skip
port");
  assert(connect(host: "example.com", tls: false) == "example.com:8080:false",
"all named");
  assert(connect("example.com", port: 443, tls: false) ==
"example.com:443:false", "mixed");
  assert(greet(name: "World") == "Hello, World!", "named with default greeting");
```

— Default argument that depends on another argument — ‘end’ defaults to len(t), computed from the earlier ‘t’ parameter.

```
  assert(substring("hello", 1, 3) == "el", "explicit start and end");
  assert(substring("hello", 2) == "llo", "end defaults to len(t)");
  assert(substring("hello") == "hello", "both defaults");
}
```

11. Vector

A vector is an ordered list of values that can grow and shrink while your program runs. Every element must have the same type — you cannot mix integers and text in one vector. Write a vector literal with square brackets: '[1, 2, 3]'. Loft automatically manages the storage, so vectors grow as you add elements without you having to say how large they will be up front.

11.0.1. Transforming vectors: map, filter, reduce

'map', 'filter', and 'reduce' each take a function and apply it to the vector. Pass the function using 'fn <name>' to refer to a named function by name.

```
map(v, f)          — apply f to every element; returns a new vector
filter(v, pred)    — keep only elements for which pred returns true
reduce(v, init, f) — combine all elements into a single value, starting from init
```

```
fn triple(x: integer) -> integer {
  x * 3
}
```

```
fn is_even_n(x: integer) -> boolean {
  x % 2 == 0
}
```

```
fn sum_acc(acc: integer, x: integer) -> integer {
  acc + x
}
```

```
fn mul_acc(acc: integer, x: integer) -> integer {
  acc * x
}
```

```
fn main() {
```

Create a vector with a literal and loop over it with 'for'. Two special loop annotations help when you need to know where you are:

```
'v#first'  is true only on the very first iteration — useful for skipping
separators.
'v#index'  holds the zero-based position of the current element.
```

```
  x =[1, 3, 6, 9];
  b = "";
  for v in x {
    if !v#first {
      b += " ";
    }
```

```
    b += "{v#index}:{v}"
  }
  assert(b == "0:1 1:3 2:6 3:9", "result {b}");
```

'+=' appends another vector to the end of an existing one. You can filter and delete elements in a single pass:

```
Add 'if condition' after 'in vector' to visit only matching elements.
Write 'v#remove' inside the loop body to delete the current element.
```

Here we keep only multiples of 3 by removing everything else. Note: you cannot append to a vector (v += [...]) while iterating over it — that is a compile error because the loop would then visit the new elements too, which could loop forever.

```
  x +=[12, 14, 15];
  for v in x if v % 3 != 0 {
    v#remove;
  }
  assert("{x}" == "[3,6,9,12,15]", "result {x}");
```

11.0.2. Clearing

'v.clear()' removes all elements, setting the length to 0. The underlying storage is kept so appending afterwards is efficient.

```
  x.clear();
  assert(x.len() == 0, "clear empties the vector");
  x += [99];
  assert(x[0] == 99, "append after clear works");
```

11.0.3. Slicing

A slice gives you a window into part of a vector without copying it. 'v[a..b]' contains elements at positions a, a+1, ..., b-1 (b is excluded). 'v[a..]' goes from position a to the very last element. 'v[..b]' goes from the beginning up to (but not including) position b.

```
  pows =[1, 2, 4, 8, 16];
  assert("{pows[1..3]}" == "[2,4]", "Sub-vector");
  assert("{pows[3..]}" == "[8,16]", "Open-ended sub-vector");
  assert("{pows[..3]}" == "[1,2,4]", "Open-start sub-vector");
```

Accessing an index that does not exist is safe: Loft returns null instead of crashing. You can check for null with '!' (logical not) because null is falsy.

```
  assert(!pows[10], "Out-of-bounds access returns null");
```

11.0.4. Comprehensions

A comprehension is a concise way to build a new vector from a formula. Syntax: '[for variable in range { expression }]' Add 'if condition' to include only elements that satisfy the condition. This is much shorter than creating an empty vector and appending in a loop.

```
  evens =[for n in 1..10 if n % 2 == 0 {
    n
  }];
  assert("{evens}" == "[2,4,6,8]", "Filtered comprehension");
  doubled =[for n in 1..6 {
    n * 2
  }];
  assert("{doubled}" == "[2,4,6,8,10]", "Doubled comprehension");
```

Embedding a for loop directly inside a format string '{...}' produces a formatted list. The format specifier after ':' is applied to every element in the result. ':02' means "at least 2 digits wide, padded with zeros on the left".

```
  assert("{for n in 1..7 {n*2}:02}" == "[02,04,06,08,10,12]", "Formatted vector
loop");
```

11.0.5. Reverse Iteration

Wrap a range in 'rev()' to step through elements from the last index to the first. 'rev(0..=3)' covers indices 0, 1, 2, 3 — the '=' makes the upper end inclusive. Here we visit pows[3]=8, pows[2]=4, pows[1]=2, pows[0]=1, building 8421 digit by digit.

```
  c = 0;
  for e in pows[rev(0..=3)] {
    c = c * 10 + e;
  }
  assert(c == 8421, "Reverse sub-vector iteration");
```

You can fill a vector with many copies of the same value using '; count' syntax:

```
[SomeStruct { field: value }; 16]
```

This creates 16 identical copies in one expression. See 08-struct.loft for examples.

11.0.6. Passing vectors to functions

When you pass a vector to a function, the function receives a **slice** — a start position and a length inside the storage of the caller. This is efficient because no data is copied, but it has an important consequence: the function can read and modify existing elements (because it shares the same storage), but it cannot grow or shrink the vector. Appending with '+=' inside the function creates a local copy that the caller never sees.

To let a function append to the caller's vector, mark the parameter with '&'. This tells the compiler to propagate structural changes (appends, clears) back to the caller when the function returns.

```
fn append_one(v: &vector<integer>, x: integer) { v += [x]; }
```

Without '&', only element-level mutations are visible to the caller:

```
fn set_first(v: vector<integer>, x: integer) { v[0] = x; }  // caller sees the
change
fn try_push(v: vector<integer>, x: integer) { v += [x]; }   // caller does NOT
see the append
```

The same rule applies to slices: 'v[2..5]' passed to a function is a narrower window into the same storage, so element writes are visible but appends are not.

11.0.7. Higher-order functions

'map' applies a function to every element and returns a new vector.

```
  nums =[1, 2, 3, 4, 5];
  tripled = map(nums,  triple);
  assert("{tripled}" == "[3,6,9,12,15]", "map triple: {tripled}");
```

'filter' keeps only elements for which the predicate returns true.

```
  evens2 = filter(nums,  is_even_n);
  assert("{evens2}" == "[2,4]", "filter evens: {evens2}");
```

'reduce' folds all elements into a single value, starting from an initial accumulator. Argument order: reduce(vector, initial_value, combiner).

```
  total = reduce(nums, 0,  sum_acc);
  assert(total == 15, "reduce sum: {total}");
  product = reduce(nums, 1,  mul_acc);
  assert(product == 120, "reduce product: {product}");
```

You can chain these calls: filter first, then map, then reduce.

```
  result = reduce(map(filter(nums,  is_even_n),  triple), 0,  sum_acc);
  assert(result == 18, "filter+map+reduce: {result}");
}
```

12. Structs

A struct groups related values under one name. Instead of keeping a product's name, price, and stock count in three separate variables that can drift out of sync, you bundle them together into a Product struct. Each piece of data inside the struct is called a field. You read and write fields using dot notation: 'product.price'. Here is a simple product record. All four fields are plain integers or text.

```
struct Product {
  name: text,
  price: integer,
  stock: integer
}
```

12.0.1. Field Constraints

You can restrict what values a field may hold. 'limit(min, max)' rejects any value outside that range at runtime. 'not null' tells the field that zero is a real data value — without it, zero is treated as "no value" (null). Colour channels can all be zero (pure black is a valid colour), so all three need 'not null'. Fields you omit in a constructor receive zero (or null for nullable fields) by default.

```
struct Colour {
  r: integer limit(0, 255),
  g: integer limit(0, 255),
  b: integer limit(0, 255)
}
```

12.0.2. Methods

A method is a function whose first parameter is named 'self'. Loft uses the type of 'self' to decide which struct the method belongs to. Call it with dot notation: 'c.to_hex()'. A method can read fields via 'self.field' and return any type.

```
fn to_hex(self: Colour) -> integer {
  self.r * 0x10000 + self.g * 0x100 + self.b
}
```

A method can also return a completely new struct value. Write '-> StructName' as the return type and construct the value in the body. The original struct is not modified — the caller gets a brand-new copy.

```
fn dimmed(self: Colour) -> Colour {
  Colour {r: self.r / 2, g: self.g / 2, b: self.b / 2 }
}
```

12.0.3. Defaults and Computed Fields

Write '= expression' after the type to set a stored default, filled at construction. Use 'computed(expression)' for a field that recalculates every time you read it — it takes no space in the record and always reflects the current state. Inside both, '$' refers to the struct.

```
struct Item {
  name: text,
  name_length: integer = len($.name)
}
```

```
struct Circle {
  radius: float,
  area: float computed(3.14159265 * $.radius * $.radius)
}
```

12.0.4. Storing many structs in a vector

'Area' uses smaller integer types to save memory: 'u16' holds values 0–65535, and 'u8' holds values 0–255. This matters when you have millions of records (such as tiles in a game map) and memory usage counts.

```
struct Area {
  height: u16,
  terrain: u8,
  water: u8,
  direction: u8
}
```

12.0.5. Value structs

Prefix a struct with 'value' to give it value (copy) semantics and zero heap overhead: it is stored inline inside a record or vector element, exactly like a plain integer. Reading one out of a field or element yields an independent COPY — writing to that copy never changes the original. Use a 'value struct' for small wrapper types (a point, a colour, a timestamp) that should be as cheap as the number they wrap.

A plain (reference) struct differs only when you read one OUT of somewhere. Two spellings look alike, so it is worth keeping them apart:

- Binding a whole value COPIES it, always. After 'b = a' the two names are

```
independent, and writing 'b' never changes 'a'. The copy goes all the way
down, including nested structs and vectors.
```

- Reading a struct-typed field or element gives you a VIEW: 'w = o.inner'

```
and 'c = v[i]' name the record in place, so 'w.n = 42' changes 'o.inner'
too. A view is not a copy — it points at the interior of what you read it
from. (A vector-typed field is a whole value, so 'av = bx.v' copies.)
```

- Write '&' when you want a live link where a bind would copy: 'b = &a' and

```
'av = &bx.v' both write through to the source.
```

```
value struct Point {
  x: integer,
  y: integer
}
```

```
fn main() {
```

Constructing a struct: name the fields you want to set. Fields you leave out default to zero (or null for nullable fields).

```
  apple = Product {name: "Apple", price: 120, stock: 50 };
  assert(apple.price == 120, "price field: {apple.price}");
  assert(apple.name == "Apple", "name field: {apple.name}");
```

You can read and write individual fields after construction.

```
  apple.stock -= 1;
  assert(apple.stock == 49, "stock after one sale: {apple.stock}");
```

A field omitted from the constructor gets zero as its default.

```
  col = Colour {r: 128, b: 128 };
  assert(col.g == 0, "omitted green channel defaults to zero");
```

Formatting a struct shows all fields compactly.

```
  assert("{col}" == "{{r:128,g:0,b:128}}", "Struct compact formatting");
```

':j' produces JSON output — a text format used by many web APIs and config tools.

```
  assert("{col:j}" == "{{\"r\":128,\"g\":0,\"b\":128}}", "JSON format");
```

Call a method on a variable using dot notation.

```
  purple = Colour {r: 128, b: 128 };
  assert("{purple.to_hex():x}" == "800080", "hex method result");
```

You can call a method directly on a constructor expression.

```
  assert(Colour {r: 255, g: 0, b: 0 }.to_hex() == 0xff0000, "method on
constructor");
```

'dimmed' returns a new Colour with each channel halved. The original variable is unchanged.

```
  dark = purple.dimmed();
  assert(dark.r == 64, "dimmed r: {dark.r}");
```

```
  assert(dark.b == 64, "dimmed b: {dark.b}");
  assert(dark.g == 0, "dimmed g: {dark.g}");
```

Stored defaults are filled once at construction time.

```
  it = Item {name: "hello" };
  assert(it.name_length == 5, "stored default name_length: {it.name_length}");
```

computed() fields recalculate on every access — changing radius updates area.

```
  ci = Circle { radius: 5.0 };
  assert(ci.area > 78.0, "computed area: {ci.area}");
  ci.radius = 10.0;
  assert(ci.area > 314.0, "recomputed area: {ci.area}");
```

Fill a vector with copies of the same struct using '; count' syntax. This creates 16 Area records, all with the same initial values.

```
  map =[Area {height: 0, terrain: 1, water: 1, direction: 1 };
  16];
  assert("{map[3]}" == "{{height:0,terrain:1,water:1,direction:1}}", "tile
record");
  map[3].height = 200;
  assert(map[3].height == 200, "individual tile update");
```

A value struct binding is a snapshot (copy), not a shared view.

```
  pts =[Point {x: 1, y: 2 }, Point {x: 3, y: 4 }];
  q = pts[0];
```

q is an independent COPY of element 0

```
  q.x = 99;
  assert(pts[0].x == 1, "value struct copy does not write back");
  assert(q.x == 99, "the copy itself did change");
```

A PLAIN struct answers the same two questions differently, so here they are side by side. Reading one out of an element gives a view, and writing the view reaches the element.

```
  shelf =[Product {name: "Pear", price: 90, stock: 10 }];
  item = shelf[0];
  item.stock = 7;
  assert(shelf[0].stock == 7, "a plain struct element read is a view");
```

Binding the whole value copies it, so the original keeps its own stock.

```
    spare = item;
    spare.stock = 3;
    assert(item.stock == 7, "a whole-value bind copies");
    assert(spare.stock == 3, "and the copy moves on its own");
```

Write '&' to get a live link where a bind would otherwise copy.

```
    linked = &item;
    linked.stock = 1;
    assert(item.stock == 1, "'&' binds a live link");
```

12.0.6. sizeof

'sizeof(Type)' returns the packed byte size used when the type is stored as a struct field or vector element. Range-constrained integer types like u8 and u16 report their packed size, not the 4-byte stack slot size.

```
    assert(sizeof(integer) == 8, "integer: 8 bytes (post-2c i64 storage)");
    assert(sizeof(u8) == 1, "u8: 1 byte (packed)");
    assert(sizeof(u16) == 2, "u16: 2 bytes (packed)");
    assert(sizeof(Colour) == 3, "Colour: 3 × u8 = 3 bytes");
    assert(sizeof(Area) == 5, "Area: u16 + 3 × u8 = 5 bytes");
}
```

13. Enums

An enum defines a fixed set of named choices. Instead of using raw numbers like 0 = north, 1 = east — which nobody can read six months later — you give each option a name. The compiler then checks every comparison and conversion, so a typo becomes a compile error instead of a silent wrong answer. Plain enums are just names that can be compared, ordered, and converted to and from text. Their order matches their declaration order.

```
enum Direction {
  North,
  East,
  South,
  West
}
```

13.0.1. Enums that carry data

Sometimes different choices have different shapes of data. A 'Circle' needs one number (radius); a 'Rect' needs two (width and height). Struct enums let each variant carry its own fields, keeping all the shape kinds in one type while still letting you work with each variant naturally.

```
enum Shape {
  Circle {radius: float },
  Rect {width: float, height: float }
}
```

13.0.2. Methods that work on any variant

Write the same function name for each variant, each taking 'self' of that variant's type. Loft automatically picks the right version at runtime based on which variant is actually stored in the variable. This lets you call shape.area() without caring which kind of shape it is.

```
fn area(self: Circle) -> float {
  PI * pow(self.radius, 2)
}
```

```
fn area(self: Rect) -> float {
  self.width * self.height
}
```

Polymorphic methods can return text just as well as numbers. Each variant can produce its own human-readable description.

```
fn describe(self: Circle) -> text {
  "circle with radius {self.radius}"
}
```

```
fn describe(self: Rect) -> text {
  "rect {self.width} by {self.height}"
}
```

13.0.3. Enum methods

Plain enum variants can also have methods. The 'self' parameter carries the current direction value, and the method can return any type. Here 'opposite()' flips North↔South and East↔West.

```
fn opposite(self: Direction) -> Direction {
  if self == North {
    South
  } else if self == South {
    North
  } else if self == East {
    West
  } else {
    East
  }
}
```

```
fn main() {
```

Plain enum values are ordered by declaration position. North comes first (smallest), West comes last (greatest).

```
  d = Direction.East;
  assert(d == East, "Enum equality");
  assert(d != North, "Enum inequality");
  assert(d > North, "East declared after North, so it is greater");
  assert(Direction.West > South, "West declared last, so it is greatest");
```

Convert between text and enum in both directions.

```
  assert("{d}" == "East", "Format plain enum as text");
  assert("West" as Direction == West, "Parse text to enum");
```

Use a plain enum method like any other method.

```
  assert(d.opposite() == West, "opposite of East is West");
  assert(Direction.North.opposite() == South, "opposite of North is South");
```

Struct enum variants are constructed exactly like regular structs.

```
  c = Circle {radius: 1.0 };
  assert(c.radius == 1.0, "Circle radius field");
  r = Rect {width: 4.0, height: 5.0 };
  assert(r.width * r.height == 20.0, "Rect manual area");
```

Calling area() on a Circle runs the Circle version; on a Rect it runs the Rect version — chosen automatically at runtime.

```
  assert(round(c.area() * 1000) == round(PI * 1000), "Circle area = PI*r^2");
  assert(r.area() == 20.0, "Rect area = width*height");
```

describe() uses format strings with field access inside each variant's method.

```
  assert(c.describe() == "circle with radius 1", "describe circle:
{c.describe()}");
  assert(r.describe() == "rect 4 by 5", "describe rect: {r.describe()}");
```

13.0.4. Stubs for missing implementations

If a variant intentionally has no implementation of a method, the compiler emits a warning. Provide an empty-body stub to silence it:

```
fn area(self: SomeVariant) -> float { }
```

A stub returns null at runtime and suppresses the warning.

13.0.5. Match expressions on enums

Match picks a code path based on the active variant. You must handle every variant, or include a _ wildcard arm that catches the rest.

```
  axis = match d {
    North | South => "vertical",
    East | West => "horizontal"
  };
  assert(axis == "horizontal", "or-pattern: East matches East|West");
```

When a variant has fields, name them inside braces to use them in the arm body.

```
  label = match c {
    Circle { radius } => "r={radius}",
    Rect { width, height } => "{width}x{height}"
  };
  assert(label == "r=1", "field destructuring in match arm");
```

13.0.6. Guard clauses

An arm can have an `if` guard after the pattern. If the guard fails, matching falls through to the next arm. Because the guard can fail, a guarded arm alone does not prove the variant is handled — you still need a wildcard _ or an unguarded arm for that variant.

```
  area = match c {
    Circle { radius } if radius > 0.0 => PI * radius * radius,
    _ => 0.0
```

```
};
assert(round(area * 1000) == round(PI * 1000), "guarded match on Circle");
```

13.0.7. Scalar match

Match also works on integers, text, floats, booleans, and characters. Arms can be literals, ranges, `null`, or `\_`.

```
grade = match 85 {
  90..=100 => "A",
  80..90   => "B",
  _        => "C"
};
assert(grade == "B", "range pattern 80..90 matches 85");
```

Or-patterns work on scalars too.

```
kind = match 2 {
  1 | 2 | 3 => "low",
  _         => "high"
};
assert(kind == "low", "scalar or-pattern");
```

A `null` pattern matches when the value is absent (e.g. a defended division by zero — `?? null` marks the division as handled, so no undefended-site warning is logged; the result is null either way).

```
zero = 0;
check = match 1 / zero ?? null {
  null => "absent",
  _    => "present"
};
assert(check == "absent", "null pattern matches div-by-zero when defended");
```

Character literals work in match arms.

```
vowel = match 'e' {
  'a' | 'e' | 'i' | 'o' | 'u' => true,
  _ => false
};
assert(vowel, "character or-pattern");
}
```

14. Sorted

A ‘sorted’ collection holds records in order by one or more fields you choose as the sort criteria (called “key fields”). You can look up a record by those fields instantly and iterate all records in that order. The collection stays sorted as you add new elements — no manual sorting needed. Declare the key fields inside angle brackets: ‘field’ sorts A⟶Z / 0⟶9 (ascending), ‘-field’ sorts Z⟶A / 9⟶0 (descending).

```
struct Elm {
  key: text,
  value: integer
}
```

```
struct Db {
  map: sorted < Elm[-key] >
}
```

@FTR-037 — a `sorted` collection keeps its order as records arrive, descending on a `-key`, and answers a lookup by that key

```
fn main() {
```

14.0.1. Adding Elements

You initialise a sorted collection the same way as a vector. The elements are automatically placed in the right position as you add them.

```
  db = Db {map: [
  Elm {key: "One", value: 1 },
  Elm {key: "Two", value: 2 },
  Elm {key: "Three", value: 3 },
  Elm {key: "Four", value: 4 }
  ] };
```

14.0.2. Looking Up by Key

Use square brackets with the key value to find a record instantly. If the key is not present the result is null — you can check it with ‘!’.

```
  assert(db.map["Two"].value == 2, "Key lookup: Two");
  assert(db.map["Three"].value == 3, "Key lookup: Three");
  assert(!db.map["Five"], "Missing key returns null");
  assert(!db.map[null], "Null key returns null");
```

Append new elements with ‘+=’; they are placed in the correct sorted position.

```
  db.map +=[Elm {key: "Zero", value: 0 }];
  assert(db.map["Zero"].value == 0, "Newly added element");
```

14.0.3. Iterating in Order

A ‘for’ loop over a sorted collection visits elements in key order. Here the key is ‘-key’ (descending text), so the order is: Zero, Two, Three, One, Four.

```
sum = 0;
for v in db.map {
  sum = sum * 10 + v.value;
}
```

Zero(0), Two(2), Three(3), One(1), Four(4) ⟶ 0*10+2=2, *10+3=23, *10+1=231, *10+4=2314

```
assert(sum == 2314, "Sorted iteration total: {sum}");
```

14.0.4. Iterating in Reverse

Wrap the collection in rev() to visit elements from last to first key order. Here the collection is sorted by ‘-key’ (descending text), so the stored order is Zero, Two, Three, One, Four. rev() visits them in the opposite order.

```
rev_sum = 0;
for v in rev(db.map) {
  rev_sum = rev_sum * 10 + v.value;
}
```

Four(4), One(1), Three(3), Two(2), Zero(0) ⟶ 4*10+1=41, *10+3=413, *10+2=4132, *10+0=41320

```
assert(rev_sum == 41320, "Reverse iteration total: {rev_sum}");
```

14.0.5. Loop Helpers: #first, #count, #index, and #remove

‘v#first’ is true for the very first element visited. ‘v#count’ is a running index starting at 0. ‘v#index’ is a sequential counter over the visited elements, also starting at 0 — on a sorted collection it advances in step with ‘#count’. ‘v#remove’ removes the current element while iterating.

```
first_key = "";
labels = "";
index_sum = 0;
for v in db.map {
  if v#first {
    first_key = v.key
  }
  if !v#first {
    labels += ","
  }
```

‘#index’ matches ‘#count’ here: both run 0,1,2,3,4 over the five elements.

```
  assert(v#index == v#count, "index tracks count: {v#index} vs {v#count}");
  index_sum += v#index;
```

```
  labels += "{v#count}:{v.key}"
}
assert(first_key == "Zero", "First element: {first_key}");
assert(labels == "0:Zero,1:Two,2:Three,3:One,4:Four", "Labels: {labels}");
assert(index_sum == 10, "0+1+2+3+4 = {index_sum}");
```

Remove elements with value > 2 while iterating.

```
for v in db.map if v.value > 2 {
  v#remove
}
sum2 = 0;
for v in db.map {
  sum2 += v.value
}
```

Remaining: Zero(0), Two(2), One(1) ⟶ sum = 3

```
assert(sum2 == 3, "Sum after remove: {sum2}");
```

14.0.6. Removing by Key

Assigning null to a sorted subscript removes the element with that key. Removing a key that is not present is a safe no-op.

```
db2 = Db {map: [
Elm {key: "A", value: 10 },
Elm {key: "B", value: 20 },
Elm {key: "C", value: 30 }
] };
db2.map["B"] = null;
assert(!db2.map["B"], "B was removed");
assert(db2.map["A"].value == 10, "A still present");
assert(db2.map["C"].value == 30, "C still present");
db2.map["missing"] = null;
```

no-op; does not panic

```
sum3 = 0;
for v in db2.map {
  sum3 += v.value
}
assert(sum3 == 40, "Sum after key removal: {sum3}");
```

Note: ‘#index’ works on a sorted collection — it is a sequential counter that advances one per visited element (see the loop above). It is only ‘index<T>’ collections that reject ‘#index’ — use ‘#count’ there instead (see 11-index).

14.0.7. Iterating a Key Range

The range examples live in `main\_ranges` below; run them here, because a file with a `main` runs only its `main`.

```
  main_ranges();
}
```

```
struct Word {
  name: text,
  score: integer
}
```

```
struct Dict {
  words: sorted<Word[name]>
}
```

```
fn main_ranges() {
```

14.0.8. Iterating a Key Range

Use an `if` guard inside a `for` loop to visit only records whose key falls within a range. Because the collection is sorted the guard still sees all elements, but the body only runs for the matching ones.

```
  d = Dict { words: [
    Word { name: "apple",      score: 5 },
    Word { name: "banana",     score: 3 },
    Word { name: "cherry",     score: 8 },
    Word { name: "date",       score: 2 },
    Word { name: "elderberry", score: 1 }
  ] };
```

Closed range: keys >= “banana” and <= “date”.

```
  range_total = 0;
  for w in d.words if w.name >= "banana" && w.name <= "date" {
    range_total += w.score
  }
  assert(range_total == 13, "banana(3)+cherry(8)+date(2) = {range_total}");
```

14.0.9. Open-ended Range (tail)

All records from “cherry” onward to the end of the sorted order.

```
  tail_count = 0;
  for w in d.words if w.name >= "cherry" {
    tail_count += 1
  }
  assert(tail_count == 3, "cherry, date, elderberry = {tail_count}");
```

14.0.10. Open-ended Range (head)

All records up to and including “banana”.

```
  head_names = "";
  for w in d.words if w.name <= "banana" {
    head_names += w.name + " "
  }
  assert(head_names == "apple banana ", "apple and banana: '{head_names}'");
```

14.0.11. Range outside the for (building a result vector)

Collect matching records into a vector for later use.

```
  subset = [for w in d.words if w.name >= "banana" && w.name <= "cherry"
{ w.score }];
  assert(subset[0] == 3, "first in range: {subset[0]}");
  assert(subset[1] == 8, "second in range: {subset[1]}");
}
```

filling and finding values

15. Index

An ‘index’ lets you find records instantly by key and iterate over ranges of keys in order. It supports multi-part keys: you can sort by a primary key and break ties with a secondary key. Declare the key fields inside angle brackets: ‘field’ sorts ascending, ‘-field’ descending. Note: each record can only belong to one index at a time.

```
struct Elm {
  nr: integer,
  key: text,
  value: integer
}
```

```
struct Db {
  map: index < Elm[nr, -key] >
}
```

@FTR-038 — an `index` on a multi-part key: lookup by key, ordered iteration, and a second key breaking ties in the other direction

```
fn main() {
```

15.0.1. Adding and Looking Up Records

Fill the index just like a vector. Elements are automatically kept in key order. This index is sorted first by ‘nr’ (ascending), then by ‘key’ (descending) for ties.

```
  db = Db {map: [
  Elm {nr: 101, key: "One", value: 1 },
  Elm {nr: 92, key: "Two", value: 2 },
  Elm {nr: 83, key: "Three", value: 3 },
  Elm {nr: 83, key: "Four", value: 4 },
  Elm {nr: 83, key: "Five", value: 5 },
  Elm {nr: 63, key: "Six", value: 6 }
  ] };
```

Provide all key fields together inside brackets to find a record. Supply fewer fields to match all records that share a prefix.

```
  assert(db.map[101, "One"].value == 1, "Key lookup");
  assert(!db.map[12, ""], "Missing key returns null");
  assert(!db.map[83, "One"], "Wrong secondary key returns null");
```

15.0.2. Iterating in Order

A ‘for’ loop visits all elements in key order.

```
  total = 0;
  for r in db.map {
```

```
  total += r.value;
}
assert(total == 21, "Sum of all values: {total}");
```

15.0.3. Range Queries

Provide a range in the first key position to visit only the matching slice. '[83..92, "Two"]' means: nr from 83 up to (not including) 92, and key from "Two" down (since the key is sorted descending, "Two" is the start of that descending segment).

```
sum = 0;
for v in db.map[83..92, "Two"] {
  sum = sum * 10 + v.value;
}
```

Elements matched: (83, Three, 3), (83, Four, 4), (83, Five, 5)

```
assert(sum == 345, "Range iteration result: {sum}");
```

15.0.4. Loop Helpers: #first and #count

'r#first' is true for the very first element visited. 'r#count' is a running counter starting at 0. Note: #index is not meaningful on index collections — use #count instead.

```
first_nr = 0;
count_total = 0;
for r in db.map {
  if r#first {
    first_nr = r.nr
  }
  count_total = r#count
}
assert(first_nr == 63, "First element by key: {first_nr}");
assert(count_total == 5, "Total elements: {count_total}");
```

15.0.5. Removing by Key

Assigning null to an index subscript removes the element with that exact key combination. Removing a key that is not present is a safe no-op.

```
db.map[92, "Two"] = null;
assert(!db.map[92, "Two"], "element (92, Two) was removed");
assert(db.map[101, "One"].value == 1, "element (101, One) still present");
db.map[999, "Z"] = null;
```

no-op; does not panic

```
total2 = 0;
for r in db.map {
```

```
    total2 += r.value
  }
  assert(total2 == 19, "Sum after key removal: {total2}");
}
```

16. Hash

A ‘hash’ lets you find a record by its key in the same time whether you have 10 records or 10 million — there is no searching through a list, just a direct jump to the answer. Unlike ‘sorted’ and ‘index’, a hash has no meaningful order — you use it purely for fast lookups. List the key fields in brackets: ‘hash<Type[field]>’. Combine a hash with a vector when you want both fast lookup and a stable iteration order.

```
struct Keyword {
  name: text
}
```

```
struct Data {
  h: hash < Keyword[name] >
}
```

16.0.1. Combining Hash and Vector

Pairing a hash with a vector on the same record type gives you the best of both worlds: the hash for instant lookup by key, and the vector for iterating in insertion order.

```
struct Count {
  t: text,
  v: integer
}
```

```
struct Counting {
  entries: vector < Count >,
  lookup: hash < Count[t] >
}
```

```
fn fill(c: Counting) {
  c.entries =[
 {t: "One", v: 1 },
 {t: "Two", v: 2 },
 {t: "Three", v: 3 },
 {t: "Four", v: 4 }
  ]
}
```

```
fn main() {
```

16.0.2. Basic Lookup

Fill a hash the same way you fill a vector. Look up by key with square brackets. A found record is truthy; a missing key returns null.

```
d = Data { };
d.h =[ {name: "one" }, {name: "two" }];
d.h +=[ {name: "three" }, {name: "four" }];
assert(d.h["three"], "Key exists");
assert(!d.h["None"], "Missing key returns null");
```

16.0.3. Hash + Vector Together

Here 'entries' is the vector (keeps insertion order) and 'lookup' is the hash (fast access). Both fields point to the same records — adding to one automatically updates the other.

```
c = Counting { };
fill(c);
assert(c.lookup["Three"].v == 3, "Hash lookup: Three");
assert(c.lookup["One"].v == 1, "Hash lookup: One");
assert(!c.lookup["Five"], "Missing key returns null");
```

Iterate in insertion order via the vector; look up by name via the hash. The vector also supports #first, #count, and #remove inside the loop.

```
add = 0;
for item in c.entries {
  add += item.v;
}
assert(add == 10, "Sum via vector iteration: {add}");
```

16.0.4. Removing a Key

Assigning null to a hash subscript removes that element. Removing a key that is not present is a safe no-op.

```
d.h["three"] = null;
assert(!d.h["three"], "three was removed");
assert(d.h["one"], "one still present");
d.h["missing"] = null;
```

no-op; does not panic

16.0.5. Iterating a Hash

`for e in h { ... }` walks the hash in ascending key order. The loop variable has the same shape as a `sorted\<T\>` iteration — a `reference\<T\>` with `\#index`, `\#count`, and `\#first` available. Multi-field keys sort lexicographically.

Under the hood the compiler builds a one-shot sorted snapshot of the hash's record-numbers and walks it, so iteration cost is O(n) + O(n log n) per `for`, not amortised across calls. For a hot loop, pair the hash with a `vector\<T\>` or `sorted\<T\[k\]\>` on the same record type and iterate the companion collection.

`e\#remove` is rejected at compile time — the snapshot is detached from the hash, so removing from it would not actually remove from the hash. Use `h\[key\] = null` to remove an entry.

```
  sum = 0;
  for e in c.lookup { sum += e.v; }
  assert(sum == 10, "Sum via hash iteration: {sum}");
}
```

17. File

A file handle lets you read and write files without worrying about when the OS opens or closes them. The file opens on the first read or write and closes automatically when the handle goes out of scope.

17.0.1. Inspecting the File System

`file(path)` creates a File handle without opening anything yet. `f\#format` tells you what kind of path you are looking at: TextFile (plain text), LittleEndian (binary, bytes stored least-significant first — common on most PCs), BigEndian (binary, bytes stored most-significant first — common in network protocols), Directory, or NotExists. `lines()` reads a text file and returns it as a vector of lines. `exists(path)` is a convenience shorthand for checking that a path is not NotExists. `delete(path)` removes a file and returns a `FileResult` enum; use `.ok()` to check success. Clean up any leftover files from a previous interrupted run. `\#cwd` opts this program into cwd-relative paths (CLI-tool semantics): a relative path resolves against the working directory, not the program's own directory. Without it, relative paths re-home next to the program (the portable-bundle default) — see @PLN9.

```
#cwd
fn cleanup() {
  delete("test.bin");
  delete("test2.bin");
  delete("buffer.bin");
}
```

```
struct Buffer {
  size: i32,
  data: vector < single >
}
```

```
fn main() {
  cleanup();
```

Asking for the format of a directory path returns Directory, not a file format. `lines()` reads the named text file and splits it on newlines.

```
  ex = file("tests/example");
  assert(ex#format == Directory, "example is a directory");
  c = file("tests/example/config/terrain.txt").lines();
  assert(c[1] == "    terrain = [", "Line was '{c[1]}'");
```

`exists` and `delete` are safe to call when the file is absent. `delete` returns false rather than crashing when nothing is there.

```
  assert(!exists("test.bin"), "File should not exist before the test.");
  assert(!delete("nonexistent_xyz.bin").ok(), "delete on missing file not ok");
```

17.0.2. Writing a Text or Binary File

Wrapping the file handle in a block ensures the file closes the moment the block ends. Set `f\#format` to LittleEndian or BigEndian before writing binary data. Use `f += value` to append the raw bytes of any scalar value (u8, u16, i32, integer, single, float, or text). An `integer` is 8 bytes wide (see the size assertions below).

```
{f = file("test.bin");
 assert(f#format == NotExists, "File should not exist yet.");
 f#format = BigEndian;  // BigEndian: most-significant byte written first.
 f += 0 as u8;
 f += 1 as u8;
 f += 0x203 as u16;
 f += 0x4050607;
 f += 0x8090a0b0c0d0e0f;
```

`f\#size` returns the total number of bytes written so far. Post-2c: integer is 8 bytes, so total = 1 + 1 + 2 + 8 + 8 = 20.

```
 assert(f#size == 20, "Should have written 20 bytes.");
```

Text is written as raw UTF-8 bytes with no length prefix.

```
 f += "Hello world!";
 assert(f#size == 32, "Size should be 32 bytes (20 + 12).");
}  // The file closes when the handle goes out of scope.
```

`move(src, dst)` renames a file. It refuses if the destination already exists.

```
assert(exists("test.bin"), "File should exist after writing.");
assert(move("test.bin", "test2.bin").ok(), "Could not move the file.");
```

17.0.3. Reading Back What You Wrote

When you open an existing file the default format is TextFile. Set `f\#format` to match the format used when writing before you read any bytes. `f\#read(n) as T` reads exactly n bytes and interprets them as type T. `f\#index` is the byte offset where the last read started. `f\#next` is the current read position; assign to it to seek anywhere.

```
{f = file("test2.bin");
 assert(f#format == TextFile, "The default format is TextFile.");
 f#format = LittleEndian;
```

The file was written as BigEndian, so reading the same 4 bytes as LittleEndian produces a byte-swapped value — that is intentional here and confirms that the bytes were stored in the order you chose.

```
 v = f#read(4) as i32;
 assert(v == 0x3020100, "BigEndian bytes 0..3 read as LittleEndian i32.");
```

```
  assert(f#index == 0, "Last read started at byte 0.");
  assert(f#next == 4, "Next read starts at byte 4.");
```

Seek to byte 20 to skip past the integers and read the text directly. (Post-2c: i32+i64 header = 1+1+2+8+8 = 20 bytes before the text.)

```
  f#next = 20;
  s = f#read(5) as text;
  assert(s == "Hello", "Partial text read from offset 20.");
  assert(f#index == 20, "Last read started at byte 20.");
  assert(f#next == 25, "Next position after 5-byte text read.");
```

Requesting more bytes than remain simply returns whatever is left.

```
  rest = f#read(100) as text;
  assert(rest == " world!", "Read continues to end of file.");
  assert(f#next == f#size, "Position should be at end of file.");
```

You can seek back to any position and re-read.

```
  f#next = 0;
  assert(f#read(4) as i32 == 0x3020100, "Seek-and-reread matches original.");
 }
  assert(delete("test2.bin").ok(), "Could not remove the test file.");
```

17.0.4. Working with Vectors and Struct Data

`f += vector\<T\>` writes every element in sequence as raw bytes, which is the fastest way to dump a whole collection to disk. Setting `f\#size = n` truncates or zero-extends the file to exactly n bytes — useful for discarding the tail after you have finished writing.

```
 {f = file("buffer.bin");
  f#format = LittleEndian;
  ints =[1, 2, 3, 4];
  f += ints;
  assert(f#size == 32, "Four integers = 32 bytes (8 each, post-2c)");
```

Truncate to the first two integers.

```
  f#size = 16;
  assert(f#size == 16, "Truncated to 16 bytes");
 }
  assert(delete("buffer.bin").ok(), "Could not remove buffer.bin after vector
write test.");
```

Write a count followed by individual float values. `sizeof(T)` returns the byte width of a type, so you can compute the correct read length without hard-coding magic numbers.

```
{buf = file("buffer.bin");
 buf#format = LittleEndian;
 buf += 4 as i32;
 buf += 1.1f;
 buf += 1.2f;
 buf += 2.1f;
 buf += 2.2f;
}
```

Read the count, then use it to read exactly that many floats directly into a struct field. This pattern lets you serialise and deserialise structs with variable-length data cleanly. A sized `f\#read` answers a raw byte buffer, so name the type you are reading it back as: the `as` is what turns those bytes into elements.

```
{b = Buffer { };
 f = file("buffer.bin");
 f#format = LittleEndian;
 n = f#read(4) as i32;
 floats = f#read(n * sizeof(single)) as vector<single>;
 b.data = floats;
 assert(len(b.data) == 4, "Read four floats back into the field.");
 assert((b.data[0] ?? 0.0f) == 1.1f, "First float survives the round-trip.");
}
 assert(delete("buffer.bin").ok(), "Could not remove buffer.bin.");
```

17.0.5. Error handling

File operations that can fail return a `FileResult` enum. Call `.ok()` to check success. The program does not crash on failure — you decide what to do.

```
  result = delete("this_file_does_not_exist.txt");
  assert(!result.ok(), "delete missing file returns not-ok");
```

Reading a non-existent file produces an empty result, not a crash.

```
  ghost = file("no_such_file.txt");
  assert(ghost#format == NotExists, "non-existent file has NotExists format");
```

`move` reports on the FILES, not on the shape of the path: a missing source is NotFound whether or not the path contains `..` (loft#712).

```
  assert(!move("any.txt", "../escape.txt").ok(), "move with a missing source
fails");
```

Writing to a read-only or invalid path also returns a FileResult. Always check `.ok()` after `delete`, `move`, `mkdir`, and `mkdir\_all`.

```
}
```

18. Image

PNG support is not built into loft — it lives in the `imaging` library, one of the installable packages in the loft registry. Add it to your project once with

```
loft install imaging
```

and import it at the top of any file with `use imaging;`. The library gives you three things: a `Pixel` struct (red / green / blue, each 0–255), an `Image` struct (name, width, height, and a flat `vector\<Pixel\>`), and the functions to move between an `Image` in memory and a `.png` file on disk.

```
file(path).png()        decode a PNG file into an Image (null if it cannot be
read)
image.save_png(path)    encode an Image back to a PNG file (true on success)
pixel.value()           pack r,g,b into a single 0xRRGGBB integer
```

Because the library builds a small native (Rust) helper the first time you use it, the very first run compiles that helper — later runs are instant.

`\#cwd` opts this program into cwd-relative paths so the bundled example file below resolves against the working directory (see 13-file). It must appear before the `use` line, which itself must come before any definition.

```
#cwd
use imaging;
```

```
fn main() {
```

18.0.1. Loading a PNG

`file(path).png()` reads and decodes the whole file in one call. Afterwards the width, height, file name, and every pixel are already in memory.

```
  img = file("tests/example/map.png").png();
  assert(img.width == 256, "width={img.width}");
  assert(img.height == 256, "height={img.height}");
```

`img.name` is just the file name, without the directory part.

```
  assert(img.name == "map.png", "name={img.name}");
```

`img.data` is a flat `vector\<Pixel\>` laid out row by row, so its length is always width * height.

```
  assert(len(img.data) == img.width * img.height, "pixel count {len(img.data)}");
```

18.0.2. The Pixel Struct

Each element of `img.data` is a `Pixel` with three channels — r, g, and b — each an integer from 0 (no intensity) to 255 (full intensity). `pixel.value()` packs the three channels into one 0xRRGGBB integer, handy for comparison or storage.

```
  first = img.data[0];
  assert(first.r >= 0 && first.r <= 255, "red channel in range: {first.r}");
  assert(first.value() == first.r * 0x10000 + first.g * 0x100 + first.b,
    "packed colour = {first.value()}");
```

18.0.3. Reading a Pixel by Coordinate

The pixel at column x, row y lives at index `y \* width + x`. Here we read the centre pixel of the 256x256 image.

```
  centre = img.data[128 * img.width + 128];
  assert(centre.r >= 0 && centre.g >= 0 && centre.b >= 0, "centre pixel is
valid");
```

18.0.4. Scanning Every Pixel

A `for` loop over `img.data` visits every pixel in row order. Count how many pixels are “bright” using a simple average-of-channels brightness.

```
  bright = 0;
  for px in img.data {
    brightness = (px.r + px.g + px.b) / 3;
    if brightness > 200 {
      bright += 1
    }
  }
  assert(bright >= 0 && bright <= len(img.data), "bright pixel count: {bright}");
```

18.0.5. Building and Saving an Image

You can also create an `Image` from scratch: fill a `vector\<Pixel\>`, wrap it in an `Image` with a matching width and height, and call `save\_png`. Here we make a 2x2 swatch — red, green, blue, and grey — and write it to a scratch file.

```
  swatch = imaging::Image {
    name: "swatch",
    width: 2,
    height: 2,
    data: [
      imaging::Pixel { r: 255, g: 0,   b: 0 },
      imaging::Pixel { r: 0,   g: 255, b: 0 },
      imaging::Pixel { r: 0,   g: 0,   b: 255 },
      imaging::Pixel { r: 128, g: 128, b: 128 }
    ]
  };
```

Write to a scratch file in the working directory, then delete it below so the example leaves nothing behind. A cwd-relative path (with `\#cwd`) works on every OS — a hard-coded `/tmp/…` would not exist on Windows.

```
  scratch = "loft-doc-swatch.png";
  saved = swatch.save_png(scratch);
  assert(saved, "save_png reported success");
```

18.0.6. Round-Tripping

Load the file we just wrote and confirm the pixels survived the encode/decode cycle unchanged.

```
  reloaded = file(scratch).png();
  assert(reloaded.width == 2 && reloaded.height == 2, "reloaded dimensions");
  assert(reloaded.data[0].r == 255 && reloaded.data[0].g == 0 &&
reloaded.data[0].b == 0,
    "reloaded red pixel");
  assert(reloaded.data[3].r == 128 && reloaded.data[3].g == 128 &&
reloaded.data[3].b == 128,
    "reloaded grey pixel");
```

Clean up the scratch file.

```
  delete(scratch);
```

18.0.7. Error Handling

`png()` returns null when the file does not exist or is not a decodable PNG. Always check the result with `!img` before you touch its fields.

```
  missing = file("no_such_image.png").png();
  assert(!missing, "png() of a missing file is null");
}
```

19. Lexer

The lexer library breaks a text into tokens so your program can understand its structure. Tokens are the smallest meaningful pieces: numbers, names (like variable and function names), operators, and quoted strings. You tell the lexer which multi-character sequences count as one token and which words are reserved keywords, and it handles the rest.

19.0.1. Setting Up the Lexer

Create a `lexer::Lexer` and register your language's rules before you parse anything. `set\_tokens` ensures operators like `+=` or `\>\>` are scanned as one token instead of two separate characters. `set\_keywords` prevents reserved words from being treated as ordinary names — the lexer will report them exactly as written so your parser can treat them specially.

```
use lexer;
fn main() {
  l = lexer::Lexer { };
  l.set_tokens(["+=", "*=", "-=", "<=", ">=", "!=", "==", ">>", "<<", "->", "=>",
">>>", "..", "..=", "&&", "||"]);
  l.set_keywords(["for", "in", "if", "else", "fn", "pub", "use", "struct",
"enum", "match", "and", "or"]);
```

19.0.2. Reading Tokens

`parse\_string(name, source)` feeds source text into the lexer. The name is used in error messages and position reports. After that, call the typed reader functions one by one to consume tokens in order.

`int()` consumes and returns the next integer token, or null if the current token is not an integer. `long\_int()` does the same for integers suffixed with `l`. `matches(s)` consumes the next token only when it equals s and returns true; otherwise it leaves the token in place and returns false. `peek()` returns the next token as text without consuming it. `position()` returns the current location as `file:line:col`.

```
  l.parse_string("Tokens", "12 += -2 * 3 >> 4");
  assert(l.int() == 12, "Integer");
  assert(!l.matches("+"), "Incorrect plus");
  assert(l.peek() != "+", "Incorrect plus");
  assert(l.matches("+="), "Incorrect plus_is");
  assert(l.int() == -2, "Second integer");
  assert(l.matches("*"), "Incorrect multiply");
  assert(l.int() == 3, "Third number");
  assert(l.position() == "Tokens:1:14", "Incorrect position {l.position()}");
  assert(!l.matches(">"), "Incorrect higher");
  assert(l.matches(">>"), "Incorrect logical shift");
  assert(l.position() == "Tokens:1:17", "Incorrect position {l.position()}");
```

19.0.3. String Literals and Comments

`constant\_text()` reads a double-quoted string and handles special codes like \n (newline) and \\ (backslash). `constant\_character()` reads a single-quoted character literal and returns it as text.

```
  l.parse_string("Texts", "\"123\" + '4'");
  assert(l.constant_text() == "123", "Incorrect text literal");
  assert(l.matches("+"), "Incorrect add");
```

`constant\_character()` returns a `character`, so compare it as one (`l.constant\_character() == '4'`), not against the text "123". The lexer collects `//` comments automatically as it scans. You do not need to handle them yourself. `last\_comment()` returns the accumulated comment text since the last consumed token. When multiple comment lines appear in a row they are joined with newlines into a single string. `comment\_behind()` is true when the comment appeared on the same line as the preceding token rather than on its own line above. `is\_finished()` returns true once every token has been consumed.

```
  l.parse_string("Comments", "// starting comments\n123 // same line comment\n//
extra comment\n4");
  assert(!l.comment_behind(), "Initial comment not behind");
  assert(l.last_comment() == "starting comments", "Initial comment");
  assert(l.int() == 123, "Content integer");
  assert(l.comment_behind(), "Second comment is behind");
  assert(l.last_comment() == "same line comment\nextra comment", "Second
comment");
  assert(!l.is_finished(), "Not Ready");
  assert(l.int() == 4, "Second integer");
  assert(l.last_comment() == "", "No remaining comment");
  assert(l.is_finished(), "Ready");
```

19.0.4. Embedded Format Expressions

Loft string literals can embed expressions with `{expr}`. The lexer exposes a protocol that lets you parse these yourself. When `constant\_text()` reaches a `{`, it returns the literal text before it and sets `is\_formatting()` to true. At that point call `set\_formatting(false)` and parse the embedded expression normally using the usual token readers. When the expression is done, call `set\_formatting(true)` and consume the closing `}}`. Then `constant\_text()` continues with the next segment of the string.

```
  l.parse_string("Formatting", "\"abc{{12 + 34}}def\"");
  assert(l.constant_text() == "abc", "Before formatting");
  assert(l.is_formatting(), "Formatting");
  l.set_formatting(false);
  assert(l.int() == 12, "First integer");
  assert(l.matches("+"), "Incorrect plus");
  assert(l.int() == 34, "Second integer");
  l.set_formatting(true);
  assert(l.matches("}}"), "Incorrect closing brace");
  assert(l.constant_text() == "def", "After formatting");
  assert(!l.is_formatting(), "Formatting");
}
```

20. Parser

The ‘parser’ library lets a Loft program read and validate other Loft source code at runtime. This is useful when you want to:

- validate configuration files written in the Loft syntax
- build tools that inspect or transform Loft source
- write a test that checks whether a generated code snippet is syntactically correct

Together the ‘lexer’ and ‘parser’ libraries are designed as a reusable foundation. You can use them to build entirely new languages: feed the lexer your own token rules, then layer a custom parser on top. This keeps the tokeniser and grammar logic separate from the language runtime so you only bring in what you need.

‘parse(name, source)’ is the single entry point. It takes a display name (used in error messages) and a Loft source string. The name does not need to correspond to a real file — it is only used when reporting parse errors.

If the source is invalid, parse() emits a diagnostic error and the call returns without producing a value.

20.0.1. What the parser understands

The parser handles all Loft syntax:

- ‘struct Name { field: type [= default] }’ — data containers with named fields
- ‘enum Name { Variant [{ field: type }] }’ — named choices, each with optional data
- ‘fn name(params) [-> type] { body }’ — functions with a block body
- ‘fn name(params) [-> type]; #rust “template”’ — operator templates backed by Rust
- ‘use module;’ — module imports
- Expressions: binary operators with precedence, function calls, field access,

```
index expressions, if/else, for loops, blocks, and formatted string literals
```

- Type expressions: plain names, generic types like ‘vector<T>’, keyed

```
collections (sorted/hash/index), and integer ranges with 'limit(min, max)'
```

```
use parser;
fn main() {
```

20.0.2. Parsing a minimal function

The double braces ‘{{‘ and ‘}}’ produce literal ‘{‘ and ‘}’ inside a Loft format string — needed here because the source code itself contains braces.

```
  assert(parser::parse("hello", "fn main() {{ println(\"Hello World\"); }}") ==
0,
         "a minimal function parses cleanly");
```

20.0.3. Parsing a struct and function together

The parser validates the entire source snippet as one unit. Both the struct definition and the function body are checked.

```
  assert(parser::parse("point", "struct Point {{ x: integer, y: integer }} fn
origin() -> Point {{ Point {{ x: 0, y: 0 }} }}") == 0,
         "a struct and a function in one unit parse cleanly");
```

20.0.4. Parsing an enum

Both a plain variant and a struct variant (with fields) are recognised.

```
  assert(parser::parse("shape", "enum Shape {{ Circle {{ radius: float }},
Rectangle {{ w: float, h: float }} }}") == 0,
         "both a plain and a struct variant parse cleanly");
```

20.0.5. Parsing several statements in a body

A body is a sequence of statements; each call below is one.

```
  assert(parser::parse("body", "fn m() {{ println(\"a\"); println(\"b\"); }}") ==
0,
         "a body of several statements parses cleanly");
```

20.0.6. Parsing assignments and operators

Every binary operator in the table above is accepted, including assignment.

```
  assert(parser::parse("assign", "fn f() {{ t = 0; t += 1; }}") == 0,
         "assignment and compound assignment parse cleanly");
```

20.0.7. Parsing a for loop and arithmetic

This exercises the expression parser, range syntax, and a tail expression after a loop body — the shape where a speculative struct-literal parse has to give its tokens back so the loop can claim them.

```
  assert(parser::parse("loop", "fn sum(n: integer) -> integer {{ t = 0; for i in
1..=n {{ t += i; }} t }}") == 0,
         "a for loop over a range parses cleanly");
```

20.0.8. Practical use: validating user-supplied code

If your application lets users write Loft snippets (for scripting or configuration), you can parse them before executing:

```
snippet = read_user_input();
if parser::parse("user_code", snippet) > 0 {
  // The snippet was invalid — the count is how many errors it had.
}
```

Combine this with the lexer (see the Lexer page) when you need to extract individual tokens from the source rather than validate all of the syntax.

20.0.9. What a failed parse looks like

The count is the whole point: a snippet that does not parse answers a NON-zero number, so a caller can tell the two apart. Watching the call not crash cannot.

```
  assert(parser::parse("broken", "fn main() {{ println(\"hi\");") > 0,
         "an unterminated body reports at least one error");
  println("parser test passed");
}
```

21. Libraries

A library is a `.loft` file you can share across projects. Place it in a `lib/` directory, import it with `use name;` at the top of your file, and then refer to its types and functions using the `name::` prefix. All types and functions in a library are accessible — there is no way to mark something as internal or hidden. `use` statements must come before any `fn` or `struct` definition in your file. Putting a `use` after a definition is a syntax error.

```
use testlib;
```

21.0.1. Constants and Enums

Library constants and enum variants are accessed with the `libname::` prefix. You can compare, order, and convert enum values exactly as you would with enums defined in the same file.

21.0.2. Struct Construction and Field Access

Construct a library struct with its full namespace prefix. Omitting the prefix is a parse error. Once you have a value, field access uses the plain dot notation with no prefix needed.

21.0.3. Calling Library Methods

Methods defined in the library are called on the value directly — no prefix on the call site. Free functions (not methods) do need the `libname::` prefix.

21.0.4. Extending a Library Type

You can add new methods to a library type from your own file. Write the `self` parameter type with the `::` separator and the compiler treats the function as a method on that type.

```
fn shifted(self: testlib::Point, dx: float, dy: float) -> testlib::Point {
  testlib::Point {x: self.x + dx, y: self.y + dy }
}
```

```
fn main() {
```

Constants from a library require the library prefix.

```
  assert(testlib::MAX_SIZE == 100, "library constant MAX_SIZE");
  assert(testlib::MIN_SIZE == 1, "library constant MIN_SIZE");
```

Library enum variants: comparison and ordering work as normal.

```
  s = testlib::Ok;
  assert(s == testlib::Ok, "enum equality");
  assert(s < testlib::Error, "enum ordering");
  assert("{s}" == "Ok", "enum to text: {s}");
  assert("Warning" as testlib::Status == testlib::Warning, "text to enum");
```

Library enum values can be passed to library free functions.

```
assert(testlib::status_text(testlib::Error) == "err", "free fn with enum arg");
```

Structs are constructed with namespace prefix and accessed by field.

```
p = testlib::Point {x: 3.0, y: 4.0 };
assert(p.x == 3.0, "field access: x={p.x}");
assert(p.y == 4.0, "field access: y={p.y}");
```

Methods defined in the library are called without a prefix.

```
dist = p.distance();
assert(round(dist) == 5.0, "method call: distance={dist}");
```

User-defined method on a library type (self: testlib::Point) works.

```
assert(p.shifted(1.0, 2.0).x == 4.0, "shifted x");
assert(p.shifted(0.0, 2.0).y == 6.0, "shifted y");
```

Scalar fields can be mutated directly.

```
b = testlib::Bag {label: "test", count: 0 };
b.label = "updated";
assert(b.label == "updated", "direct scalar field mutation");
```

Methods that mutate scalar fields work.

```
b.bump();
assert(b.count == 1, "method scalar mutation: count={b.count}");
b.bump();
assert(b.count == 2, "method scalar mutation: count={b.count}");
```

Free functions: called with the library prefix.

```
assert(testlib::add(3, 4) == 7, "free function add");
assert(testlib::add(0, -5) == -5, "free function negative");
```

Library types work as function parameter types when written with their full namespace prefix in the function signature. A struct with a vector field can be initialised via a struct literal, including elements that are themselves library structs.

```
bag2 = testlib::Bag {label: "full", count: 3, items: [
testlib::Point {x: 1.0, y: 0.0 },
testlib::Point {x: 0.0, y: 1.0 }
] };
assert(len(bag2.items) == 2, "initialized vector field: {len(bag2.items)}");
assert(bag2.items[0].x == 1.0, "vector field element access");
```

Appending a struct variable with `+= \[var\]` and a whole vector with `+= vec` both work.

```
  extra = testlib::Point {x: 7.0, y: 8.0 };
  bag2.items +=[extra];
  assert(len(bag2.items) == 3, "append via var: len={len(bag2.items)}");
  assert(bag2.items[2].x == 7.0, "append via var: x={bag2.items[2].x}");
  more_pts =[testlib::Point {x: 9.0, y: 10.0 }, testlib::Point {x: 11.0, y:
12.0 }];
  bag2.items += more_pts;
  assert(len(bag2.items) == 5, "append vector: len={len(bag2.items)}");
  assert(bag2.items[3].x == 9.0, "append vector[3].x={bag2.items[3].x}");
```

Importing the same library twice is silently ignored. A library can itself import other libraries using `use`, so dependency chains work.

21.0.5. Package Layout and loft.toml

A library can be distributed as a directory package instead of a single flat file. The packaged directory layout is:

```
mylib/
  loft.toml           optional manifest
  src/
    mylib.loft        library source (default entry)
```

When the interpreter searches a lib directory and finds `\<dir\>/mylib/` it looks for `\<dir\>/mylib/src/mylib.loft` automatically.

The optional `loft.toml` manifest supports two settings:

```
[package]
loft = ">=1.0"        minimum interpreter version required
```

```
[library]
entry = "src/mylib.loft"   override the default entry path
```

If the interpreter version is below the stated minimum, loading the library produces a fatal compile error describing the version mismatch. If no manifest is present, the default entry `src/\<name\>.loft` is used.

21.0.6. Wildcard and Selective Imports

By default, library names require the `libname::` prefix. You can import names directly into your namespace with `use lib::\*` (wildcard) or `use lib::Name, Other` (selective). Only `pub`-marked definitions in the library are imported this way. Non-pub definitions remain accessible via the `lib::name` prefix.

21.0.7. Limitations

`use` must appear before all definitions. If you write a function first and then a `use`, the compiler reports a syntax error.

`pub` controls import visibility. Only `pub`-marked definitions are available via wildcard (`use lib::\*`) or selective import. Non-pub definitions are still accessible with the `lib::name` prefix.

Native extensions. A library can be pure `.loft`, or ship a Rust crate beside it: declare `native = "..."` in `loft.toml` and loft builds and links it for you. See PACKAGES.md for the build + binding model.

```
}
```

22. Store Locks

Loft gives you two separate ways to protect a value from accidental changes:

1. **Compile-time const**: mark a variable or parameter with ‘const’ and the

```
  compiler refuses to compile any code that tries to reassign it. The check
  happens before the program even runs — zero runtime cost.
```

2. **Runtime lock**: the ‘#lock’ attribute on a reference lets you lock a

```
  store at runtime so that any write attempt panics immediately, even across
  function boundaries. This is useful for debugging: turn it on when you
  suspect an unexpected mutation.
```

```
struct Counter {
  value: integer
}
```

22.0.1. const parameters

‘const’ on a parameter is a compile-time promise: “this function will not modify this value.” The compiler enforces it — any assignment to a const parameter is a compile error, caught before you run anything.

```
fn read_value(self: const Counter) -> integer {
  self.value
}
```

A non-const parameter leaves the store unlocked so the function can write.

```
fn increment(self: Counter) {
  self.value += 1
}
```

```
fn main() {
```

22.0.2. const local variables

Declare a local variable with ‘const’ to signal that it will not change after its first assignment. The compiler rejects any later assignment to it — reassigning or appending to a const variable is a compile error.

This is handy for configuration values or lookup tables that should never be overwritten by accident deep inside a long function.

```
  const limit = 100;
  assert(limit == 100, "const integer is readable");
```

const also works for struct references.

```
  const cfg = Counter {value: 42 };
  assert(cfg.value == 42, "const reference is readable");
```

Passing a const reference to a const parameter is always allowed.

```
  assert(read_value(cfg) == 42, "const passed to const param");
```

22.0.3. Calling methods on const references

A non-const method can still be called on a non-const variable even after you have manually locked the store; the lock is a runtime check.

```
  c = Counter {value: 10 };
  increment(c);
  assert(c.value == 11, "increment modified c");
```

22.0.4. Runtime store locks with #lock

'#lock' is an attribute on any reference variable. Setting it to true turns on a runtime guard: any write to that store will panic immediately, wherever it happens. A freshly created reference starts unlocked.

```
  d = Counter {value: 5 };
  assert(!d#lock, "new store starts unlocked");
  d#lock = true;
  assert(d#lock, "store is locked after assignment");
```

You can still read from a locked store — only writes are blocked.

```
  assert(read_value(d) == 5, "locked store is still readable");
```

22.0.5. When to use each approach

- Use 'const' on parameters and locals to express your design intent

```
and get compile-time safety at zero cost.
```

- Use '#lock = true' when you want a runtime tripwire: you suspect some

```
code path is mutating a value it should not touch, and you want the
program to panic with a precise location rather than corrupt silently.
```

get_store_lock() is the function form of the #lock attribute. Both return the same boolean.

```
  e = Counter {value: 99 };
  assert(get_store_lock(e) == e#lock, "function form matches attribute before
lock");
  e#lock = true;
  assert(get_store_lock(e) == e#lock, "function form matches attribute after
```

```
lock");
}
```

23. Parallel execution

The `par(b=worker\_call, threads)` clause on a `for` loop runs a function on every element of a vector in parallel and gives you the results one by one in the loop body.

23.0.1. Why parallel loops?

When you have a large collection and each element can be processed independently — image filters, score calculations, data transforms — `par()` splits the work across CPU cores automatically. You write a normal for loop; adding `par(...)` makes it parallel with no other changes.

23.0.2. Syntax

```
`for a in vec par(b=func(a), N) { body }`
```

- `a` — loop variable, read-only reference to the current element
- `b` — result variable, holds the return value of `func(a)`
- `func(a)` — worker function called on each element
- `N` — number of threads (1 = sequential, 4 = typical)

23.0.3. Two call forms

- **Form 1** — global function: `par(b=my\_func(a), 4)`
- **Form 2** — method on element: `par(b=a.my\_method(), 4)`

23.0.4. Worker function rules

The worker function:

- Takes a `const` reference to the element (read-only, no mutation)
- Returns a value (integer, float, boolean, text, or a struct)
- Must not use global state or I/O (no println, no file access)
- Can accept extra arguments forwarded from the calling scope

Results are delivered in the original order regardless of which thread finishes first.

```
struct Score {
  value: integer
}
```

```
struct ScoreList {
  items: vector < Score >
}
```

```
struct DoubledScore {
  label: text,
  doubled: integer
}
```

```
fn double_score(thr_r: const Score) -> integer {
  thr_r.value * 2
}
```

```
fn scale_score(thr_r: const Score, thr_factor: integer) -> integer {
  thr_r.value * thr_factor
}
```

```
fn make_doubled(thr_r: const Score) -> DoubledScore {
  DoubledScore { label: "v{thr_r.value}", doubled: thr_r.value * 2 }
}
```

```
fn get_value(self: const Score) -> integer {
  self.value
}
```

```
fn make_scores() -> ScoreList {
  thr_q = ScoreList { };
  thr_q.items +=[Score {value: 10 }, Score {value: 20 }, Score {value: 30 }];
  thr_q
}
```

@FTR-070 — a `par` loop across threads: the worker runs on every element and the results come back in order, whatever the thread count

```
fn main() {
```

23.0.5. Global Function (Form 1)

Each Score's value is doubled by `double\_score` across 4 threads. The loop body sees `b` with the doubled value, in original order.

```
  q = make_scores();
  sum = 0;
  for thr_a in q.items par(thr_b = double_score(thr_a), 4) {
    sum += thr_b;
  }
```

10*2 + 20*2 + 30*2 = 120

```
  assert(sum == 120, "parallel double: sum == 120");
```

23.0.6. Extra Arguments

The worker function can take extra arguments from the calling scope. Here `scale\_score(a, factor)` passes `factor=3` to every worker.

```
q1b = make_scores();
factor = 3;
scaled_sum = 0;
for thr_a in q1b.items par(thr_b = scale_score(thr_a, factor), 4) {
  scaled_sum += thr_b;
}
```

10*3 + 20*3 + 30*3 = 180

```
assert(scaled_sum == 180, "extra arg: scaled sum == 180");
```

23.0.7. Struct Return

Workers can return a struct. Text fields are deep-copied so they remain valid after the worker thread exits.

```
q2 = make_scores();
labels = "";
for thr_sa in q2.items par(thr_ds = make_doubled(thr_sa), 1) {
  labels += "{thr_ds.label},";
}
assert(labels == "v10,v20,v30,", "struct return: {labels}");
```

23.0.8. Method Call (Form 2)

`b=a.get\_value()` dispatches the method on each element in parallel. This is syntactic sugar — equivalent to `b=get\_value(a)`.

```
q3 = make_scores();
total = 0;
for thr_a in q3.items par(thr_b = thr_a.get_value(), 4) {
  total += thr_b;
}
assert(total == 60, "method call: total == 60");
```

23.0.9. Empty Vector

An empty vector is safe — the loop body never executes, no threads are spawned.

```
empty = ScoreList { };
thr_n = 0;
for thr_a in empty.items par(thr_b = double_score(thr_a), 1) {
  thr_n += 1;
}
assert(thr_n == 0, "empty: 0 iterations");
```

23.0.10. Sequential fallback

Using `par(..., 1)` runs the worker on a single thread — useful for debugging. The behaviour is identical; only the parallelism changes.

```
  q4 = make_scores();
  seq_sum = 0;
  for thr_a in q4.items par(thr_b = double_score(thr_a), 1) {
    seq_sum += thr_b;
  }
  assert(seq_sum == 120, "sequential par(1): same result");
}
```

24. Logging

Printing everything to the console is fine during development, but in a real application you usually want more control: write to a file, filter by severity, and keep the output even after the terminal session ends. Loft's logging functions give you that without changing your source code.

24.0.1. The four severity levels

Choose the level that matches how serious the event is:

- 'log_info' — routine progress; fine to see during development but often

```
silenced in production to keep log files small.
Example: "processing file X", "connected to database".
```

- 'log_warn' — something unexpected happened but the program recovered.

```
Example: "config key missing, using default", "retrying after timeout".
```

- 'log_error' — something went wrong and you need to investigate, but the

```
program can continue (perhaps degraded).
Example: "failed to save record", "unexpected null value".
```

- 'log_fatal' — a condition so serious that normal operation is impossible.

```
Example: "cannot open database", "required config file not found".
```

24.0.2. Configuring the log destination

By default, log calls do nothing — no file is written, no console output is produced. To switch logging on, place a 'log.conf' file in the same directory as your '.loft' file, or pass '–log-conf path/to/log.conf' on the command line.

Generate a documented template with all defaults by running:

```
loft --generate-log-config
```

A minimal 'log.conf' looks like this:

```
[log]
file  = log.txt    # write messages here (relative to the .loft file)
level = info       # minimum level to record; choices: info warn error fatal
```

```
[rotation]
max_size_mb = 500  # start a new log file after this many megabytes
daily       = true # also start a new file at midnight UTC
max_files   = 10   # keep at most this many log files (older ones are deleted)
```

```
[rate_limit]
per_site = 5       # suppress messages from the same source line after 5/minute
```

```
[levels]
# Override the global level for a specific file:
# "debug_tool.loft" = info
# "src/"            = error
```

24.0.3. Production mode

When 'production = true' is set in log.conf:

- 'panic()' becomes a fatal log entry instead of aborting the process.
- A failing 'assert()' becomes an error log entry instead of aborting.

The program keeps running and the problem is captured in the log — useful for long-running services where a single error should not bring everything down.

24.0.4. Using format strings in log messages

Log messages are plain text, but you can embed any expression using the same '{...}' format syntax as everywhere else in Loft. The interpolation happens only when the message is actually going to be written; if the configured level is higher than the call, the string is never evaluated (no performance cost for suppressed messages).

```
fn main() {
```

Without a log.conf these calls do nothing — the tests below pass even though no output is produced.

```
  log_info("starting up");
  log_warn("this is a warning");
  log_error("something went wrong");
  log_fatal("critical failure");
```

24.0.5. Example log.txt output

When a 'log.conf' is present with 'level = info', the four calls above produce entries like this in 'log.txt':

```
2026-03-24 09:15:00 INFO   app.loft:3  starting up
2026-03-24 09:15:00 WARN   app.loft:4  this is a warning
2026-03-24 09:15:00 ERROR  app.loft:5  something went wrong
2026-03-24 09:15:00 FATAL  app.loft:6  critical failure
```

Each line contains: timestamp, severity level, source file and line number, then the message. This makes it easy to search the file for errors or trace back to the exact line that produced a message. A true assert never logs anything — it is only the false case that logs.

```
  assert(true, "this should never fail");
```

24.0.6. Typical usage pattern

In a real program you would write something like:

```
fn process(item: Item) {
    log_info("processing item {item.id}");
    result = do_work(item);
    if !result {
        log_error("work failed for item {item.id}");
        return;
    }
    log_info("finished item {item.id} successfully");
}
```

The log messages give you a record of exactly what the program did and where it went wrong, without cluttering the terminal during normal runs.

```
  println("logging test passed");
}
```

25. Random

Randomness lives in the `random` library, an installable package in the loft registry. Add it to a project once with

```
loft install random
```

and import it at the top of your file with `use random;`. The library gives you three functions, all backed by a fast PCG-64 generator:

```
rand(lo, hi)      a uniform random integer in [lo, hi] inclusive; null if lo > hi
rand_seed(seed)   seed the generator so a run is reproducible
rand_indices(n)   a vector holding 0, 1, ..., n-1 in a random order
```

The generator is thread-local and starts from a fixed default seed, so results are reproducible across runs unless you seed it with something varying (for example `rand\_seed(now() as integer)`).

```
use random;
```

```
fn main() {
```

25.0.1. Basic Random Integers

`rand(lo, hi)` returns a uniformly distributed integer between lo and hi, both included. Seed first with `rand\_seed` when you want a repeatable sequence.

```
  rand_seed(42);
  roll = rand(1, 6);       // a simulated die roll, 1..6
  assert(roll >= 1 && roll <= 6, "die roll out of range: {roll}");
```

25.0.2. Reproducibility

The same seed always produces the same sequence — essential for tests and for replaying a game from a saved seed.

```
  rand_seed(0);
  a = rand(0, 999);
  rand_seed(0);
  assert(rand(0, 999) == a, "the same seed must reproduce the sequence");
```

`rand` returns null when the range is empty (lo > hi), so you can detect a bad range with `!` instead of getting a crash.

```
  assert(!rand(10, 5), "an inverted range returns null");
```

25.0.3. Random Ordering: rand_indices

`rand\_indices(n)` returns a vector containing 0, 1, ..., n-1 in a random order. Use it to visit another collection in random order without copying the data.

```
rand_seed(7);
order = rand_indices(5);
assert(len(order) == 5, "rand_indices(5) has five entries");
```

Every value 0..4 appears exactly once — it is a permutation, not a sample with repeats. We verify that by marking each position as seen.

```
seen = [for _ in 0..5 { false }];
for idx in order {
  seen[idx] = true
}
all_seen = true;
for s in seen {
  if !s {
    all_seen = false
  }
}
assert(all_seen, "rand_indices must cover every position exactly once");
```

25.0.4. Sampling Without Replacement

Take the first k entries of a shuffled index vector to pick k distinct items.

```
items = ["apple", "banana", "cherry", "date", "elderberry"];
rand_seed(1);
indices = rand_indices(len(items));
```

Note on null-flow: indexing a vector by a *variable* (here `indices\[i\]`) yields a NULLABLE element — `text?` — because the compiler cannot prove the index is in range. Appending a `text?` to the non-null `picked` would change its type, so we supply a fallback with `?? ""`; the value is never actually null here.

```
picked = "";
for i in 0..3 {
  if i > 0 {
    picked += ", "
  }
  picked += items[indices[i]] ?? ""
}
```

`picked` now holds 3 distinct fruit names in a random order.

```
assert(len(picked) > 0, "should have picked some items: {picked}");
```

The same guard is needed for numbers — use `?? 0` when accumulating an integer that you read through a variable index.

```
  values = [10, 20, 30, 40, 50];
  total = 0;
  for i in order {
    total += values[i] ?? 0
  }
  assert(total == 150, "every value summed once: {total}");
}
```

26. Time

Loft provides two time functions:

```
now()     — milliseconds since the Unix epoch (wall-clock time).
ticks()   — microseconds elapsed since program start (monotonic clock).
```

Both return an 'integer' (loft's 64-bit integer).

Use 'now()' for timestamps, log entries, and date calculations. Use 'ticks()' for benchmarks and frame timing — it is unaffected by system clock changes or NTP adjustments.

```
fn main() {
```

26.0.1. Wall-clock time: now()

'now()' returns the current time as milliseconds since 1970-01-01T00:00:00 UTC. The value is always positive and grows over time.

```
  t = now();
  assert(t > 0, "now() must be positive");
```

Two successive calls return non-decreasing values.

```
  t2 = now();
  assert(t2 >= t, "now() must be non-decreasing");
```

26.0.2. Elapsed time: ticks()

'ticks()' measures microseconds since the program started. It uses a monotonic clock so it never jumps backward.

```
  start = ticks();
  assert(start >= 0, "ticks() must be non-negative");
  end = ticks();
  assert(end >= start, "ticks() must be monotonically non-decreasing");
```

26.0.3. Measuring elapsed time

Subtract two 'ticks()' values to get the duration in microseconds. Divide by 1000 to convert to milliseconds.

```
elapsed_us  = end_ticks - start_ticks
elapsed_ms  = elapsed_us / 1000
```

Example: measure how long a loop takes.

```
  t_before = ticks();
  sum = 0;
  for i in 0..1000 {
```

```
    sum += i
  }
  t_after = ticks();
  elapsed = t_after - t_before;
  assert(elapsed >= 0, "elapsed time must be non-negative: {elapsed}");
  assert(sum == 499500, "loop produced wrong sum: {sum}");
```

26.0.4. Common patterns

Timestamp a log entry (seconds since epoch):

```
seconds = now() / 1000
```

Seed the random number generator with the current time (now() already returns an integer, so no cast is needed):

```
rand_seed(now())
```

Simple stopwatch:

```
start = ticks()
... do work ...
log_info("Done in {(ticks() - start) / 1000} ms")
```

```
}
```

27. Safety

Loft catches many errors at compile time, but a few surprises remain at runtime. This page catalogues every known trap so you can write confident code from day one. Each section includes a live example that proves the described behavior. Helper used in the '??' double-evaluation example below. Increments the call counter and returns the given value unchanged.

```
fn counted_call(calls: &integer, value: integer) -> integer {
  calls += 1;
  value
}
```

Used to obtain a null text value (an uninitialized field is the NUL sentinel).

```
struct NullTextHolder {
  s: text,
}
```

```
struct OptTextHolder {
  s: text?,
}
```

```
fn main() {
```

27.0.1. Null values — hidden reserved values

Every type reserves one special value to mean "nothing here" (null). That reserved value looks like any other value, so be aware of what it is for each type:

- `boolean` — `false` is the null value
- `integer` — the most negative 64-bit integer (-9 223 372 036 854 775 808) is null
- `float`/`single` — `NaN` (Not a Number) is null
- `character` — the NUL character (code point 0) is null
- `text` — a single NUL character (`'\\0'`) is null; the empty string `""` is NOT null
- `reference` — record 0 is null
- plain `enum` — byte value 255 is null (limits enums to 255 variants)

This means there is one value per type that you cannot distinguish from null. For integers, that is –9 223 372 036 854 775 808. When defended with `?? null` (or a following null-check), division by zero produces this same value, so both paths look the same to your code:

```
  zero = 0;
  n = 1 / zero ?? null;
  assert(!n, "div-by-zero defended with `?? null` is null");
  assert(n != 0, "null is not zero; it is -9 223 372 036 854 775 808");
  assert(n < 0, "i64::MIN is the most negative 64-bit integer");
```

Arithmetic on null propagates: null plus anything is null.

```
  assert(! (n + 1), "null + 1 is still null");
```

Text null is the NUL character (‘\0’), not the empty string. A `text?` local or field can hold it; a plain `text` cannot, and a parse respects that.

```
  holder = NullTextHolder.parse(`{{}}`);
  assert(holder.s == "", "a missing `text` field is empty, because it cannot be
null");
  opt = OptTextHolder.parse(`{{}}`);
  assert(!opt.s, "a missing `text?` field IS null — the declaration decides");
  empty = "";
  assert(empty, "empty string is NOT null — this surprises most newcomers");
```

27.0.2. Parsing text to a number needs a fallback

An unchecked `text as integer` is a COMPILE error: parsing can fail, and loft refuses to let that failure slip through as a silent null. Ask for a checked cast with `integer?` (yields the number or null), or supply a default with `?? \<value\>`. The same rule applies to `float`.

```
  parsed = "42" as integer?;
  assert(parsed == 42, "checked cast `as integer?` yields the number");
  fallback = "oops" as integer? ?? -1;
  assert(fallback == -1, "unparseable text falls back to the default");
```

27.0.3. Integer overflow yields null

A calculation that overflows the integer range is uncomputable, so loft yields null and keeps running — it never wraps to a silently-wrong number (the way C, or a Rust release build, would) and it never crashes. One bad value degrades locally; the rest of the program still computes.

```
huge = 9223372036854775807; huge + 1  →  null (not a wrapped negative)
```

To trace where overflows arise, opt into the debug log level. Mitigation: check the result for null, or keep intermediate values within range.

```
  huge = 9223372036854775807;
  assert(! (huge + 1), "overflow yields null, not a wrapped value");
```

27.0.4. Bitwise operators with zero

All bitwise operators (AND, OR, XOR, shift) work correctly with zero. Zero is the identity element for OR, XOR, and shift; zero for AND.

```
  assert(0b1010&0 == 0, "AND with 0: zero");
  assert(0b1010|0 == 0b1010, "OR with 0: identity");
  assert(0b1010^0 == 0b1010, "XOR with 0: identity");
  assert(5<<0 == 5, "shift left by 0: identity");
  assert(5>>0 == 5, "shift right by 0: identity");
```

27.0.5. Float null compares uniformly with every other scalar

Floats store null as `NaN` internally, but null COMPARISON is uniform with every other scalar type (@PLN102 null model): `null == null` is TRUE, null is not equal to any real value, and null orders as the low extreme. Detect a null float with `f == null` (or `!f` / `f ?? default`) — NOT the old `f != f` trick, which no longer works now that `null == null` is true. Both the defended and the undefended division yield null and continue (C80 / E-Uncomp — see tests/scripts/184-i333-div-zero-null-continues.loft); the undefended site additionally reports a warning.

```
bad = 0.0 / 0.0 ?? null;
assert(!bad, "a null float is falsy");
assert(bad == null, "detect a null float with `== null`");
assert(bad == bad, "null == null is true (uniform null model)");
assert(! (bad != null), "a null float `!= null` is false");
assert(bad != 0.0, "null != 0.0 is true");
```

Use `f == null`, `!f`, or `f ?? default` to check for null floats.

27.0.6. Text length counts characters; size counts bytes

`len()` on text returns the number of characters (Unicode code points) — the human count. `size()` returns the number of UTF-8 bytes; multi-byte characters (accented letters, emoji, CJK) each occupy 2-4 bytes. The two values differ whenever the text contains any multi-byte character.

```
emoji = "Hi 😊!";
assert(len(emoji) == 5, "5 characters (H, i, space, 😊, !)");
assert(size(emoji) == 8, "8 UTF-8 bytes (the emoji is 4): {size(emoji)}");
```

Slicing and indexing also use byte offsets. Slicing in the middle of a multi-byte character is an error. Mitigation: Use `for c in text` to iterate by character. Use `c\#index` and `c\#next` to get the byte boundaries of each character.

```
count = 0;
for c in emoji { count += 1; }
assert(count == 5, "for-loop iterates by character, not byte");
```

27.0.7. `\#index` means different things on text and vectors

On a text loop, `c\#index` is the byte offset of the current character. On a vector loop, `v\#index` is the 0-based element position. Both are called `\#index` but represent different units.

```
v = [10, 20, 30];
idx = 0;
for x in v { idx = x#index; }
assert(idx == 2, "vector #index: element position (0-based)");
```

Text #index is a byte offset, not a character count:

```
t = "aé";
byte_pos = 0;
```

```
  for c in t { byte_pos = c#index; }
  assert(byte_pos == 1, "text #index: byte offset of 'é' (byte 1, not char 1)");
```

27.0.8. ?? evaluates the left side exactly once

The `??` operator means "use this value, or if it is null, use the right side instead". The left-hand expression is evaluated exactly once regardless of whether the result is null. The example below uses `counted\_call` (defined above) to verify this.

```
  calls = 0;
  qq_result = counted_call(calls, 7) ?? 99;
  assert(calls == 1, "?? evaluated left side exactly once: {calls}");
  assert(qq_result == 7, "result is the value from that single evaluation:
{qq_result}");
```

This means `result = expensive\_call() ?? default` is safe: the function is called once. If it returns null, `default` is used. There is no double call.

27.0.9. Text indexing and slicing return different types

`txt\[i\]` returns a `character` (a single Unicode scalar value). `txt\[i..j\]` returns `text` (a UTF-8 string). These are different types.

```
  txt = "hello";
  ch = txt[0];
  slice = txt[0..1];
  assert(ch == 'h', "indexing returns a character");
  assert(slice == "h", "slicing returns text");
```

Building text from characters requires format interpolation:

```
  result = "";
  for c in "abc" { result += "{c}"; }
  assert(result == "abc", "characters must be formatted into text");
```

27.0.10. Format strings: braces are always interpreted

Every string literal in loft is a format string. Literal braces must be escaped as `{{` and `}}`:

```
  n = 42;
  assert("{n}" == "42", "single braces: format expression");
  assert(len("{{}}") == 2, "double braces produce literal brace chars");
```

Forgetting to escape braces in expected output is a common mistake in assertions and comparisons.

27.0.11. Hash collections cannot be iterated

Hashes are lookup structures, not ordered collections. You cannot write `for item in my\_hash { }`. If you need both fast lookup and ordered iteration, keep a vector and a hash pointing at the same record type. See the Hash documentation page for the recommended pattern.

27.0.12. Mutation guard blocks appending during iteration

The compiler prevents `v += \[x\]` inside `for e in v`. This protects against infinite loops. The guard also catches field access: `for e in db.items { db.items += \[x\]; }` is blocked too. The only allowed mutation is `e\#remove` inside a filtered loop.

27.0.13. If-expression requires else when used as a value

Using `if` as a value expression without an `else` clause is a compile error. This prevents accidental null values from missing branches. If-statements (where the body has no value) do not need else. For example, `x = if cond { 1 }` is an error; write `x = if cond { 1 } else { 0 }`.

27.0.14. Match guards do not prove a variant is handled

A guarded arm like `Red if cond =\> ...` does not count as handling the `Red` variant because the guard can fail at runtime. Even if every variant has a guarded arm, you still need a wildcard `\_` or an unguarded arm so the compiler knows every case is covered.

27.0.15. Ref-parameter semantics

Without &, appending to a vector parameter is local — the caller's vector does not grow. With &, the caller sees the new elements. Field-level mutations (e.g. `v\[i\].field = x`) are always visible to the caller because both sides share the same underlying database reference. Rule of thumb: Use `&vector\<T\>` when the function needs to grow the vector. Use plain `vector\<T\>` when the function only reads or modifies existing elements.

27.0.16. Text file reading assumes UTF-8

`lines()` and `content()` read a file as UTF-8 text and crash on invalid UTF-8 (a binary file, or a different encoding such as Latin-1). For binary data, set a binary format on the file — `f\#format = LittleEndian` (or `BigEndian`) — and read raw bytes / integers directly (see 13-file.loft). So: read UTF-8 text in text mode, everything else in a binary format.

27.0.17. XOR is ^, not exponentiation

Unlike some languages where ^ means "power", in loft ^ is bitwise XOR. For exponentiation use the `\*\*` operator (`2 \*\* 10 == 1024`, `2.0 \*\* 3.0 == 8.0`) or the `pow()` function. Watch one precedence footgun: `-x \*\* y` parses as `-(x \*\* y)`, so loft warns on it — write `(-x) \*\* y` when you mean to raise a negated base.

```
  assert((0b1010^0b1100) == 0b0110, "^ is XOR");
  assert(2**10 == 1024, "** is the power operator");
  assert(pow(2.0, 3.0) == 8.0, "pow() also computes powers");
}
```

28. JSON

Loft has built-in JSON support: any struct can be serialised to JSON with a format flag, and parsed back from JSON text. No annotations or code generation needed — it works on every struct automatically.

28.0.1. Serialisation — struct to JSON

Use the ':j' format flag inside a format string to produce JSON output. Field names become quoted keys; strings are escaped; numbers and booleans are written as JSON literals.

```
"{my_struct:j}"
```

```
struct User {
  id: integer,
  name: text,
  email: text
}
```

```
struct MaybeUser {
  id: integer,
  name: text?,
}
```

28.0.2. Parsing — JSON to struct

Call 'Type.parse(text)' to create a struct from JSON text. Text arguments are auto-wrapped through 'json_parse' internally.

A field the JSON does not mention — and a field written 'null' — gets the DECLARED type's absent value. For a plain field that is its zero, because a plain field cannot hold null; write the field 'integer?' / 'float?' if you need to tell "absent" from "zero" apart:

```
struct Reading { id: integer, drift: float? }
r = Reading.parse("{}")     // r.id == 0, r.drift == null
```

Caveat (Q1): the auto-wrap form DROPS diagnostics — malformed input and schema mismatches leave the struct at its defaults with 'json_errors()' empty. For error reporting, stage explicitly: 'User.parse(json_parse(text))' — that form pushes both parse and schema errors to 'json_errors()'.

```
user = User.parse(json_text)
```

to inspect 'json_errors()' between the two steps.

28.0.3. Vectors

Parse a JSON array into a vector of structs with 'vector<T>.parse(text)'.

```
scores = vector<Score>.parse("[{\"v\":1},{\"v\":2}]")
```

```
struct Score {
  value: integer
}
```

28.0.4. Parse Errors

Call 'json_errors()' to see what went wrong with a malformed input. Schema-level mismatches (e.g. a field declared integer but receiving a JSON string) currently land as the loft null sentinel in the struct; Q1 schema-side diagnostics will add path-qualified reports in a follow-up.

```
data = MyType.parse(bad_json);
if len(json_errors()) > 0 { log_warn(json_errors()); }
```

28.0.5. Nested Structs

Structs with struct-typed fields parse nested JSON objects automatically.

```
struct Address {
  city: text,
  zip: text
}
```

```
struct Contact {
  name: text,
  address: Address
}
```

```
fn main() {
  u = User { id: 42, name: "Alice", email: "alice@example.com" };
  json = "{u:j}";
  expected = `{{"id":42,"name":"Alice","email":"alice@example.com"}}`;
  assert(json == expected, "to json");
```

```
  bob = User.parse(`{{"id":7,"name":"Bob","email":"bob@test.org"}}`);
  assert(bob.id == 7, "parsed id: {bob.id}");
  assert(bob.name == "Bob", "parsed name");
  assert(bob.email == "bob@test.org", "parsed email");
```

```
  u2 = User.parse("{u:j}");
  assert(u2.id == u.id, "round-trip id");
  assert(u2.name == u.name, "round-trip name");
```

Type-mismatched fields (id: string, name: number) parse as JSON fine, but the struct unwrap abandons the record at its defaults; path-qualified diagnostics on the mismatch are collected in `json\_errors`. `id` is declared plain, so its default is 0 — the mismatch is reported through `json\_errors`, never by putting a null in a slot the declared type says cannot hold one. Here we verify the unwrap does not crash on mismatched shapes.

```
  bad = User.parse(`{{"id":"not_a_number","name":42}}`);
  assert(bad.id == 0, "type-mismatched id keeps its default: {bad.id}");
```

```
  scores = vector<Score>.parse(`[{{"value":10}},{{"value":20}},{{"value":30}}]`);
  total = 0;
  for s in scores {
    total += s.value;
  }
  assert(total == 60, "vector sum: {total}");
```

```
  c = Contact.parse(`{{"name":"Carol","address":
{{"city":"Amsterdam","zip":"1012"}}}}`);
  assert(c.name == "Carol", "nested name");
  assert(c.address.city == "Amsterdam", "nested city: {c.address.city}");
  assert(c.address.zip == "1012", "nested zip");
```

28.0.6. A missing field gets its own declared absence

When a field is absent from the JSON, it gets the absent value its DECLARATION allows — not the type's null sentinel. A plain field cannot hold null, so it takes its empty value: 0 for a number, "" for text, false for a boolean. Declare the field 'T?' when "the document did not say" has to be tellable from "the document said zero", and check it with '!' or '??' before use.

```
  partial = User.parse(`{{"id":1}}`);
  assert(partial.id == 1, "partial id");
  assert(partial.name == "", "a missing `text` field is empty:
[{partial.name}]");
  maybe = MaybeUser.parse(`{{"id":1}}`);
  assert(!maybe.name, "a missing `text?` field is null");
}
```

29. Generics

A generic function uses a type variable to work with any type. Write the function once; the compiler creates a specialised copy for each concrete type you call it with.

29.0.1. Declaring a generic function

Place a single type variable in angle brackets after the function name. The type variable must appear in the first parameter (directly or as a container element like vector<T>).

```
fn identity<T>(x: T) -> T { x }
```

29.0.2. Calling a generic function

No special syntax at the call site — the compiler infers T from the first argument's type and creates a specialised copy automatically.

```
fn test_identity() {
  assert(identity(42) == 42, "identity integer");
  assert(identity(3.14) == 3.14, "identity float");
  assert(identity("hello") == "hello", "identity text");
  assert(identity(true) == true, "identity bool");
}
```

29.0.3. Multiple parameters of the same type

Additional parameters can also use T. They all share the same concrete type.

```
fn pick_second<T>(gen_a: T, gen_b: T) -> T {
  _x = gen_a;
  gen_b
}
fn test_pick_second() {
  assert(pick_second(1, 99) == 99, "pick second int");
  assert(pick_second("a", "b") == "b", "pick second text");
}
```

29.0.4. Generic functions on vectors

The most common use of generics: write a function that works on any vector. T is inferred from the vector's element type.

```
fn first_element<T>(gen_v: vector<T>) -> T {
  gen_v[0]
}
```

@PLN25 index flip — a computed index `gen\_v.len() - 1` is not provably in-bounds, so the read is nullable; a generic has no `?? default`, so the honest result type is `T?`.

```
fn last_element<T>(gen_v: vector<T>) -> T? {
  gen_v[gen_v.len() - 1]
```

```
}
fn test_vector_generics() {
  ints = [10, 20, 30];
  assert(first_element(ints) == 10, "first int");
  assert(last_element(ints) == 30, "last int");
  words = ["alpha", "beta", "gamma"];
  assert(first_element(words) == "alpha", "first text");
  assert(last_element(words) == "gamma", "last text");
}
```

29.0.5. Bounded generics with interfaces

By default, you can only assign and return T — no arithmetic, no comparison. To use operators on T, add an interface bound: `\<T: Ordered\>` means T must support `\<`, `\<=`, `\>`, `\>=` comparisons.

Built-in interfaces:

```
Ordered    — comparison operators (<, <=, >, >=)
Equatable  — equality operators (==, !=)
Addable    — addition and subtraction (+, -)
Numeric    — all four scalar operators (+, -, *, /)
Scalable   — multiplication by a float factor (* float)
Printable  — text conversion (to_text)
```

```
fn gen_max<T: Ordered>(gen_x: T, gen_y: T) -> T {
  if gen_x > gen_y { gen_x } else { gen_y }
}
fn gen_min<T: Ordered>(gen_x: T, gen_y: T) -> T {
  if gen_x < gen_y { gen_x } else { gen_y }
}
fn test_bounded() {
  assert(gen_max(3, 7) == 7, "max int");
  assert(gen_max(2.5, 1.5) == 2.5, "max float");
  assert(gen_min(3, 7) == 3, "min int");
  assert(gen_min("apple", "banana") == "apple", "min text");
}
```

29.0.6. Combining bounds

Use + to require multiple interfaces: `\<T: Ordered + Addable\>`.

```
fn clamped_add<T: Ordered + Addable>(gen_a: T, gen_b: T, gen_hi: T) -> T {
  result = gen_a + gen_b;
  if result > gen_hi { gen_hi } else { result }
}
fn test_combined_bounds() {
  assert(clamped_add(3, 4, 10) == 7, "under limit");
  assert(clamped_add(8, 5, 10) == 10, "clamped to limit");
  assert(clamped_add(1.0, 2.0, 2.5) == 2.5, "float clamped");
}
```

29.0.7. Allowed operations on T

Inside a generic function you may only use operations that the bound guarantees. Without any bound: assign, return, store in variables. With Ordered: comparisons. With Addable: + and -.

29.0.8. Disallowed operations

The compiler rejects operations not covered by the bound. `x + y` on unbounded T gives:

```
"generic type T: operator '+' requires a concrete type"
```

`x.field` on T gives:

```
"generic type T: field access requires a concrete type"
```

```
fn main() {
  test_identity();
  test_pick_second();
  test_vector_generics();
  test_bounded();
  test_combined_bounds();
}
```

30. Closures

A closure is a lambda that reads variables from the function scope in which it is written. No explicit capture list is needed — the compiler detects which outer variables the lambda body uses and packages them automatically.

30.0.1. Integer capture

A lambda may read any integer variable in scope. Store the lambda, then call it by name.

30.0.2. Multiple integer captures

A lambda can read more than one outer variable at once. All are captured at the moment the lambda is written.

30.0.3. Text capture

Text values are captured by deep-copy, so they are independent of the original variable after capture.

30.0.4. Capture timing

Loft captures variables at the moment the lambda is written (definition time), not when it is called. If a variable changes after the lambda is written, the lambda still sees the original value.

30.0.5. Cross-scope closures

A function can return a capturing lambda to the caller. The captured values travel with the lambda — no dangling references.

30.0.6. Closures with higher-order functions

A capturing closure stored in a variable can be called directly.

30.0.7. Non-capturing lambdas with higher-order functions

Lambdas that use only their own parameters (no capture) also work fine.

```
fn make_adder(base_val: integer) -> fn(integer) -> integer {
  fn(n: integer) -> integer { base_val + n }
}
```

```
fn main() {
```

30.0.8. Integer capture

```
  offset = 100;
  shift = fn(x: integer) -> integer { x + offset };
  assert(shift(5) == 105, "shift(5): {shift(5)}");
  assert(shift(-3) == 97, "shift(-3): {shift(-3)}");
```

30.0.9. Multiple integer captures

```
  lo = 10;
  hi = 20;
  clamp_val = fn(x: integer) -> integer {
    if x < lo { lo } else if x > hi { hi } else { x }
```

```
};
assert(clamp_val(5) == 10, "clamp below: {clamp_val(5)}");
assert(clamp_val(15) == 15, "clamp mid: {clamp_val(15)}");
assert(clamp_val(25) == 20, "clamp above: {clamp_val(25)}");
```

30.0.10. Text capture

```
greeting = "Hello";
make_msg = fn(name: text) -> text { "{greeting}, {name}!" };
assert(make_msg("World") == "Hello, World!", "greeting capture");
assert(make_msg("Loft") == "Hello, Loft!", "greeting capture 2");
```

30.0.11. Capture timing

Closures capture at definition time. 'base' is 10 when the lambda is written; reassigning 'base' to 20 afterwards does not affect the captured value.

```
base = 10;
add_base = fn(n: integer) -> integer { base + n };
base = 20;
assert(add_base(5) == 15, "sees base=10 at definition time: {add_base(5)}");
```

30.0.12. Cross-scope closures

make_adder returns a lambda that captured base_val from its parameter.

```
add10 = make_adder(10);
add100 = make_adder(100);
assert(add10(5) == 15, "add10(5): {add10(5)}");
assert(add100(5) == 105, "add100(5): {add100(5)}");
```

30.0.13. Closures with higher-order functions

```
factor = 3;
scale = fn(x: integer) -> integer { x * factor };
assert(scale(5) == 15, "scale(5): {scale(5)}");
assert(scale(10) == 30, "scale(10): {scale(10)}");
```

30.0.14. Non-capturing lambdas with higher-order functions

No capture needed here — the lambda uses only its own parameter.

```
nums = [1, 2, 3, 4, 5];
doubled = map(nums, |x| { x * 2 });
assert(doubled[0] == 2, "doubled[0]: {doubled[0]}");
assert(doubled[4] == 10, "doubled[4]: {doubled[4]}");
```

```
evens = filter(nums, |x| { x % 2 == 0 });
assert(evens[0] == 2, "evens[0]: {evens[0]}");
```

```
  assert(evens[1] == 4, "evens[1]: {evens[1]}");
}
```

31. Coroutines

#warn Variable d is never read @TITLE: Lazy sequences with generators and yield A generator function produces values one at a time. Declare the return type as ‘iterator<T>’ and use ‘yield’ to emit each value. The function body is suspended between yields and resumed when the next value is requested. When the body finishes the generator is exhausted. Generators are useful when you want to produce values on demand, stop early with ‘break’, or chain sequences without building intermediate collections.

31.0.1. Declaring a generator function

A function whose return type is ‘iterator<T>’ is a generator. ‘yield value’ emits one item and pauses; the next call resumes from there.

31.0.2. Iterating with a for loop

A ‘for’ loop drives the generator forward automatically. The body runs once per yielded value. Use ‘break’ to stop early without consuming the rest.

31.0.3. Manual advance with next() and exhausted()

Call ‘next(gen)’ yourself when you need fine-grained control. ‘exhausted(gen)’ returns true once there are no more values.

31.0.4. Generators with parameters

A generator function may take any parameters, including text. Parameter values are preserved across yields so the generator can use them throughout its lifetime.

31.0.5. Delegation with yield from

‘yield from sub_gen()’ forwards all values from another generator inline, as if each of its yields appeared at this point directly.

31.0.6. Early termination with break

Stopping a ‘for’ loop before the generator is exhausted is safe. The abandoned generator frame is freed automatically.

```
fn count_up(start: integer, stop: integer) -> iterator<integer> {
  for i in start..stop {
    yield i;
  }
}
```

```
fn squares(n: integer) -> iterator<integer> {
  for i in 1..=n {
    yield i * i;
  }
}
```

Yields the length of ‘s’ twice — shows that a text parameter survives yields.

```
fn len_twice(s: text) -> iterator<integer> {
  yield len(s);
  yield len(s);
}
```

```
fn inner_vals() -> iterator<integer> {
  yield 10;
  yield 20;
}
```

```
fn outer_combined() -> iterator<integer> {
  yield 1;
  yield from inner_vals();
  yield 2;
}
```

```
fn large_range(limit: integer) -> iterator<integer> {
  for i in 1..=limit {
    yield i;
  }
}
```

```
fn main() {
```

31.0.7. Iterating with a for loop

```
  total = 0;
  for n in count_up(1, 6) {
    total += n;
  }
  assert(total == 15, "sum 1..5: {total}");
```

Squares via for loop with index tracking.

```
  idx = 0;
  for s in squares(4) {
    idx += 1;
    if idx == 1 { assert(s == 1, "sq1: {s}"); }
    if idx == 2 { assert(s == 4, "sq2: {s}"); }
    if idx == 3 { assert(s == 9, "sq3: {s}"); }
    if idx == 4 { assert(s == 16, "sq4: {s}"); }
  }
  assert(idx == 4, "squares count: {idx}");
```

31.0.8. Manual advance with next() and exhausted()

```
  gen = count_up(10, 13);
  a = next(gen);
  b = next(gen);
  c = next(gen);
  assert(a == 10 && b == 11 && c == 12, "manual next: {a},{b},{c}");
```

One extra advance past the last element; exhausted() is true after that.

```
  d = next(gen);
  done = exhausted(gen);
  assert(done, "exhausted after last value");
```

31.0.9. Generators with parameters

'len_twice' takes a text parameter and yields its length twice. The text value is preserved across the yield.

```
  lt_total = 0;
  for n in len_twice("hello") {
    lt_total += n;
  }
  assert(lt_total == 10, "len_twice: {lt_total}");
```

31.0.10. Delegation with yield from

```
  yf_total = 0;
  for n in outer_combined() {
    yf_total += n;
  }
  assert(yf_total == 33, "yield from: 1+10+20+2={yf_total}");
```

31.0.11. Early termination with break

Stop after the first 5 values from a 1000-element generator.

```
  limit_total = 0;
  for n in large_range(1000) {
    if n > 5 { break; }
    limit_total += n;
  }
  assert(limit_total == 15, "early break sum 1..5: {limit_total}");
}
```

32. Tuples

A tuple groups a fixed number of values of possibly different types. You do not need to give a tuple a name — use it directly where you need to pass or return multiple values together. Access individual elements with '.0', '.1', etc.

32.0.1. Creating tuples

Write the elements in parentheses, separated by commas. The compiler infers the element types from what you put in. You can add an explicit type annotation when needed.

32.0.2. Accessing elements

Use '.0', '.1', '.2', … to read individual elements.

32.0.3. Modifying elements

Tuple elements can be reassigned like ordinary variables.

32.0.4. Tuples as function parameters

Pass a tuple to a function by value (a copy) or by reference. A value parameter gets its own copy — changes inside the function do not affect the caller. A reference parameter ('&') gives the function a direct link to the caller's tuple so it can modify individual elements in place.

32.0.5. Returning tuples

A function can return a tuple to give back more than one value at once.

32.0.6. Destructuring

Assign a tuple to multiple names in one step using the '(a, b) = expr' form. This is concise when a function returns a tuple and you need both values.

32.0.7. Three or more elements

Tuples can have any number of elements.

```
fn min_max(lo: integer, hi: integer) -> (integer, integer) {
  if lo <= hi {
    (lo, hi)
  } else {
    (hi, lo)
  }
}
```

```
fn swap_ints(pair: &(integer, integer)) {
  tmp = pair.0;
  pair.0 = pair.1;
  pair.1 = tmp;
}
```

```
fn main() {
```

32.0.8. Creating tuples

```
t = (10, 20);
assert(t.0 == 10, "t.0: {t.0}");
assert(t.1 == 20, "t.1: {t.1}");
```

Type annotation

```
w: (integer, integer) = (3, 4);
assert(w.0 + w.1 == 7, "annotated sum: {w.0+w.1}");
```

32.0.9. Modifying elements

```
v = (1, 2);
v.0 = 10;
assert(v.0 + v.1 == 12, "after assign: {v.0}+{v.1}");
```

32.0.10. Tuples as value parameters

```
p = (10, 20);
sum = p.0 + p.1;
assert(sum == 30, "sum: {sum}");
```

32.0.11. Tuples as reference parameters (swap in place)

```
pair = (3, 7);
swap_ints(pair);
assert(pair.0 == 7 && pair.1 == 3, "after swap: ({pair.0},{pair.1})");
```

32.0.12. Returning a tuple from a function

```
bounds = min_max(5, 2);
assert(bounds.0 == 2, "min: {bounds.0}");
assert(bounds.1 == 5, "max: {bounds.1}");
```

Already in order

```
bounds2 = min_max(1, 9);
assert(bounds2.0 == 1, "min2: {bounds2.0}");
assert(bounds2.1 == 9, "max2: {bounds2.1}");
```

32.0.13. Destructuring

```
(lo, hi) = min_max(8, 3);
assert(lo == 3, "destructured lo: {lo}");
assert(hi == 8, "destructured hi: {hi}");
```

32.0.14. Three or more elements

```
  u = (1, 2, 3);
  assert(u.0 + u.1 + u.2 == 6, "triple sum: {u.0}+{u.1}+{u.2}");
```

```
  triple = (10, 20, 30);
  triple.1 = 99;
  assert(triple.0 == 10 && triple.1 == 99 && triple.2 == 30,
    "triple modified: ({triple.0},{triple.1},{triple.2})");
}
```

33. Match

The `match` expression lets you compare a value against a series of patterns and run different code for each case. It is similar to `switch` in other languages but more powerful: patterns can destructure structs, bind fields to local variables, and include `if` guards. Match is an expression — it returns a value, so you can assign the result or use it directly inside a larger expression.

33.0.1. Simple enum matching

The most common use: branch on an enum variant. Every variant must be covered or a `\_` wildcard must appear — the compiler checks exhaustiveness.

```
enum Direction { North, South, East, West }
```

A struct-enum: its variants carry named fields, which a match arm can destructure.

```
enum Shape {
  Circle { radius: integer },
  Rect { w: integer, h: integer },
}
```

```
fn direction_name(d: Direction) -> text {
  match d {
    North => "north",
    South => "south",
    East  => "east",
    West  => "west",
  }
}
```

```
fn main() {
  assert(direction_name(North) == "north", "simple enum match");
  assert(direction_name(West) == "west", "west");
```

33.0.2. Match as expression

Because match returns a value you can use it in assignments and arguments.

```
  score = match Direction.East {
    North | South => 10,
    East | West   => 20,
  };
  assert(score == 20, "match expression: {score}");
```

33.0.3. Struct-enum destructuring

When an enum has variants with fields (a struct-enum) you can bind the fields to variables in the match arm.

```
Circle { radius } — binds the 'radius' field to a local variable
Rect { w, h }     — binds both 'w' and 'h'
```

The variable names must match the field names exactly.

```
shape = Rect { w: 3, h: 4 };
shape_area = match shape {
  Circle { radius } => radius * radius * 3,
  Rect { w, h }     => w * h,
};
assert(shape_area == 12, "struct-enum destructure: {shape_area}");
```

33.0.4. Guards

Add `if condition` after a pattern to restrict when the arm matches. The guard can reference variables bound by the pattern.

```
v = 42;
label = match v {
  0      => "zero",
  _ if v < 0 => "negative",
  _ if v > 100 => "large",
  _      => "normal",
};
assert(label == "normal", "guard: {label}");
```

33.0.5. Wildcard _

The underscore _ matches anything. Put it last as a catch-all. Without it, the compiler will reject the match if any value could fall through without matching.

```
x = 7;
kind = match x {
  1 => "one",
  2 => "two",
  _ => "other",
};
assert(kind == "other", "wildcard: {kind}");
```

33.0.6. Integer matching

Match works on integers — each arm is compared with ==.

```
name = match 3 {
  1 => "one",
  2 => "two",
  3 => "three",
  _ => "many",
};
assert(name == "three", "integer match: {name}");
```

33.0.7. Range patterns

A numeric arm can match a whole inclusive range with `low..=high` — handy for grading, bucketing, or any "is it in this band" test.

```
grade = match 95 {
  90..=100 => "A",
  80..=89  => "B",
  _        => "lower",
};
assert(grade == "A", "range pattern: {grade}");
```

33.0.8. Null patterns

When the subject is nullable, a `null` arm handles the absent case explicitly. With null-flow this is the idiomatic way to branch on a `T?` inside a match.

```
maybe: integer? = null;
presence = match maybe {
  null => "absent",
  0    => "zero",
  _    => "present",
};
assert(presence == "absent", "null pattern: {presence}");
```

33.0.9. Text matching

Match also works on text values.

```
greeting = "hi";
reply = match greeting {
  "hello" => "formal",
  "hi"    => "casual",
  _       => "unknown",
};
assert(reply == "casual", "text match: {reply}");
```

33.0.10. Multiple patterns with |

Combine patterns with `|` to share the same arm body.

```
d = Direction.South;
axis = match d {
  North | South => "vertical",
  East | West   => "horizontal",
};
assert(axis == "vertical", "multi-pattern: {axis}");
```

33.0.11. Nested match

Match can appear inside other expressions, including other match arms.

```
  n = 15;
  result = match n % 2 == 0 {
    true  => "even",
    false => match n % 3 == 0 {
      true  => "odd multiple of 3",
      false => "odd",
    },
  };
  assert(result == "odd multiple of 3", "nested: {result}");
}
```

34. Formatting

Loft strings can embed expressions inside `{...}` braces. A colon after the expression introduces a format specifier that controls width, alignment, precision, number base, and output style.

34.0.1. Basic interpolation

Any expression inside `{...}` is evaluated and converted to text.

```
struct Point { x: integer, y: integer }
```

A type that receives the PARTS of a format string. See the last section.

```
struct Query {
  const parts: vector<text>,
  const values: vector<text>,
}
```

`lit` gets the bytes you wrote in the source file.

```
fn lit(self: Query, s: text) { self.parts += [s]; }
```

`hole\_text` gets an interpolated value. It is never added to `parts`.

```
fn hole_text(self: Query, v: text?) { self.values += [v ?? ""]; }
```

```
fn hole_int(self: Query, v: integer) { self.values += ["{v}"]; }
```

Show the two lists separately, so you can see what went where.

```
fn shape(self: Query) -> text {
  out = "";
  for q_part in self.parts { out += "<{q_part}>" }
  for q_val in self.values { out += "[{q_val}]" }
  return out;
}
```

```
fn main() {
  name = "world";
  assert("hello {name}" == "hello world", "basic interpolation");
  assert("1 + 2 = {1 + 2}" == "1 + 2 = 3", "expression in braces");
```

34.0.2. Width and alignment

A number after `:` sets the minimum width. The value is padded with spaces to fill the width.

```
{val:6}   — right-aligned (default for numbers)
{val:>6}  — right-aligned (explicit)
```

```
{val:<6}  — left-aligned
{val:^6}  — centered
```

For text, the default alignment is left. For numbers, right.

```
assert("{42:6}" == "    42", "default number align is right");
assert("{42:>6}" == "    42", "explicit right-align");
assert("{42:<6}" == "42    ", "left-align number");
assert("{42:^6}" == "  42  ", "center-align number");
s = "hi";
assert("{s:6}" == "hi    ", "default text align is left");
assert("{s:>6}" == "    hi", "right-align text");
assert("{s:^6}" == "  hi  ", "center-align text");
```

34.0.3. Zero padding

Prefix the width with `0` to pad with zeros instead of spaces.

```
assert("{7:03}" == "007", "zero-padded 3 digits");
assert("{42:05}" == "00042", "zero-padded 5 digits");
assert("{-1:04}" == "-001", "zero-padded negative: sign before zeros");
```

34.0.4. Signed format

`+` forces a sign on positive numbers.

```
assert("{42:+}" == "+42", "explicit positive sign");
assert("{-42:+}" == "-42", "negative sign always shown");
assert("{0:+}" == "+0", "sign on zero");
```

34.0.5. Hexadecimal, binary, octal

`:x` for lowercase hex, `:\#x` for hex with `0x` prefix. `:b` for binary, `:o` for octal.

```
assert("{255:x}" == "ff", "hex lowercase");
assert("{255:#x}" == "0xff", "hex with prefix");
assert("{10:b}" == "1010", "binary");
assert("{8:o}" == "10", "octal");
```

34.0.6. Float precision

`.N` after the colon limits decimal places.

```
assert("{3.125:.1}" == "3.1", "1 decimal place");
assert("{3.125:.2}" == "3.12", "2 decimal places");
assert("{3.125:.0}" == "3", "0 decimal places");
assert("{0.0:.3}" == "0.000", "trailing zeros");
```

34.0.7. Width + precision

Combine width and precision: `{val:W.P}` where W is total width and P is decimal places.

```
  assert("{3.125:8.2}" == "    3.12", "width 8 precision 2");
  assert("{-3.125:8.2}" == "   -3.12", "negative width+precision");
```

34.0.8. JSON format

`:j` serialises a struct or value as JSON. `:j` serialises a struct as JSON. Use it on struct values, not primitives. See the JSON documentation page for full details.

34.0.9. Vector format

Vectors are formatted as `\[a,b,c\]` by default. A format specifier inside a `for` loop applies to each element.

```
  v = [1, 2, 3];
  assert("{v}" == "[1,2,3]", "default vector format");
  assert("{for fmt_n in 1..4 {fmt_n * 10}:04}" == "[0010,0020,0030]", "formatted
vector elements");
```

34.0.10. Large integers and single-precision floats

`integer` is a 64-bit type, so even large values format like any other number; single-precision floats use the same specifiers too.

```
  n = 1000000;
  assert("{n}" == "1000000", "integer default");
  assert("{n:>10}" == "   1000000", "integer right-aligned");
  f = 1.5f;
  assert("{f}" == "1.5", "single default");
```

34.0.11. Pretty-printing a struct

The `:\#` specifier expands a struct across a spaced, readable layout instead of the compact default form.

```
  p = Point { x: 1, y: 2 };
  assert("{p}" == "{{x:1,y:2}}", "default struct format is compact");
  assert("{p:#}" == "{{ x: 1, y: 2 }}", "`:#` pretty-prints with spaces");
```

34.0.12. Character format

```
  c = 'A';
  assert("{c}" == "A", "character default");
```

34.0.13. Boolean format

```
  assert("{true}" == "true", "boolean true");
  assert("{false}" == "false", "boolean false");
```

34.0.14. Building a value instead of text

Everything above joins the pieces into one `text`. When the type you assign to says so, the format string **builds that type instead**, and the type is told which bytes you wrote and which came from a value.

A type opts in by defining `lit` (for your bytes) and one `hole\_…` method per value kind it accepts: `hole\_text`, `hole\_int`, `hole\_float`, `hole\_single`, `hole\_boolean`, `hole\_character`. There is no new syntax — the type you assign to decides — and plain `text` is unchanged.

```
  who = "ada";
  q: Query = "SELECT * FROM t WHERE name = {who}";
```

The text you wrote is in `parts`; the value is in `values`, never in the text. That separation is the point: a value cannot turn into syntax, because the only way into the text is `lit`, and `lit` only ever receives bytes from your source file.

```
  assert(q.shape() == "<SELECT * FROM t WHERE name = >[ada]",
         "the literal and the value stay apart");
```

The value can be anything at all, including text that would otherwise change the meaning of the statement. It is still just a value.

```
  danger = "'; DROP TABLE t; --";
  q2: Query = "WHERE name = {danger}";
  assert(q2.shape() == "<WHERE name = >['; DROP TABLE t; --]",
         "a value that looks like syntax is still a value");
```

Each kind goes to its own method, so the type sees the real type.

```
  q3: Query = "id={7} name={who}";
  assert(q3.shape() == "<id=>< name=>[7][ada]", "an integer hole calls
hole_int");
```

The database clients use this, so a query needs no placeholders to count:

```
  q: SqlText = "SELECT id FROM users WHERE name = {name}";
  db.db_rows(q)
```

```
}
```

35. Reference parameter forwarding

@ARGS: –lib tests/lib When a function receives a `&Struct` (mutable reference) parameter and passes it to another function that also takes `&Struct`, the reference must be forwarded without dereferencing. This is P144.

```
use p144_entry;
```

```
fn main() {
```

box_set_val calls box_ensure(b) internally — forwarding the & param.

```
  b = Box { items: [] };
  box_set_val(b, 42);
  assert(b.items[0].val == 42, "P144: & forwarded — item set to 42");
```

Calling the inner function directly also works.

```
  b2 = Box { items: [] };
  box_ensure(b2);
  assert(len(b2.items) == 1, "ensure added one item");
  assert(b2.items[0].val == 0, "default val is 0");
```

Multiple forwards: ensure is idempotent.

```
  box_ensure(b2);
  assert(len(b2.items) == 1, "ensure idempotent");
```

Set a second value via box_set_val on a pre-populated box.

```
  b3 = Box { items: [] };
  b3.items += [Inner { val: 10 }];
  b3.items += [Inner { val: 0 }];
  box_set_val(b3, 99);
  assert(b3.items[0].val == 10, "existing item preserved");
  assert(b3.items[1].val == 99, "zero item replaced");
}
```

36. Time library

@ARGS: –lib tests/fixtures/libs

The `time` library represents a point in time as a plain `integer` (i64) holding milliseconds since the Unix epoch — the same unit `now()` returns. All calendar math is UTC and uses the proleptic Gregorian calendar, so results are identical on the interpreter, `--native`, and WebAssembly. Import it with `use time;` and call its free functions with the `time::` prefix.

Timezone model: field accessors and formatters are UTC. For local-day bucketing pass a fixed UTC offset in minutes (`to\_local` / `local\_day` / `today`). There is no timezone database and no daylight-saving handling — a fixed offset only.

```
use time;
```

36.0.1. Parsing and field access

`parse` reads an ISO date "YYYY-MM-DD", optionally followed by a time. The accessors return UTC calendar fields; `weekday` is 0=Mon..6=Sun.

```
fn show_fields() {
  d = time::parse("2026-05-25");
  assert(time::year(d) == 2026, "year");
  assert(time::month(d) == 5, "month");
  assert(time::day(d) == 25, "day");
  assert(time::weekday(d) == 0, "2026-05-25 is a Monday");
  assert(time::weekday_name(d) == "Mon", "weekday name");
  assert(time::month_name(d) == "May", "month name");
```

```
  t = time::parse("2026-05-25 14:30:07");
  assert(time::hour(t) == 14, "hour");
  assert(time::minute(t) == 30, "minute");
  assert(time::second(t) == 7, "second");
```

The "T…Z" ISO form parses to the same instant as the space form.

```
  assert(time::parse("2026-05-25T14:30:07Z") == t, "ISO T/Z form");
```

Malformed input parses to null — test with `!`.

```
  bad = time::parse("not-a-date");
  assert(!bad, "garbage is null");
}
```

36.0.2. Formatting

```
fn show_formatting() {
  t = time::parse("2026-05-25 14:30:07");
  assert(time::format_date(t) == "2026-05-25", "format_date");
```

```
  assert(time::format_time(t) == "14:30", "format_time");
  assert(time::format_seconds(t) == "14:30:07", "format_seconds");
  assert(time::format_datetime(t) == "2026-05-25 14:30", "format_datetime");
  assert(time::format_iso(t) == "2026-05-25T14:30:07Z", "format_iso");
}
```

36.0.3. Arithmetic and differences

Stepping never desynchronises across leap years — the math is exact.

```
fn show_arithmetic() {
  jan1 = time::parse("2026-01-01");
  assert(time::format_date(time::add_weeks(jan1, 1)) == "2026-01-08",
"add_weeks");
  assert(time::format_date(time::add_days(jan1, 365)) == "2027-01-01",
"add_days");
```

Leap-day boundary.

```
  feb28 = time::parse("2024-02-28");
  assert(time::format_date(time::add_days(feb28, 1)) == "2024-02-29", "leap
day");
```

```
  dec31 = time::parse("2026-12-31");
  assert(time::days_between(jan1, dec31) == 364, "days_between");
```

```
  hourly = time::seconds_between(time::parse("2026-01-01 00:00:00"),
                                 time::parse("2026-01-01 01:00:00"));
  assert(hourly == 3600, "seconds_between");
}
```

36.0.4. Week boundaries and ISO week numbering

```
fn show_weeks() {
```

start_of_week snaps back to Monday 00:00.

```
  wed = time::parse("2026-05-27");
  assert(time::format_date(time::start_of_week(wed)) == "2026-05-25", "Monday of
week");
```

2026-01-01 is a Thursday -> ISO week 1 of 2026.

```
  assert(time::iso_year(time::parse("2026-01-01")) == 2026, "iso_year");
  assert(time::iso_week(time::parse("2026-01-01")) == 1, "iso_week");
```

2021-01-01 is a Friday -> still ISO week 53 of 2020.

```
  jan1_2021 = time::parse("2021-01-01");
  assert(time::iso_year(jan1_2021) == 2020, "iso_year rollback");
  assert(time::iso_week(jan1_2021) == 53, "iso_week 53");
}
```

36.0.5. Local day at a fixed offset

23:30 UTC at +120 minutes is 01:30 the next local day.

```
fn show_local() {
  t = time::parse("2026-05-25 23:30:00");
  assert(time::format_datetime(time::to_local(t, 120)) == "2026-05-26 01:30",
"to_local");
  assert(time::format_date(time::local_day(t, 120)) == "2026-05-26",
"local_day");
}
```

```
fn main() {
  show_fields();
  show_formatting();
  show_arithmetic();
  show_weeks();
  show_local();
  println("time library: all checks passed");
}
```

37. Feature catalogue

GENERATED by tools/features/gen.loft — DO NOT EDIT. @NAME: Feature catalogue @TITLE: Every language and tooling feature, with what each one is for.

Everything loft can do, in one list. Each entry has a page of its own with what it is, how it aids you, and a runnable example — see `doc/features/` in the repository, where `\@F2` is `F2.md`.

The catalogue is generated from the `loft-lang/features` issue tracker, which is the single source of truth: every entry is an issue, so nothing here can drift from what was decided without the drift guard failing.

37.0.1. Language features

- **@F1** — Nullable value semantics — in-band sentinels + null-fallback arithmetic
- **@F2** — `??` null-coalescing operator (incl. `?? return`)
- **@F3** — Primitive scalar types (integer, float, single, boolean, character)
- **@F4** — Ranged/width integer types (u8/i8/u16/i16/i32/u32)
- **@F5** — Type conversions — implicit, format-only, and explicit `as`
- **@F6** — `vector\<T\>` — append, index, slice, comprehensions, aggregates, map/filter/reduce
- **@F7** — `hash\<T\[keys\]\>` keyed collection
- **@F8** — `sorted\<T\[keys\]\>` collection
- **@F9** — `index\<T\[keys\]\>` B-tree index (asc/desc, multi-key)
- **@F10** — `iterator\<T\>` values
- **@F11** — Tuples — anonymous fixed-arity `(T1, T2, …)`
- **@F12** — Struct records — fields, `= default`, `computed`, `limit`/`not null`/`assert`
- **@F13** — Simple enums (ordered value types)
- **@F14** — Polymorphic struct-enums (per-variant fields)
- **@F15** — Enum-scoped variant names + context inference
- **@F16** — Functions & declarations (`pub`, parameters, return)
- **@F17** — Named arguments + default parameter values
- **@F18** — `const` parameters
- **@F19** — Method dispatch via `self` / `both`
- **@F20** — Variant-based dynamic dispatch
- **@F21** — References `&T` — parameters + write-back bindings
- **@F22** — Closures & lambdas (value capture, cross-scope)
- **@F23** — Function references as first-class values
- **@F24** — Higher-order functions (map / filter / reduce)
- **@F25** — Generics — single type variable `\<T\>`, inferred
- **@F26** — Interfaces & bounded generics (`\<T: A + B\>`, operator interfaces)
- **@F27** — `if` / `else` as an expression
- **@F28** — `for`-in loops — ranges, loop attributes, filtered, `rev()`
- **@F29** — Pattern matching — enum/scalar/tuple, guards, or-patterns, exhaustiveness
- **@F30** — `is` variant check (+ field capture)
- **@F31** — `break` / `continue` + labelled forms
- **@F32** — Custom iterators via `fn next(self) -\> T?`
- **@F33** — `par(...)` parallel for-loop

- **@F34** — Coroutines / generators — `yield`, `yield from`
- **@F35** — String literals — `{expr}` interpolation + backtick multiline
- **@F36** — String formatting / format specifiers (+ for-expressions)
- **@F37** — Operator set — arithmetic/comparison/logical/bitwise/unary, `\*\*`
- **@F38** — Arithmetic safety — overflow/÷0 ⟶ null, nullable peers
- **@F39** — Math & trigonometry library
- **@F40** — File & directory I/O (+ durable-store binding)
- **@F41** — Environment & arguments (env vars, arguments(), program dirs, path resolution)
- **@F42** — JSON — json_parse, JsonValue, Type.parse, to_json
- **@F43** — Random numbers (rand_seed / rand / rand_indices)
- **@F44** — Logging & diagnostics API (log_*, print, assert, panic)
- **@F45** — `sizeof()`
- **@F46** — Type aliases (`type X = …`)
- **@F47** — Library imports / module system (`use` forms, `pub`)
- **@F48** — The `loft` CLI — run a program, `--interpret` / `--native`, `--timeout`, `--help`
- **@F49** — REPL — interactive sessions
- **@F50** — Introspection — `loft introspect` (bytecode / native Rust / slot tables)
- **@F51** — Debugger — breakpoints, frame capture, scripted RPC
- **@F52** — Source formatter — canonical `.loft` output
- **@F53** — Native-binary backend (`--native` ⟶ rustc)
- **@F54** — Browser / WASM target (`--html` / `--native-wasm`)
- **@F55** — Package management (`loft install`, `loft.toml`, lockfile)
- **@F56** — Live code reload — patch a running program
- **@F88** — Stack traces (stack_trace)
- **@F89** — Test runner (fn test_*, loft –tests / loft test)
- **@F92** — Direct C binding — `\#c "symbol" "signature"`, no rustc and no glue crate
- **@F93** — Paged & remote store loading — read part of a dataset over HTTP range
- **@F94** — Type-directed interpolation — a type receives the literal/hole boundary
- **@F95** — Value structs (`value struct`) — copy semantics, stored inline
- **@F96** — `x?` default-fallback operator — discharge a nullable to its type default
- **@F97** — `len` vs `size` — logical count vs bytes occupied
- **@F98** — `spatial\<T\[x,y\]\>` — position-keyed collection for proximity queries
- **@F99** — Sequence match patterns — alternation, optionals, repetition, capture
- **@F100** — Dead-code lint — a local never read, and a write whose value is lost
- **@F101** — The build phase — `loft build`, declared targets, zero-config defaults
- **@F102** — Android target — `--native-android`, APK packaging, GL surface and touch input
- **@F103** — `deliver` / `expose` — hand a loft value to JavaScript with no copy
- **@F104** — `store\_reclaim()` — give a store's unused file space back
- **@F105** — Custom `to\_text` format hook — a type owns how it prints
- **@F106** — Copy and move semantics — when two names share data, and when they do not
- **@F107** — Type reflection — the declared shape of a type, as data
- **@F108** — Lazy store binding — a collection fetches on a miss
- **@F109** — `\#superseded` — steer callers to a newer form without breaking the old one

- **@F110** — Diagnostic suggestions — what to write instead, checked by running it
- **@F111** — `for f in s\#fields` — a compile-time loop over a struct's scalar fields
- **@F112** — `store\_release()` — say a record is finished, and stop holding it in memory
- **@F113** — Associated types — an interface names a companion type
- **@F114** — `x\[i\]` on a library type — `OpIndex` dispatch
- **@F115** — `OpDrop` — a type runs code when its scope lets it die

37.0.2. Tooling and infrastructure

- **@I57** — Lexer
- **@I58** — Parser (two-pass recursive descent)
- **@I59** — Type resolver
- **@I60** — Scope & dependency/lifetime tracker (deps)
- **@I61** — Stack slot allocator
- **@I62** — IR data model (Value/Type/Data)
- **@I63** — Store-resident IR (reader + handle + materializer)
- **@I64** — Bytecode compiler (IR -> bytecode)
- **@I65** — Bytecode code generator
- **@I66** — Bytecode VM / executor
- **@I67** — Opcode implementations
- **@I68** — Native Rust generator
- **@I69** — Word-addressed store
- **@I70** — Database — alloc / persistence / journal / snapshot / schema
- **@I71** — DbRef pointers & collection keys
- **@I72** — Parallel execution runtime
- **@I73** — Native function registry
- **@I74** — CDylib extension loader
- **@I75** — Diagnostics collector
- **@I76** — Logger runtime
- **@I77** — Registry / manifest / lockfile resolution
- **@I78** — Live-reload dispatch
- **@I79** — Documentation generator (maintainer tooling)
- **@I80** — Tracker indexer (maintainer tooling)
- **@I81** — Feature-catalogue sync generator (issues ⟶ in-project mirror + example tests)
- **@I82** — Sandbox policy & enforcement
- **@I83** — Dev server (–serve / –rpc)
- **@I84** — Coroutine runtime / yield codec
- **@I85** — Engine-host kernel natives
- **@I86** — Startup cache & embedded stdlib
- **@I87** — Auto-use / lazy library triggers
- **@I90** — Shared utilities & data structures
- **@I91** — Editor tooling: language server (LSP), debug adapter (DAP), resolution index
- **@I116** — Library placement — in-process, a worker, or another machine
- **@I117** — Git query natives — a repository as a typed library, not a subprocess

```
fn main() {
  println("the feature catalogue is generated from the issue tracker");
}
```

38. Running your program

The pages before this one show how to WRITE loft. This one shows how to RUN it. You need one command to start, and everything else is optional.

38.0.1. Run a file

Put your program in a file ending in '.loft' and pass it to loft:

```
$ loft hello.loft
hello, world!
```

That is the whole thing. There is no build step to run first and no project to set up. A single file is a complete program.

38.0.2. Give your program some arguments

Put them after the file name, and declare one `vector\<text\>` parameter on 'main' to read them:

```
fn main(args: vector<text>) {
  for who in args { println("hello, {who}!"); }
}
```

```
$ loft greet.loft Ada Grace
hello, Ada!
hello, Grace!
```

That one parameter is the only shape 'main' accepts. Any other — a plain 'text', two of them — is refused, because nothing would fill it.

38.0.3. Try something without making a file

Type 'loft' on its own and you get a prompt where you can type one line at a time and see the answer straight away:

```
$ printf '2 + 3\n' | loft repl --fresh
5
```

Typed by hand it looks like this — start it, then type at the prompt:

```
loft
loft> 2 + 3
5
```

This is called a REPL, which is short for "read, evaluate, print, loop". It is the quickest way to check what a function does or try an idea. Your session is remembered, so you can close it and come back later — '–fresh' is how you ask for a clean one instead.

38.0.4. Two ways to run, and why you usually ignore this

loft can run your program in two ways. It compiles your program to a fast program first when it can, and otherwise it reads and runs your program directly. The second way is called interpreting.

You do not have to choose. loft picks for you, and the ANSWER is the same either way — that is a rule the language holds itself to, not a hope. You can ask for one on purpose when you want to:

```
$ loft --interpret hello.loft
hello, world!
```

A downloaded loft always interprets. That is normal and not something to fix.

38.0.5. Stop a program that runs too long

A loop with a mistake in it can run forever. Give loft a time limit and it stops the program for you:

```
$ loft --timeout 10 hello.loft
hello, world!
```

This is worth using whenever you run something for the first time.

38.0.6. Check your work without running it

Sometimes you only want to know whether the program is correct so far:

```
$ loft check hello.loft
ok
```

This reports mistakes and runs nothing. It is fast, and it is a good habit while you are still writing.

38.0.7. When loft tells you something is wrong

A message from loft names the file, the line, and what to do. Some messages can also show you the exact replacement to write:

```
$ loft --explain --interpret hello.loft
hello, world!
```

That prints the suggested fix under each message. It only shows you the fix; it changes nothing, so it is always safe to run.

```
fn main() {
```

A program is just a file with a `main` function in it, like this one. Everything above is how you would run this file.

```
  greeting = "hello";
  who = "world";
  message = "{greeting}, {who}!";
  assert(message == "hello, world!", "string interpolation builds the message");
  println(message);
}
```

39. Testing your code

A test is a function that checks your code still does what you meant. You do not install anything to write one, and there is no special syntax.

39.0.1. Write a test

Give a function a name starting with ‘test’ and put an ‘assert’ in it. That is all a test is:

```
fn double(n: integer) -> integer { n * 2 }
```

```
fn test_double_doubles() {
  assert(double(4) == 8, "double(4) should be 8");
}
```

‘assert’ takes something that should be true, and a message. If it is true, nothing happens. If it is not, the message is what you will read, so write the message for the person who has to fix it — usually you, later.

39.0.2. Run your tests

Point loft at the file:

```
$ loft --tests calc.loft
test result: ok. 2 passed; 1 file
```

and it finds every ‘test’ function and runs it, naming the file and the functions it found.

Give it a directory instead of a file and it looks in every ‘.loft’ file underneath. Give it nothing and it starts from where you are:

```
$ loft --tests calc.loft::test_double_doubles
test result: ok. 1 passed; 1 file
```

39.0.3. Run just one test

While you are fixing one thing, run only that one. Put ‘::’ and the test’s name after the file:

```
$ loft --tests calc.loft::test_double_of_zero
test result: ok. 1 passed; 1 file
```

39.0.4. When a test fails

A failure names the test and shows your message:

```
FAIL  calc.loft::test_that_fails  —  assertion failed: arithmetic still works
test result: FAILED. 1 failed; 0 passed; 1 total; 1 file
```

The message is the part you wrote, which is why a vague message like “it works” costs you time and a specific one saves it.

39.0.5. Read the line after the result

The end of the report says what the run did NOT do, in square brackets — for example that it ran on the interpreter and did not try the compiled build. That is deliberate. A run that says only "ok" cannot tell you whether the other half was checked or never ran at all.

To also run your tests the compiled way:

```
$ loft --tests --native calc.loft
test result: ok. 2 passed; 1 file
```

39.0.6. Once you have a project

When your code grows into a project with a 'loft.toml' file (the next page), tests live in a 'tests/' folder and you run them with:

```
$ cd greeter && loft test
test result: ok. 1 passed; 1 file
```

```
fn double(n: integer) -> integer { n * 2 }
```

This page is itself run as a test, so the example below really does pass.

```
fn main() {
  assert(double(4) == 8, "double(4) should be 8");
  assert(double(0) == 0, "double(0) should be 0");
```

An assert that holds produces no output at all — silence is success.

```
  println("both checks passed");
}
```

40. Debugging

When a program does the wrong thing, the usual reaction is to add printing to it, run it, read the output, and take the printing out again. loft gives you a better tool: stop the program on the line you care about and look around.

The program used below is ‘count.loft’. It adds 1, 2 and 3 into a total.

40.0.1. Stop on a line

Say ‘debug’, then your file, then a colon and a line number:

```
$ printf ':continue\n' | loft debug count.loft:4
paused in main | total = 0, i = 1
```

The program runs until it reaches line 4 and then waits for you. The line it prints tells you where you are and what every local variable holds right now.

Line 4 is inside a loop that goes round three times, so the program stops there three times. ‘:continue’ means “carry on until you reach this line again”, not “run to the end” — which is why the examples below say ‘:continue’ more than once when they want the program to finish.

40.0.2. Look at a value

Type the name of a variable and press enter. Any expression works, not just a name — it is worked out where the program is paused:

```
$ printf 'total\n:continue\n:continue\n:continue\n' | loft debug count.loft:4
0
```

This is the part that replaces printing. You do not have to guess in advance which values you will want; ask for them when you are there.

40.0.3. Move one step at a time

Four commands move the program forward:

- ‘:step’ runs the next line, going INTO any function it calls
- ‘:next’ runs the next line, going OVER any function it calls
- ‘:finish’ runs until the current function returns
- ‘:continue’ carries on to the next time this line is reached

```
  $ printf ':step\ntotal\n:continue\n:continue\n:continue\n' | loft debug
count.loft:4
  1
```

After that single ‘:step’ the total is 1, because the loop has now added its first number. Each stop prints the same “paused” line, so you can watch a value change as the loop goes round.

40.0.4. Change a value while it is running

You can write to a local, not only read it. This answers “would it work if this were right?” without editing the file and starting again:

```
  $ printf 'total = 100\n:continue\n:continue\n:continue\n' | loft debug
count.loft:4
  total=106
  run finished
```

The program carries on with the value you gave it. It finished with 106 instead of 6, because the loop still had 2 and 3 to add after the change.

40.0.5. Getting out

‘:continue’ carries on. ‘:quit’ stops right away. ‘:help’ lists every command if you forget one.

40.0.6. Typing, or feeding it a script

Every example here pipes its commands in, which is how they are checked automatically — the output you see above is compared against what the command really prints. Run ‘loft debug count.loft:4’ on its own and you get the ‘(dbg)’ prompt to type at instead. The commands are the same either way.

```
fn main() {
```

This is what `count.loft` does — the program the examples above debug.

```
  total = 0;
  for i in 1..4 {
    total += i;
  }
  assert(total == 6, "1 + 2 + 3 is 6, which is what the debugger shows you");
  println("total={total}");
}
```

41. Projects and libraries

One file is a complete loft program, and for a long time that is all you need. When you want to share code between programs, or keep tests beside it, you turn the folder into a package.

41.0.1. What a package looks like

A package is a folder with a ‘loft.toml’ file in it and two sub-folders:

```
greeter/
  loft.toml          what this package is called
  src/greeter.loft   the code
  tests/greet.loft   the tests
```

‘src’ is where your code lives and ‘tests’ is where your tests live. The names matter — loft looks in exactly those places.

41.0.2. The manifest

‘loft.toml’ names the package and says which file is its front door:

```
[package]
name    = "greeter"
version = "0.1.0"
```

```
[library]
entry = "src/greeter.loft"
```

The ‘entry’ file is the one that other code sees. Anything you want to share from it is marked ‘pub’:

```
pub fn greet(who: text) -> text { "hello, {who}!" }
```

Without ‘pub’ a function is private to the package, which is the default.

41.0.3. Testing a package

Inside the package folder, ‘loft test’ finds and runs everything in ‘tests’:

```
$ cd greeter && loft test
test result: ok. 1 passed; 1 file
```

The tests reach your code by importing the package by name:

```
use greeter::*;
fn test_greet_names_the_person() {
  assert(greet("Ada") == "hello, Ada!", "greet builds the sentence");
}
```

‘use greeter::*;’ brings in everything public. Write ‘use greeter;’ instead and you call it as ‘greeter::greet(…)’, which is longer but says where each name came from — useful once you import several packages.

41.0.4. Running one test while you work

Give ‘loft test’ the file, or the file and one function:

```
$ cd greeter && loft test greet
test result: ok. 1 passed; 1 file
```

41.0.5. Check it compiles, without running anything

Inside the package, ‘loft check’ compiles everything and reports problems:

```
$ cd greeter && loft check
loft build: `native` ✓
```

A package that is only a library has no program to start, so there is nothing to run — ‘check’ still tells you the code compiles, which is what you wanted to know.

41.0.6. Coverage — what the tests did not reach

After the result, ‘loft test’ tells you which of your functions no test ever entered:

```
$ cd greeter && loft test
coverage: all 1 functions were entered by these tests
```

It is a list, never a percentage and never a gate. A percentage becomes a target, and tests written to move a number check nothing.

41.0.7. Using someone else’s package

Add it to the manifest under ‘[dependencies]’ and install:

```
[dependencies]
math = ">=0.1"
```

```
loft install
```

Then ‘use math;’ in your code. ‘loft install’ also writes a lock file that records the exact versions, so the same code builds the same way later.

```
fn main() {
```

The package these examples describe is `greeter`, and this is its one function — a `pub fn` in `src/greeter.loft` that its tests import.

```
  greeting = "hello, Ada!";
  assert(greeting == "hello, Ada!", "what greet(\"Ada\") returns");
  println(greeting);
}
```

42. Standard Library

42.1. Types

```
pub type boolean size(1)
```

Primitive types built into lav. True or false value.

```
pub type integer size(8)
```

64-bit signed integer.

```
pub type single size(4)
```

32-bit floating-point. Good for graphics and performance-sensitive math.

```
pub type float size(8)
```

64-bit floating-point. Use when precision matters.

```
pub type text size(4)
```

UTF-8 string.

```
pub type character size(4)
```

A single Unicode code point.

```
pub type u8 = integer limit(0, 255) size(1)
```

Integer subtypes for compact storage in struct fields. Behave as integer in expressions. 0 – 255, 1 byte.

```
pub type i8 = integer limit(-128, 127) size(1)
```

–128 – 127, 1 byte.

```
pub type u16 = integer limit(0, 65535) size(2)
```

0 – 65535, 2 bytes.

```
pub type i16 = integer limit(-32768, 32767) size(2)
```

–32768 – 32767, 2 bytes.

```
pub type i32 = integer size(4)
```

4-byte signed: −2_147_483_647 – 2_147_483_647. The bottom of the 32-bit range (`-2147483648`) is the null sentinel, so it is not a value an `i32` can hold — the same reservation `u32` makes at the top of its range.

```
pub type u32 = integer limit(0, 4294967294) size(4)
```

4-byte unsigned: 0 – 4_294_967_294. size(4) forces 4-byte storage and serialisation symmetric with `i32`. Stored as Parts::Int so the in-memory representation is signed i32; values in the upper half (2^31..2^32-2) round-trip through file I/O via 4-byte raw bytes but read back as negative i64s in expressions. For full 0..2^32-1 values needed in expression-time arithmetic, declare the field / variable as plain `integer` (8-byte) instead — `u32` is for the file / wire-format use cases (magic numbers, counts, tick fields) where the byte width is the load-bearing property.

42.2. Interfaces

Standard interfaces for bounded generic functions. A type satisfies an interface by defining the required operator or method.

```
pub interface Ordered
```

Types that support the `\<` comparison operator. Satisfied by integer, single, float, text, and any user type defining OpLt. ONE method is all a type has to define: inside a generic bounded by this, `\>`, `\<=` and `\>=` all derive from `\<` — `a \> b` is `b \< a`, `a \<= b` is `!(b \< a)`, `a \>= b` is `!(a \< b)`. Each evaluates its operands exactly once.

```
pub interface Equatable
```

Types that support the == equality operator. Satisfied by integer, single, float, text, boolean, and user types defining OpEq. `!=` derives from it (`a != b` is `!(a == b)`), so a type defining `op ==` gets both.

```
pub interface Addable
```

Types that support the + addition operator, returning the same type. Satisfied by integer, single, float, and user types defining OpAdd.

```
pub interface Numeric
```

Types that support `\*` and - (unary negation). Separate from `Addable` to allow fine-grained bounds without stub-name collisions. Satisfied by integer, single, float, and user types defining OpMul and OpMin.

```
pub interface Scalable
```

Types that support integer scaling via a `scale` method. Uses a method (not `op \*`) to avoid stub-name collision with `Numeric`. User types satisfy Scalable by defining `fn scale(self: T, factor: integer) -\> integer`.

```
pub interface Printable
```

Types that can be converted to text via a `to\_text` method. User types satisfy Printable by defining `fn to\_text(self: T) -\> text`.

42.3. Math

Functions for numeric computation. All trigonometric functions work in radians. Both single and float variants exist for every function — choose single for speed, float for precision. Integer operations loft#984 — the declared-RANGE guard, applied to the VALUE at a store into a slot that declares one (`integer limit(lo, hi)`, a narrow width). A value outside `lo..=hi` takes the slot's DEFAULT — `dflt` is the lowest value in range, or the null sentinel where the slot admits null — and reports it, rather than being wrapped, aliased, or dropped.

It guards the VALUE rather than the store, which is what lets one op reach both a FIELD and a VARIABLE: a variable has no store op to carry the range, and that is why a declared range on a local went unenforced entirely.

A null passes straight through: whether a null may land in this slot is `(N-Store)`'s question, answered at compile time, and substituting `lo` for it here would quietly invent a value the program never computed.

Emitted only where the value is NOT provably in range, so ordinary in-range code pays nothing (`parser::expressions::range\_guard`).

```
pub fn abs(both: integer) -> integer
```

Absolute value. Removes the sign from a negative integer.

```
pub fn floor_mod(both: integer, divisor: integer) -> integer?
```

Floor modulo — the remainder that takes the sign of the DIVISOR, so `x.floor\_mod(n)` lands in `\[0, n)` for a positive `n`. The `%` operator truncates toward zero and keeps the DIVIDEND's sign (`-1 % 3 == -1`); `floor\_mod` wraps (`(-1).floor\_mod(3) == 2`), which is what circular indexing / wrap-around wants (`grid\[(i - 1).floor\_mod(w)\]`). `x.floor\_mod(0)` is null, like `%` (C80 — an undefined value is a null, never a fault).

```
pub fn abs(both: single) -> single
```

Absolute value for single-precision floats.

```
pub fn cos(both: single) -> single
```

Cosine. Use for circular motion: x = r * cos(angle).

```
pub fn sin(both: single) -> single
```

```
pub fn tan(both: single) -> single
```

```
pub fn acos(both: single) -> single?
```

```
pub fn asin(both: single) -> single?
```

```
pub fn atan(both: single) -> single
```

```
pub fn ceil(both: single) -> single
```

```
pub fn floor(both: single) -> single
```

```
pub fn round(both: single) -> single
```

```
pub fn sqrt(both: single) -> single?
```

```
pub fn atan2(both: single, v2: single) -> single
```

Arc tangent of y/x, preserving the correct quadrant. Use instead of atan when you have separate x/y components.

```
pub fn log(both: single, v2: single) -> single?
```

```
pub fn pow(both: single, v2: single) -> single?
```

Raises base to the power exp. Use for exponential growth curves and scaling.

```
pub fn abs(both: float) -> float
```

Absolute value for double-precision floats.

```
pub PI = OpMathPiFloat()
```

The ratio of a circle's circumference to its diameter (3.14159...).

```
pub E = OpMathEFloat()
```

Euler's number, the base of natural logarithms (2.71828...).

```
pub fn cos(both: float) -> float
```

Cosine. Use for circular motion: x = r * cos(angle).

```
pub fn sin(both: float) -> float
```

```
pub fn tan(both: float) -> float
```

```
pub fn acos(both: float) -> float?
```

```
pub fn asin(both: float) -> float?
```

```
pub fn atan(both: float) -> float
```

```
pub fn ceil(both: float) -> float
```

```
pub fn floor(both: float) -> float
```

```
pub fn round(both: float) -> float
```

```
pub fn sqrt(both: float) -> float?
```

```
pub fn atan2(both: float, v2: float) -> float
```

Arc tangent of y/x, preserving the correct quadrant. Use instead of atan when you have separate x/y components.

```
pub fn log(both: float, v2: float) -> float?
```

```
pub fn pow(both: float, v2: float) -> float?
```

Raises base to the power exp. Use for exponential growth curves and scaling.

42.4. exp / ln / log2 / log10

Raises E (2.71828…) to the power v. Use for exponential growth models.

```
pub fn exp(both: single) -> single
```

```
pub fn exp(both: float) -> float
```

Double-precision exp (e^v).

```
pub fn ln(both: single) -> single?
```

```
pub fn ln(both: float) -> float?
```

Double-precision natural logarithm.

```
pub fn log2(both: single) -> single?
```

```
pub fn log2(both: float) -> float?
```

Double-precision base-2 logarithm.

```
pub fn log10(both: single) -> single?
```

```
pub fn log10(both: float) -> float?
```

Double-precision base-10 logarithm.

42.5. min / max / clamp

```
pub fn min(both: integer, b: integer) -> integer
```

@PLN25 F1b(b): each has a NON-NULL overload (`-\> τ`) and a `τ?` overload (`-\> τ?`). The call dispatch picks the `τ?` overload whenever ANY argument is STATICALLY nullable, so an explicit `integer?`/`single?`/`float?` argument — INCLUDING a division result (`1 / z`, now typed `τ?` under DN3) — gets a correctly-typed nullable RESULT and routes to the `τ?` body. The non-null bodies are therefore CLEAN (no `return null` guard): they only ever receive truly-non-null args, so they no longer rely on the STD_SOURCE (N-Store) exemption, which is now retired. The `τ?` overloads carry the propagation: null if either argument is null. Smallest of two integer values.

```
pub fn max(both: integer, b: integer) -> integer
```

Largest of two integer values.

```
pub fn clamp(both: integer, lo: integer, hi: integer) -> integer
```

Clamps v into the inclusive range [lo, hi]. Returns null if any argument is null.

```
pub fn min(both: single, b: single) -> single
```

Smallest of two single-precision float values.

```
pub fn max(both: single, b: single) -> single
```

Largest of two single-precision float values.

```
pub fn clamp(both: single, lo: single, hi: single) -> single
```

Clamps v into the inclusive range [lo, hi]. Returns null if any argument is null.

```
pub fn min(both: float, b: float) -> float
```

Smallest of two double-precision float values.

```
pub fn max(both: float, b: float) -> float
```

Largest of two double-precision float values.

```
pub fn clamp(both: float, lo: float, hi: float) -> float
```

Clamps v into the inclusive range [lo, hi]. Returns null if any argument is null.

```
pub fn approx(both: float, b: float, eps: float) -> boolean
```

Approximate equality: true when a and b differ by at most eps (inclusive). == on float/single is EXACT IEEE equality (@PLN102) — use `approx` when you want a tolerance, e.g. comparing computed transcendentals: `approx(sqrt(2.0) \* sqrt(2.0), 2.0, 1e-9)`. A null (NaN) operand is not approximately equal to anything, so the result is false.

```
pub fn approx(both: single, b: single, eps: single) -> boolean
```

Approximate equality for single-precision floats (see the float overload).

42.6. Text

Functions for working with text (UTF-8 strings) and character values. Read the value of a variable and put a reference to it on the stack

```
pub fn len(both: text) -> integer
```

Number of characters (Unicode code points) in the text (@PLN110) — the human count, as in every mainstream language. For a byte length (bounds checks, the limit of byte-positioned indexing / slicing) use `size`.

```
pub fn size(both: text) -> integer
```

Number of bytes in the text (@PLN110). This is the bound for byte-positioned operations: `s\[i\]`, slices `s\[a..b\]`, and `find`/`rfind` all use byte offsets. For the human character count use `len`.

```
pub fn len(both: character) -> integer
```

Byte length of the character's UTF-8 encoding (1–4).

Splits self on every occurrence of separator and returns the parts as a vector. Use to parse CSV lines or space-separated tokens.

```
pub fn split(self: text, separator: character) -> vector<text>
```

```
pub fn split_text(self: text, separator: text) -> vector<text>
```

```
pub fn starts_with_at(self: text, pos: integer, prefix: text) -> boolean
```

Functions for searching, transforming, and classifying text and character values. Character classification functions return true only if every character in the text satisfies the condition. The single-character variants test one code point. (`starts\_with` / `ends\_with` moved to `02\_files.loft` so the path helpers there can call them; both still available as `text.starts\_with` / `text.ends\_with`.) Returns true if self contains `prefix` starting at byte position `pos`. Sugar over `self\[pos..pos + prefix.size()\] == prefix` for the common "is this token at this offset?" pattern in scanners / parsers. Returns false (not an error) when pos + prefix.size() exceeds self.size() — same shape as `starts\_with` for the "prefix too long for input" case. Faster than comparing characters one at a time for a known prefix at `pos`. Example: @STD-001

```
pub fn char_slice(self: text, from: integer, to: integer) -> text
```

The characters of `self` from `from` up to (not including) `to` — a slice that counts CHARACTERS where `self\[a..b\]` counts BYTES. ⚠️ **This is the cure for the commonest text bug in this stack.** `len(text)` counts characters while `self\[a..b\]` slices bytes, and loft snaps a byte cut outward to a character boundary — so a count computed from `len` and applied as a byte range fits FEWER characters than it measured. The failure is not mojibake but something quieter: the text comes back SHORT, silently, and only when it is not ASCII, so every ASCII test stays green. `"héllo"\[0..4\]` is `"hél"`; `"héllo".char\_slice(0, 4)` is `"héll"`. Use it wherever a character COUNT becomes a cut: fitting a label to a width, wrapping a line, a caret or a selection in a text field, truncating with an ellipsis. Reach for `self\[a..b\]` only when the numbers came from a byte source — `find`, `size`, `byte\_at`. Negative bounds count from the end, as a byte slice's do (`char\_slice(-2, n)` is the last two characters); both ends clamp, and a reversed or empty range answers `""` rather than failing.

```
pub fn trim(both: text) -> text[both]
```

(Path helpers `dir` / `basename` / `join` / `resolve` moved to `02\_files.loft` § Path helpers, so they're available before `03\_text.loft` loads — same call shape, same `pub fn` signatures.) Removes leading and trailing whitespace. Use when processing user input or file content.

```
pub fn trim_start(self: text) -> text[self]
```

Removes leading whitespace only.

```
pub fn trim_end(self: text) -> text[self]
```

Removes trailing whitespace only.

```
pub fn find(self: text, value: text) -> integer?
```

Returns the byte index of the first occurrence of value, or null if not found. Use to locate substrings before slicing. Nullable: null when value is absent (@PLN102 keystone step 4 — the type is honest about the not-found case).

```
pub fn rfind(self: text, value: text) -> integer?
```

Returns the byte index of the last occurrence of value, or null if not found. Use to find file extensions or the last path separator. Nullable: null when value is absent (@PLN102 keystone step 4 — the type is honest about not-found).

```
pub fn contains(self: text, value: text) -> boolean
```

Returns true if value appears anywhere in self. `s.contains(v) == s.find(v) != null` — `find` returns the position, `contains` is its found?/not-found? projection. NOT superseded: `contains` is the clearer idiom for a membership test (a boolean predicate), and folding it onto `find` (below) is an implementation detail, not a reason to steer callers away from it.

```
pub fn replace(self: text, value: text, with: text) -> text
```

Returns a copy of self with every occurrence of value replaced by with.

```
pub fn to_lowercase(self: text) -> text
```

Returns a lowercase copy. Use for case-insensitive comparisons.

```
pub fn to_uppercase(self: text) -> text
```

Returns an uppercase copy.

```
pub fn is_lowercase(self: text) -> boolean
```

True if the text is non-empty and all characters are lowercase letters.

```
pub fn is_lowercase(self: character) -> boolean
```

True if the character is a lowercase letter.

```
pub fn is_uppercase(self: text) -> boolean
```

True if the text is non-empty and all characters are uppercase letters.

```
pub fn is_uppercase(self: character) -> boolean
```

True if the character is an uppercase letter.

```
pub fn is_numeric(self: text) -> boolean
```

True if the text is non-empty and all characters are numeric digits (Unicode numeric, not just ASCII 0–9).

```
pub fn is_numeric(self: character) -> boolean
```

True if the character is a numeric digit.

```
pub fn is_alphanumeric(self: text) -> boolean
```

True if the text is non-empty and all characters are letters or digits. Use to validate identifiers or tokens.

```
pub fn is_alphanumeric(self: character) -> boolean
```

True if the character is a letter or digit.

```
pub fn is_alphabetic(self: text) -> boolean
```

True if the text is non-empty and all characters are alphabetic.

```
pub fn is_alphabetic(self: character) -> boolean
```

True if the character is alphabetic.

```
pub fn is_whitespace(self: text) -> boolean
```

True if the text is non-empty and all characters are whitespace. Use to detect blank lines.

```
pub fn is_whitespace(self: character) -> boolean
```

True if the character is whitespace.

```
pub fn is_control(self: text) -> boolean
```

True if the text is non-empty and all characters are control characters.

```
pub fn is_control(self: character) -> boolean
```

True if the character is a control character.

```
pub fn join(self: vector<text>, sep: text) -> text
```

Joins parts with sep between each consecutive pair. Returns "" for an empty vector. Use to build comma-separated lists, path segments, or any delimited output. Example: @STD-003

```
pub fn byte_at(self: text, i: integer) -> integer
```

These two are TEXT functions that happen to sit after the Environment marker. Re-opening the section files them under Text in the generated reference, which is where a reader looks for them — `text\_from\_bytes` existed for two releases and was reported as missing (loft#748) because `doc/stdlib-text.html` did not list it. Declaring the section rather than MOVING the definitions is deliberate: definition order is the order types are minted, and the native `init()` replays that order, so relocating `vector\<u8\>`'s first mention shifts every type id after it (loft#739 / #742). A comment cannot. Return the BYTE at position `i` (0..len) as integer 0-255, or 0 for out-of-bounds. Unlike `text\[i\]`

which decodes the UTF-8 codepoint containing byte `i` (walking back through continuation bytes), `byte\_at(i)` is a pure O(1) byte read. Use in ASCII-heavy scanning hot paths (tokenisers, regex- like loops) where the UTF-8 decode is wasted work — every non-ASCII byte still returns a valid 0-255 number; the caller compares against ASCII constants so byte semantics suffice. 5-10× faster than `text\[i\]` for pure-ASCII checks.

```
pub fn text_from_bytes(bytes: vector<u8>) -> text
```

Build a text from the raw UTF-8 bytes of a vector<u8> — the inverse of byte_at. Use in binary decoders (CBOR text, HPKE byte composition) that assemble a byte buffer and need to turn it back into text. Bytes that are not valid UTF-8 yield the empty text (never a crash); validate first if you must tell "empty input" from "invalid bytes" apart.

```
pub fn chr(cp: integer) -> text
```

Build a one-character text from a Unicode CODE POINT — the inverse of the `ch as integer` that text iteration already gives you, and the code-point twin of text_from_bytes' byte route. Use when you have a number and need the character it names: decoding an escape (`\\u{...}`, an HTML entity), walking a code-point table, or reassembling text a code point at a time. chr(65) "A" chr(233) "é" chr(20013) "中" chr(128512) "😀" A code point that names no character answers the EMPTY text, never a crash: a surrogate (D800-DFFF), anything past U+10FFFF, a negative number — and also 0, because `character` uses 0 as its null and text iteration stops there, so a NUL built here could not be read back. If you need an embedded NUL, go through the byte route: `text\_from\_bytes(\[0\])` carries one. Example: @STD-002

42.7. Collections

Operations on vector<T> — the primary ordered collection type. Vectors are grown by appending with += and elements are accessed by index. All structures are passed by reference instead of by value

```
pub fn len(both: vector) -> integer
```

Number of elements in the vector. Use in loop bounds: for i in 0..v.len().

```
pub fn len(both: sorted) -> integer
```

Number of elements in a sorted collection.

```
pub fn clear(both: vector)
```

Remove all elements from the vector, setting its length to 0.

```
pub fn len(both: hash) -> integer
```

Number of elements in a hash collection.

```
pub fn size(both: hash) -> integer
```

The byte footprint of a hash (@PLN110): its full bucket table, holes included. The table is the hash's own allocation — `elms` slots, each a 4-byte record-id (an empty slot is a hole and still counts, because open addressing's spare capacity IS the format). Allocation-local: the entry records live in separate allocations and are not counted. Grows in steps as the table rehashes (load factor 0.75). 0 for an empty (unallocated) hash.

42.8. Output and Diagnostics

```
pub fn assert(test: boolean, message: text, file: text, line: integer)
```

Panics with message if test is false. Use to verify invariants during development. In production mode (–production), logs an error instead of aborting. The file and line are injected by the compiler; do not pass them manually. Safe to call from `par` workers: failures write to the log/stderr sink, which the host serialises across threads.

```
pub fn panic(message: text, file: text, line: integer)
```

Immediately terminates execution with message. Use for unrecoverable error states. In production mode (–production), logs a fatal entry instead of aborting. The file and line are injected by the compiler; do not pass them manually.

```
pub fn log_info(message: text, file: text, line: integer)
```

Write a structured log record at the chosen severity. The file and line are injected by the compiler; do not pass them manually. Safe to call from `par` workers: the host serialises log writes across threads.

```
pub fn log_warn(message: text, file: text, line: integer)
```

Like log_info, at warning severity.

```
pub fn log_error(message: text, file: text, line: integer)
```

Like log_info, at error severity.

```
pub fn log_fatal(message: text, file: text, line: integer)
```

Like log_info, at fatal severity.

```
pub fn print(v1: text)
```

Writes v to standard output without a newline. Use for progress output and building up a line incrementally.

```
pub fn println(v1: text)
```

```
pub fn eprint(v1: text)
```

Writes v to standard ERROR without a newline. Companion to print() — use to separate human-readable status / progress / summary lines from machine-readable stdout (JSON, structured data piped to downstream consumers).

```
pub fn eprintln(v1: text)
```

Writes v followed by a newline to standard ERROR.

```
pub fn len(both: spatial) -> integer
```

Number of elements in a spatial (radix / Morton tree) collection.

```
pub fn len(both: trie) -> integer
```

42.9. Vector aggregates

```
pub fn min_of < T: Ordered > (v: vector<T>) -> T?
```

Smallest element in a vector, or null when the vector is empty (@PLN102 keystone step 4 — the type is honest about the empty case). Works on any Ordered type (op <). Example: @STD-004

```
pub fn max_of < T: Ordered > (v: vector<T>) -> T?
```

Largest element in a vector, or null when the vector is empty (@PLN102 keystone step 4 — the type is honest about the empty case). Works on any Ordered type (op <). Example: @STD-004

```
pub fn sum < T: Addable > (v: vector<T>, init: T? = null) -> T
```

Sum of vector elements. Works on any Addable type. `init` is the identity to start from; leave it out and the element type's own zero is used (0, 0.0, ""). Example: sum([10, 20, 12], 0) == 42 Example: sum([10, 20, 12]) == 42 Example: @STD-005 @PLN102 arc C — `init` gained its default so `sum\_of(v)` can be REWRITTEN to `sum(v)`: `superseded-call` offers that rename and `loft fix` verifies every edit by compiling the result, so while `init` was required the only `\#superseded` symbol loft ships had its fix rejected on every program (loft#1003). A literal default is not spellable here (`init: T = 0` is "expected T, got integer"), so it is the nullable-with-discharge form, which needs loft#1016's per-monomorph `construct\_default(T)`. Additive: every existing `sum(v, init)` call is unchanged.

```
pub fn sum_of(v: vector<integer>) -> integer
```

Sum of all integer elements. Returns 0 for an empty vector. @PLN102 arc C — superseded by the general `sum(v, init)`; kept as a shim over it (the old form keeps working). `sum\_of(v)` == `sum(v, 0)`. Defined AFTER `sum` so the shim's call resolves — a forward reference to a generic is not yet supported.

```
pub interface Walkable
```

```
pub fn tree_walk < T: Walkable > (wk_root: T, cap: integer) -> vector<T>
```

Example: @STD-006

42.10. Assertions that report both sides

```
pub fn assert_eq < AssertValue: Equatable + Printable > (got: AssertValue, want:
AssertValue, what: text, file: text, line: integer)
```

Assert that two values are equal, naming BOTH of them when they are not. `assert(got == want, "…")` puts the expected value in the CONDITION, so a failure says what was got and leaves the reader to recover what was wanted by reading the expression. This says both. Works on any type that is `Equatable` (op ==) and `Printable` (to_text) — every built-in scalar, and a user type defining both. Failure behaves exactly as `assert` does, because it IS `assert`: the run aborts (and in `--production` logs an error instead), and the reported position is the CALLER's, not this function's. The file and line are injected by the compiler; do not pass them manually. The type variable is spelled `AssertValue` and NOT `T` on purpose. A type variable's bound stubs are keyed by its NAME (`t\_1T\_to\_text`), so every generic in the program spelling its variable `T` shares one namespace with a user `struct T` — and a bound declaring a method as common as `to\_text` or `op ==` then blocks that struct from defining its own. Measured: with these bounds on `T`, every user struct named `T` formatted as EMPTY and could not declare `to\_text` at all. loft#1153 tracks the collision itself; until it closes, a stdlib generic bound to a common method needs a name no one would write.

```
pub fn assert_ne < AssertValue: Equatable + Printable > (got: AssertValue, want:
AssertValue, what: text, file: text, line: integer)
```

Assert that two values DIFFER, naming the value both sides share when they do not. The mirror of assert_eq; same bounds, same failure behaviour, same position injection.

42.11. Field iteration support types

```
pub enum FieldValue {
  FvBool { v: boolean },
  FvInt { v: integer },
  FvLong { v: integer },
  FvFloat { v: float },
  FvSingle { v: single },
  FvChar { v: character },
  FvText { v: text },
}
```

Wraps a field value in a type-discriminated enum for compile-time field iteration.

```
pub struct StructField {
  name: text,
```

```
  value: FieldValue,
  // Was the field DECLARED nullable (`text?` rather than `text`)?
  //
  // The payload variants above are typed non-null, and `τ?` shares `τ`'s
  // runtime layout (@PLN25) — a nullable field spells absence with a SENTINEL
  // rather than a wrapper. So a null field's value arrives inside `FvText` /
  // `FvInt` / … as that sentinel, and this flag is what says so; without it a
  // reader would have to discover the possibility by being surprised.
  //
  // Mirrors `FieldInfo.nullable` from `type_of` (07_reflect.loft) on purpose:
  // the two field lists describe the same declaration and must agree.
  nullable: boolean,
}
```

Represents a single struct field during compile-time iteration.

```
pub fn to_text(self: integer) -> text
```

Built-in `to\_text` impls so primitives satisfy `Printable` automatically. Placed after every OpFormat* / OpAppend* declaration above so format-string interpolation in the bodies can resolve to the relevant built-in. Without these impls, Converts an integer to its decimal text. Built-in types define `to\_text` so they satisfy the `Printable` interface (used by `print`, format strings, …).

```
pub fn to_text(self: float) -> text
```

Converts a float to its text representation.

```
pub fn to_text(self: single) -> text
```

Converts a single-precision float to its text representation.

```
pub fn to_text(self: boolean) -> text
```

Converts a boolean to “true” or “false”.

```
pub fn to_text(self: character) -> text
```

Converts a character to a one-character text.

```
pub fn to_text(self: text) -> text
```

Returns the text unchanged — the identity case, so `text` satisfies `Printable` too.

42.12. File System

Types and functions for reading and writing files. A File value is obtained via file() and carries the path, format, and an internal reference. An environment variable as a name/value pair. Capability groups

(@PLN86) — the fs/env split a sandbox profile grants via `fs\#read` / `fs\#update` / `env\#read`; the functions below link in their signature.

```
pub struct EnvVariable {
  name: text,
  value: text,
}
```

```
pub enum Format {
  TextFile,
  LittleEndian,
  BigEndian,
  Directory,
  NotExists,
}
```

```
pub enum FileResult {
  Ok,
  NotFound,
  PermissionDenied,
  IsDirectory,
  Other,
}
```

Result of a filesystem-mutating operation (delete, move, mkdir). Use ok() to get a simple boolean, or match on specific variants for detailed error handling. Every variant can actually be produced: Ok on success, NotFound for a missing / out-of-project path, PermissionDenied when the OS refuses access, IsDirectory when a file op targets a directory (e.g. deleting a directory with delete()), and Other for anything else. (–native and –interpret classify from the OS error; the wasm host reports only Ok / Other, and NotFound still comes from the loft-level existence check.)

```
pub fn ok(self: FileResult) -> boolean
```

True when the result is `FileResult.Ok`. Use to test a file op for success. Example: @STD-012

```
pub struct File {
  path: text,

  size: integer,

  // @PLN116 — a bare enum field needs an explicit default (an enum's 0 is null,
and a
  // non-null field may not hold null).  A fresh File is `NotExists` until
`OpGetFile`
  // determines the real format for a valid path (see `file()`), so `NotExists`
is the
  // honest placeholder — it is overwritten before any read.
```

```
  format: Format = NotExists,

  ref: i32?,

  current: integer,

  next: integer,
}
```

```
pub fn content(self: File) -> text?fs#read
```

Reads the entire file as a UTF-8 text value. Use for small configuration files or scripts. Example: @STD-010

```
pub fn lines(self: File) -> vector<text> fs#read
```

Par-safe: reads the file into a worker-local store; the host bridge serialises filesystem access. Reads the file and splits it into lines. Strips trailing '\r' so CRLF files (Windows) and LF files (Unix) produce identical results. Use when processing line-by-line (logs, CSV, etc.). Example: @STD-011

Returns the platform path separator character: '\' on Windows, '/' elsewhere. Detected once at startup from the runtime filesystem.

```
pub fn path_sep() -> character
```

```
pub fn file(path: text) -> File fs#read
```

```
pub fn exists(path: text) -> boolean fs#read
```

```
pub fn exists(both: File) -> boolean fs#read
```

Filesystem stat (via file()); par-safe. Method form: f = file("path"); if f.exists() { ... } Also callable as exists(file_obj) via the 'both' parameter name.

```
pub fn delete(path: text) -> FileResult fs#update
```

```
pub fn move(from: text, to: text) -> FileResult fs#update
```

```
pub fn mkdir(path: text) -> FileResult fs#update
```

```
pub fn mkdir_all(path: text) -> FileResult fs#update
```

```
pub fn is_dir(path: text) -> boolean fs#read
```

Returns true if `path` exists and is a directory. Use to branch on a path's kind before descending into / reading it.

```
pub fn is_file(path: text) -> boolean fs#read
```

Returns true if `path` exists and is a regular file. Use to confirm a directory entry is a readable file before opening it.

```
pub fn list_dir(path: text) -> vector<text> ?fs#read
```

Lists the entry NAMES of directory `path` (base names, not full paths), sorted. @PLN102 H4 — a MISSING / non-directory path lists as NULL (distinct from an EMPTY directory, `\[\]`); discharge with `?? \[\]` to keep the old shape. Use to enumerate a directory; join with `path` to build full child paths. Example: @STD-010

```
pub fn read_bytes(path: text) -> vector<u8> ?fs#read
```

Reads the whole file `path` as raw bytes. @PLN102 H4 — a MISSING / unreadable file reads as NULL (distinct from an EMPTY file, `\[\]`); discharge with `?? \[\]` to keep the old shape. Binary-exact (round-trips with write_bytes); use for non-UTF-8 data — for text prefer `file(path).content()`. Example: @STD-010

```
pub fn write_bytes(path: text, bytes: vector<u8>) -> boolean fs#update
```

Writes `bytes` to file `path`, truncating any existing content. Returns true on success. The inverse of read_bytes; the pair round-trips a non-UTF-8 blob byte-for-byte.

```
pub fn mtime(path: text) -> integer fs#read
```

Modification time of `path` as Unix epoch SECONDS (integer — same i64 representation as file.size). Returns 0 on missing file / IO error / pre-epoch dates — caller treats 0 as "unknown" (matches scan.sh's `stat -c %Y || echo 0` fallback). Takes a path string rather than a File handle so the native + interp dispatch both use the same `n\_mtime` registration. Use for date-window filters: convert the returned seconds to YYYY-MM-DD and compare lexicographically against `ymd\_days\_ago(N)`.

```
pub fn store_durable_check(path: text) -> boolean fs#read
```

Durable-store integrity check. Returns true iff the `.dmeta` sidecar at `\<path\>.dmeta` validates against the main file at `\<path\>` (signature, header CRC, payload length, payload CRC, tier_id all OK). Returns false on any failure or missing file. Pair with `store\_durable\_seal` after a clean write session to record the new state. Desktop-only: the sidecar is built over memory-mapped files, so on the wasm targets this answers false — "nothing validates", which is the state that sends a caller down its rebuild branch (loft#1063).

```
pub fn store_durable_seal(path: text) -> boolean fs#update
```

Write a fresh `.dmeta` sidecar capturing the current main-file's byte length + CRC32 + a clean-close timestamp. Returns true on success, false on any I/O error. Call this after finishing a write session; if the program crashes between the last write and the seal, the sidecar stays stale and the next `store\_durable\_check` returns false → caller rebuilds. Desktop-only, like `store\_durable\_check`: on the wasm targets this answers false — the seal did not happen — so seal where the store is written and read it back with `store\_load` (loft#1063).

```
pub fn store_persist_bind(r: reference, path: text) -> boolean fs#update
```

"The collection IS the file." Re-root the Store backing the given reference at a file path so mutations are durable via mmap without any explicit save/load loop. Works for any store-rooted collection — `hash`, `sorted`, `ordered`, `index` (each keyed local/field is a dedicated Store). `hash` carries its bucket seed in its own record and `sorted`/`ordered`/`index` are comparison-based (no per-process state), so the persisted image is portable: a different process (a restart, or a remote reader) both iterates AND key-looks-up correctly. A reference that is NOT its own Store root fails soft (returns `false`), same as any I/O error. First call on a path that does NOT yet exist: serialises the current in-memory Store at the reference's slot to disk (padded to a valid ≥1024-word image with a tail free block), then mmaps it back. Caller's existing DbRefs into that slot remain valid. Call on a path that DOES exist: opens the file via mmap; the caller's prior in-memory contents at that slot are dropped in favour of the on-disk image. This is the load-on-startup path, assumes the on-disk layout matches the declared type. Both modes return `true` on success, `false` on any I/O / format error (no panic — the binding is fail-soft, callers fall back to JSON or rebuild-from-source). Typical dryopea-style pattern: pw = PaintedWorld { painted: [] } It snapshots the whole STORE `r` lives in, which is not always a store of just `r` (loft#757). A keyed LOCAL owns its store, so binding it writes a file for that collection. A keyed FIELD shares its container's store, so binding `pw.painted` above writes a file for `PaintedWorld` — carrying the container and every sibling collection — and that file will NOT load back into a bare `hash\<PaintedHex\[q, r\]\>`. Both are usable; they are just different files. Bind through the container consistently, or bind a local of the collection's own type when another program has to read the file. The compiler advises at the call when the argument is a field. `hash` carries its bucket seed in its own record and the comparison-based kinds hold no per-process state, so every persisted image is portable across processes.

```
pub fn store_persist_copy(r: reference, path: text) -> boolean fs#update
```

Write an image of `r` laid out for PAGING, and keep the live collection as it is. Use this for a file another program (or a browser) will READ — a vocabulary, a map, any shipped dataset — where what matters is how few pages a query touches. `store\_persist\_bind` copies the live bytes, so every record keeps its place and your references stay valid. That place is the layout: records sit where they were INSERTED, so a `trie` prefix query returning 20 records reads 20 scattered pages. This writes a rebuilt copy instead, with each record placed in its collection's own order — key order for a `trie` — so one prefix is one run. Measured on a 74,692-word vocabulary, one 20-record prefix query: 22 requests and 1.25 MB from a bound image, 4 and 0.26 MB from this one. The live collection is untouched: nothing moves, so every reference you hold stays valid, and the file is NOT bound (writes after this do not reach it). Call it when the data is final. To keep writing, use `store\_persist\_bind`. words: trie<Word[w]> = [] for w in vocabulary() { words += [Word { w: w }] } store_persist_copy(words, "vocab.store") Returns `false`

on an I/O or format error, and for a collection whose shape the rebuild cannot carry. The image is sized to its content, with none of the growth slack a bound file keeps.

```
pub fn store_load(r: reference, path: text) -> boolean fs#read
```

Load a persisted store IMAGE fully into memory, populating the empty store-rooted collection `r` so it can be queried like any in-memory collection. The portable, read-only counterpart of `store\_persist\_bind`: it HEAP-COPIES the file (no mmap), so it works on EVERY backend — including wasm, which has no mmap — but is NOT durable (writes stay in memory and never reach the file). Use it to open a snapshot for querying where `store\_persist\_bind` can't run (a browser / wasm target) or where a durable live binding isn't wanted. Returns `false` on a missing / truncated / wrong-format file (no panic; assumes the on-disk layout matches `r`'s declared type). @PLN97 arc G Phase 1 — the whole-file load the browser working-set path builds on (loft#522). h: hash<Rec[id]> = [] store_load(h, "world.store")

```
pub fn store_load_url(r: reference, url: text, sha256: text) -> boolean fs#read
```

Load a persisted store IMAGE over HTTP(S) from a TRUSTED source, establishing authenticity BEFORE the bytes are adopted: fetch the whole image at `url`, verify its SHA-256 against the caller-pinned `sha256` (lowercase hex), and only on a match heap-load it into `r` — the bytes never touch disk. A fetch error or a hash mismatch REFUSES the load (returns false, loads nothing). The fetch⟶verify⟶trust discipline the registry install uses, bridged onto the store loader; `url` may be `http(s)://` or `file://`. Whole-file counterpart of the paged `store\_load\_key(s)/store\_load\_range` loaders. @PLN97 arc G Phase 0. h: hash<Rec[id]> = [] store_load_url(h, "https://cdn.example/world.store", "<sha256-hex>")

```
pub fn store_load_url_trusted(r: reference, url: text) -> boolean fs#read
```

Load a whole store IMAGE over HTTP(S)/file:

```
pub fn store_load_untrusted(r: reference, path: text) -> boolean fs#read
```

Load a store IMAGE from a local file that may be UNTRUSTED into `r` — the structurally-validated counterpart of store_load. Reads the file and validates its block structure BEFORE adoption, so a crafted or corrupt file cannot hang (0-size block) or drive a heap over-read; it is rejected (false). store_load is faster (validates only in debug) for a file you produced yourself; use this for a file whose provenance you don't control. @PLN97 arc G Phase 2. h: hash<Rec[id]> = [] if !store_load_untrusted(h, "downloaded.store") { /* rejected — malformed */ }

```
pub fn store_verify(r: reference) -> boolean
```

Structural integrity check of a store-rooted collection's heap graph: verify every internal pointer targets a live record — no dangling / out-of-bounds / cyclic edge. Returns true when sound, false (with a reason on stderr) when not. The verifier behind the working-set loader's confidence: after any `store\_load\*`, `store\_verify(local)` proves the (partial) copy produced a structurally valid heap,

not one with a pointer left aimed at the source. h: hash<Rec[id]> = [] store_load_key(h, “block.store”, 42) assert(store_verify(h))

```
pub fn store_reclaim(r: reference) -> integer fs#update
```

Give back the free space at the END of a store-rooted collection’s store, and return the BYTES handed back (0 when there was nothing to give). For a store bound with `store\_persist\_bind` that is the FILE shrinking; otherwise it is memory returned to the allocator. Records are never moved, so every reference into the collection stays valid. You say when, because only the program knows whether a drop is permanent: a collection that shrinks and grows again would just pay to re-grow. Read `store\_memory()` first, and read it as a RANKING rather than a quantity: its `tail%` says which store is worth reclaiming, and it is NOT the size of the return. A reclaimed store lands at `tail 11%` — a growth reserve the allocator keeps — so what comes back is `tail% - 11%` of the resulting capacity, never the whole tail. Measured across eight shapes with tails from 13% to 60%: at a 13% tail the report suggests 36 KB and the call hands back 5832 bytes. Treat a reading near 11% as nothing to get back, not a little. Its `inner%` is the space BETWEEN records, which this does not touch — though the number RISES after a reclaim, because the capacity it is a percentage of has shrunk. WHERE the drop was matters as much as how big it was. `mergeable` counts adjacent free neighbours that never coalesced, so it measures how CONTIGUOUS the drop was, and that is exactly the part this call can fix: the same 2000 records dropped contiguously merge 2004 free blocks down to 7 and hand back 25400 bytes, while dropped alternately they leave 1997 blocks standing forever — the live records between them are what keeps them apart — and hand back 16312. Example: @FTR-001 Example: @FTR-002 You do NOT need this to right-size a file at the END of a run. A bound store keeps its file AS the live arena, and the arena’s capacity grows by 7/3 and never shrinks by itself — so mid-run the file is a rung on a ladder, not a measure of content, and can sit 57% above what it holds. Releasing the collection hands that tail back on its own, so the file a program leaves behind follows its content whether or not this was ever called (loft#752). Call it MID-RUN, when a live set has dropped for good and the memory (or the disk) is wanted back before the end. world: hash<Hex[q, r]> = [] store_persist_bind(world, “world.store”)

```
pub fn store_release(r: reference) -> integer fs#update
```

Say “everything I have written so far is finished”: start writing it out to the file and stop holding it in memory. Returns the BYTES dropped from the resident set (0 when there was nothing to drop, or the collection is not bound to a file). For a GENERATOR streaming a large collection into a store bound with `store\_persist\_bind`. Without it the resident set grows with everything written so far, and the kernel only learns which pages are finished by evicting the wrong ones first. Measured on a 20 000-record build, one call per record: peak memory 44.3 MB -> 2.2 MB (20x), at no cost in wall clock. tiles: hash<TTile[tkey]> = [] store_persist_bind(tiles, “tiles.store”) Content is untouched and every reference into the collection stays valid: nothing moves and nothing is freed. Reading a released record simply re-reads it from the file, at the cost of one page fault. So this is a HINT — calling it too often, or on the wrong collection, costs a little speed and can never cost an answer. It pays when records are written IN KEY ORDER and not returned to. A generator that keeps many records open at once leaves the store scattered with free blocks (measured: 3 691 against 10 for the same data written in order) and the allocator then keeps re-reading them, so the same call gives back 1.0x instead of 20x. If your

build streams in cell order, this is close to free; if it does not, sort it first and this is the reason to. Not `store\_reclaim`, which gives back the FILE's unused TAIL and changes its size. This changes only what is resident, and never the file's length. Not a durability barrier either — it asks for writeback to START; `store\_durable\_seal` is what promises the bytes have landed.

```
pub fn store_bind_lazy(local: reference, source: text) -> boolean
```

Working-set load: fetch ONE integer-keyed entry from a persisted HASH image into the empty local hash `local`, reading only the pages the lookup touches — not the whole file. The bounded-fetch counterpart of `store\_load`, for when `local` should hold only the entries actually asked for (a phone pulling the few map tiles a route needs from a large block). `path` is a local file or an `http(s)://` URL served with `Range`, on every target including the browser (`--html`), where the fetch goes through the same bridge `store\_load\_url\_trusted` uses. Returns false when the key is absent, the file is unreadable, or the collection is not an integer-keyed hash. The entry's own fields are RELOCATED into the local store, so `text`, nested structs and flat vectors all come across; only a `vector\<text\>` or a `vector\<vector\>` is refused (its element pointers would dangle), and a refusal says so on stderr rather than looking like an absent key. @PLN97 arc G (loft#522). tiles: hash<Tile[id]> = [] store_load_key(tiles, "block.store", 42) @PLN129 arc A — bind a COLLECTION to a lazy source. After this, a lookup that MISSES fetches that one entry and inserts it, so the next lookup is an ordinary resident hit; a lookup that hits never leaves the process. The collection is therefore automatically the cached data set — there is no separate cache. Per COLLECTION, not per store: `persons` and `companies` are different sources, and two collections of one type can bind differently. Binding replaces, and may be done before the collection holds anything. `source` is either an IMAGE — what `store\_load\_key` accepts: a local `.store` file or an `http(s)://` URL served with Range — or a DATABASE, named by a driver prefix. Returns false for a null collection. persons: hash<Person[id]> = [] store_bind_lazy(persons, "people.store") p = persons[42] @PLN129 arc B — `sqlite:\<path\>` binds to a table instead, and the query is DERIVED from the collection's own type: the table is the element type's name lowercased, the columns are its fields, and the `WHERE` is the collection's key. Nothing is written down twice. persons: hash<Person[id]> = [] Read-only, and the connection enforces it. A binding that cannot be served — a field that is not a column, a collection whose KIND the source cannot read — is REFUSED rather than served wrongly, and says so through `store\_lazy\_error`. sqlite is opened on the first fault, so a program that binds no database loads nothing. FALSE means the binding was not made, and it is worth checking. A `.store` IMAGE is read a page at a time, which only a `hash` or a `trie` supports: a `sorted`, `index` or `spatial` bound to one is refused HERE, at the call that is wrong, rather than answering `null` at every later lookup (loft#802). Those kinds load whole — `store\_load` / `store\_load\_url\_trusted` carry all of them. A DATABASE source judges its own schema on the first fault instead, since what it can serve is a fact about the other end. if !store_bind_lazy(tiles, "tiles.store") { store_load(tiles, "tiles.store");

```
pub fn store_lazy_query(local: reference, condition: text) -> integer
```

@PLN129 arc B2 — run an explicit condition against this collection's bound DATABASE source and pull every matching row into the collection. Answers how many records the collection gained. The escape hatch for what the collection's KEY cannot express — a predicate on another column, a pattern, a range nobody declared an index for. A keyed lookup derives its own query and needs no call; this

one cannot be derived, so it is written down and visible rather than happening behind a lookup. found = store_lazy_query(persons, “name LIKE ‘Ada%’”); p = persons[42] The rows land IN the collection, not in a separate result: a person reached by this query and the same person reached by a later lookup are ONE record, and a row already resident is left alone rather than fetched twice. So `len` and iteration keep answering “what have I got” — after this, more. `condition` is SQL, sent as written (the connection is read-only). Answers 0 both when nothing matched and when the query could not run; `store\_lazy\_error` tells those apart.

```
pub fn store_lazy_range(local: reference, lo: integer, hi: integer) -> integer
```

@PLN129 arc B step 8 — pull a whole KEY RANGE from this collection’s bound DATABASE source in ONE query. Answers how many records the collection gained. This is what keeps lazy reading usable rather than merely correct. Fetching 500 records one lookup at a time is 500 round trips; the same 500 as a range is one. So when you know the span you want, ask for the span: store_lazy_range(events, 100, 199); The collection must be ORDERED (`sorted` or `index`) — a `hash` has no order to range over — and keyed on ONE column, since two numbers cannot say which value pins a composite key’s leading column; use `store\_lazy\_query` for that. Both bounds are inclusive, in the collection’s own key order. A record already resident is left alone. Answers 0 when nothing matched AND when the query could not run; `store\_lazy\_error` tells those apart.

```
pub fn store_lazy_error(local: reference) -> text
```

@PLN129 arc C — why a lazy fetch for this collection could not REACH its source, or “” when it is healthy. A lookup cannot tell you this. C80 says a value read never raises, so a miss answers `null` whether the key is genuinely absent or the source is unreachable — and those are different facts, one stable and one not. Ask this after a null to tell them apart: p = persons[42]; if p == null { why = store_lazy_error(persons); if why == “” { /* really no such person */ } else { /* could not reach: {why} */ } } The FIRST failure’s reason, kept — not the last: it names the original cause, and later ones are usually the same failure repeating. Nothing clears it but `store\_lazy\_clear`. An absence does NOT, and neither does a later success: reaching the source now says nothing about what an earlier failure already lost, and answering “healthy” over a traversal that missed data is the silent wrong answer this channel exists to prevent.

```
pub fn store_lazy_faults(local: reference) -> integer
```

@PLN129 arc C — how many fetches could not REACH this collection’s source. 0 is healthy. The magnitude behind `store\_lazy\_error`: after a traversal it answers “how incomplete am I”.

```
pub fn store_lazy_clear(local: reference) -> boolean
```

@PLN129 arc C — acknowledge this collection’s fetch failures, returning whether there was anything to acknowledge. The ONLY thing that clears them. A later fetch happening to succeed does NOT: a traversal whose first lookup could not reach the source and whose second could is MISSING data, and answering “healthy” afterwards would be exactly the silent wrong answer this channel exists to prevent. Clearing is a caller saying “I have seen this”, which is a different event entirely.

```
pub fn store_lazy_fail(local: reference, why: text) fs#read
```

@PLN133 S8 — a loft DRIVER reporting that it could not reach its source. The writing end of the channel `store\_lazy\_error` reads. A driver written in loft (`fn lazy\_fetch(...)`) has the same three answers a Rust source has, and two of them are an integer: `1` inserted, `0` absent. The third is not — "the source is down" carries a REASON, and answering `0` for it is exactly the silent wrong answer arc C exists to prevent, because a caller cannot tell it from "no such person". fn lazy_fetch(coll: hash<Person[id]>, source: text, key_int: integer, key_text: text) -> integer { if !db.db_open(source) { store_lazy_fail(coll, "cannot open {source}: {db.db_last_error()}"); return 0; } ... } Sticky and counted exactly like a Rust source's failure: the FIRST reason is kept, every failure is counted, and only `store\_lazy\_clear` clears them.

```
pub fn store_load_key(local: reference, path: text, key: integer) -> boolean
fs#read
```

```
pub fn store_load_key_text(local: reference, path: text, key: text) -> boolean
fs#read
```

Text-keyed form of `store\_load\_key`: fetch ONE entry from a persisted `hash\<T\[textkey\]\>` or `trie\<T\[textkey\]\>` (a place-name / string-id index) into `local`, reading only the pages the lookup touches. Returns false when the key is absent or the collection isn't a copyable text-keyed hash or trie. @PLN97 arc G (loft#522), @PLN134 for the trie. places: hash<Place[name]> = [] store_load_key_text(places, "gazetteer.store", "Amsterdam")

```
pub fn store_load_prefix(local: reference, path: text, pre: text, limit: integer)
-> integer fs#read
```

Prefix form: fetch every entry whose text key begins with `pre` from a persisted `trie\<T\[k\]\>` into `local`, reading only the pages the prefix walk touches — what a search box needs, and what a `sorted` range cannot express without a hand-built successor string. Returns the count loaded. `limit` caps the WALK, not just the answer: with `limit` 8 the ninth record is never stepped to, so its pages are never fetched. A negative `limit` means no cap, which on a common prefix reads the whole run. @PLN134. words: trie<Word[w]> = [] store_load_prefix(words, "vocab.store", "kerk", 20)

```
pub fn store_load_box(local: reference, path: text, from: vector<integer>, till:
vector<integer>, limit: integer) -> integer fs#read
```

Box form: fetch every entry inside the closed bounding box `from..till` from a persisted `spatial\<T\[x, y\]\>` into `local`, reading only the pages the box walk touches — what a map viewport needs. Returns the count loaded. The corners are vectors so the same call serves 1, 2 or 3 axes, and writing them the other way round names the same box. TWO bounds, and a map needs both. `limit` caps the WALK, not just the answer: with `limit` 200 the 201st marker is never stepped to, so its pages are never fetched (a negative `limit` means no cap). And the BOX bounds it — the Morton interval between two corners is a superset the Z-order curve threads in and out of, so a wide, shallow viewport would otherwise read 1.46 M records to return 4 k. Measured on a 3.2 M-point map index:

5.3 pages of 64 KB for the first viewport, 1.5 for each pan after it, against a 158 MB whole-image download. Those numbers assume a dataset written with locality in every axis, which is the one thing this call cannot do for you. The Morton walk is symmetric; a file laid out along ONE axis is not, so a box crossing that axis pays for every stride it crosses. On a 62 500-point grid returning the same 500 records, a 250x2 box read 107 pages of 64 KiB against 7 for its 2x250 mirror, and the two swapped when the identical data was written in the other axis order — 81 % of the image to return 0.8 % of the records. Write such a dataset TILED: it has no bad orientation, and for the square-ish boxes a viewport uses it beats either linear order. @PLN136. pins: spatial<Pin[x, y]> = [] store_load_box(pins, "map.store", [x1, y1], [x2, y2], 200)

```
pub fn store_load_keys(local: reference, path: text, keys: vector<integer>) ->
integer fs#read
```

Plural form of `store\_load\_key`: fetch the given integer keys' entries into `local` in one call (the paged reader is opened once and its cache reused), returning how many were found. Keys absent from the remote are skipped. Same relocation rules as `store\_load\_key`. @PLN97 arc G (loft#522). tiles: hash<Tile[id]> = [] got = store_load_keys(tiles, "block.store", [7, 13, 42])

```
pub fn store_load_keys_text(local: reference, path: text, keys: vector<text>) ->
integer fs#read
```

Plural form of `store\_load\_key\_text`, and the text twin of `store\_load\_keys`: fetch the given text keys' entries into `local` in ONE call, returning how many were found. Keys absent from the remote are skipped. Serves a `hash\<T\[k\]\>` and a `trie\<T\[k\]\>` alike, like the single-key form. loft#1064. Looping `store\_load\_key\_text` is not the same call: each one opens its own reader, so a ring of twenty asset names re-fetches the same 64 KiB bucket-table page twenty times. This opens the reader once and shares its page cache — which is what makes it the PREFETCH call: ask for the ring at a load or level boundary, so no frame is ever the thing that discovers it needs an asset. art: hash<Blob[bl_key]> = [] got = store_load_keys_text(art, "pack.store", ["page/mobs", "hit.ogg"])

```
pub fn store_load_range(local: reference, path: text, lo: integer, hi: integer) -
> integer fs#read
```

Range form: fetch every entry whose integer key is in [lo, hi] from a persisted `sorted\<T\[k\]\>` into `local`, reading only the pages the range walk touches — the ordered-collection counterpart of `store\_load\_keys` (a phone pulling the corridor of map tiles a route crosses). Returns the count loaded. @PLN97 arc G (loft#522). tiles: sorted<Tile[tkey]> = [] store_load_range(tiles, "block.store", lo_cell, hi_cell)

```
pub fn set_file_size(self: File, size: integer) -> FileResult fs#update
```

Truncates or extends the file to `size` bytes. Returns FileResult.Ok, or an error variant (IsDirectory / NotFound / Other).

```
pub fn seek(self: File, pos: integer) -> boolean fs#update
```

Moves the read/write position to `pos` bytes from the start, for random access into a binary file: read an index, jump to the record it names, read that. Equivalent to `self\#next = pos`, which is the operator form and has always worked; this is the NAME the operation was already documented under, and a consumer who reached for it (three call forms, all of them this one) concluded random access was unsupported and restructured their file format around it. Returns false — a no-op — when there is nothing to seek in: a directory, an absent file, a negative `pos`, or a file the process has not yet read from or written to (the OS handle is opened by the first I/O, so seeking before it exists has nothing to move). Seeking PAST the end is allowed: the position is remembered and a following write extends the file, which is how a free-list or an update-in-place walk lands.

```
pub fn position(self: File) -> integer
```

The current read/write position in bytes, i.e. where the next read or write will land. The read side of [seek]; `self\#next` is the operator form. Distinct from `self\#index`, which is where the LAST read STARTED — after `x = f\#read as i32` on a fresh file, `position` is 4 and `f\#index` is 0. A file this process has not read from or written to yet has no position, and reports null rather than 0 — 0 is a legitimate position, so returning it would make "not opened" indistinguishable from "at the start". Discharge with `?? 0` when the distinction does not matter.

```
pub fn sync(self: File) -> boolean fs#update
```

```
pub fn files(self: File) -> vector<File> fs#read
```

```
pub fn write(self: File, v: text) -> FileResult fs#update
```

Writes v as UTF-8 text to the file, overwriting existing content. Returns FileResult.Ok on success and FileResult.Other on an OS write failure (disk full, permission denied, a bad path) — a failed write is OBSERVABLE, not silently swallowed. Discarding callers (`f.write(s)` as a statement) are unaffected; check with `f.write(s).ok()` or match the result.

42.13. Environment

Returns all environment variables as a vector of EnvVariable records (fields: name, value). Use to inspect or forward the full environment.

```
pub fn env_variables() -> vector<EnvVariable> env#read
```

```
pub fn env_variable(name: text) -> text env#read
```

Returns the value of the environment variable name, or null if it is not set. Use to read configuration from the shell environment.

Functions for interacting with the host operating system. Returns the script-level arguments passed after the script path. Does not include the loft binary name or loft CLI flags.

```
pub fn arguments() -> vector<text>
```

```
pub fn ymd_days_ago(days: integer) -> text
```

Returns the UTC calendar date `days` days before today as `YYYY-MM-DD`. Use for cutoff dates in time-window filters; negative `days` clamps to today.

```
pub fn directory(v: &text = "") -> text
```

Returns the current working directory, optionally with v appended as a subpath. Use to construct absolute paths relative to where the program was launched.

```
pub fn user_directory(v: &text = "") -> text
```

Returns the current user's home directory, optionally with v appended. Use for storing user-specific data or configuration.

```
pub fn program_directory(v: &text = "") -> text
```

Returns the directory containing the running executable, optionally with v appended. Use to locate assets bundled alongside the program.

```
pub fn source_dir() -> text
```

Returns the directory containing the main source file being executed. Use to locate data files relative to the script, regardless of working directory.

42.14. Optional C libraries (@PLN24 arc G)

```
pub fn c_library_available(soname: text) -> boolean env#read
```

Returns whether the C library soname is usable right now: it loads, and every `\#c` binding declared against it resolves in it. Ask this before calling into a library declared with `\[c\] optional-libs`, which is not installed on every machine and is loaded only when a binding needs it. Both halves of the answer matter — a library of the wrong version loads and then exports only some of the symbols, so "the file is there" would say yes where the call still fails. A library the program never declared answers false.

42.15. System directories (#635)

Private native: the OS temp dir, "" only where there is no filesystem.

```
pub fn temp_dir() -> text?env#read
```

Returns the system temporary directory, or null on a target with no filesystem (the browser). Prefer this over reading TMPDIR / TEMP / TMP directly: those differ by platform (Unix uses TMPDIR, Windows TEMP/TMP) and this also applies the OS fallback (e.g. /tmp) when none is set.

```
pub fn cache_dir() -> text?env#read
```

Returns loft's per-user cache directory ($XDG_CACHE_HOME/loft, else $HOME/.cache/loft) — the same root the engine caches into — or null on a target with no filesystem (the browser).

```
pub fn host_input(wait_ms: integer = -1) -> text
```

Reads pending host input as one text; empty when there is none. The source is per-target: stdin on native and –native-wasm (WASI); the JS host's input QUEUE on –html — seed it with globalThis.loftInput before loft_start, and push live messages any time with globalThis.loftPush(msg) (e.g. fetch() completions). Use for a headless compute module or a polled input loop — the inbound mirror of host_output. The engine only moves opaque bytes; the program parses them. wait_ms says how long to wait for input that has not arrived yet. The default –1 reads the WHOLE stream, waiting for stdin to end; 0 takes what is already there and returns at once; a positive value waits that many milliseconds for the first byte. Ask with a bound whenever the answer might be that nobody is listening: an absent host never closes stdin, so the default form waits forever for a reply that is not coming. host_output("MODE?"); mode = host_input(200); On –html every read already polls (a page cannot wait for a message its own thread has to deliver), and on –native-wasm, which has no threads, a bounded read falls back to waiting for the whole stream.

```
pub fn host_output(msg: text)
```

Sends one STRUCTURED message to the host shell — the outbound mirror of host_input, distinct from user-facing print. Per target: a line on stderr (native and WASI — the invoking process can script it); the page's globalThis.loftOutput(msg) handler on –html. The browser pattern for loft-initiated work: host_output a request (e.g. "fetch <url>"), the JS shell acts on it, and pushes the completion back via loftPush for host_input to pop — JS owns the network, loft stays pure compute.

42.16. Time

```
pub fn now() -> integer
```

Returns the current wall-clock time as milliseconds since the Unix epoch (00:00 UTC on 1 Jan 1970). Use for timestamps and logging.

```
pub fn ticks() -> integer
```

Returns microseconds elapsed since program start (monotonic clock). Unaffected by system clock adjustments. Use for benchmarks and frame timing.

42.17. Memory diagnostics

```
pub fn store_memory() -> text
```

Returns a multi-line snapshot of all LIVE stores' internal memory utilisation: total capacity vs actual claimed data vs free space, record + free-block counts, mergeable adjacent-free pairs (free neighbours that should have coalesced), and the largest stores by capacity with their creation site (`bc:\<pos\>` — a bytecode position on the interpreter; 0 on –native) and type name. Use to watch memory growth in a running program.

42.18. Vector operations (T2-8, T2-5)

reverse(v) and sort(v) are compiler special-cased in parse_call.

```
pub fn starts_with(self: text, value: text) -> boolean
```

Returns true if self begins with value. Use for prefix matching (e.g., protocol detection).

```
pub fn ends_with(self: text, value: text) -> boolean
```

Returns true if self ends with value. Use for suffix matching (e.g., file extension checks).

```
pub fn dir(self: text) -> text
```

Directory part of a path — strips the trailing `/\<segment\>`. “doc/claude/PROBLEMS.md”.dir() ⟶ “doc/claude” “CLAUDE.md”.dir() ⟶ “” (no `/`) “/etc/passwd”.dir() ⟶ “/etc” “”.dir() ⟶ “”

```
pub fn basename(self: text) -> text
```

Last segment of a path — the part after the trailing `/`. “doc/claude/PROBLEMS.md”.basename() ⟶ “PROBLEMS.md” “CLAUDE.md”.basename() ⟶ “CLAUDE.md” “/foo/”.basename() ⟶ “” (trailing slash) “”.basename() ⟶ “”

```
pub fn join(self: text, other: text) -> text
```

Join two paths — `self/other` with a single `/` separator. Special cases: - other starts with `/` (absolute) ⟶ return other unchanged - self is empty ⟶ return other - self ends with `/` ⟶ no extra separator inserted

```
pub fn resolve(self: text, target: text) -> text
```

Resolve `target` against `self` (where self is a base directory). Strips leading `./` repeatedly; each `../` segment trims the last component from `self`. Mirrors scan.loft’s `resolve\_link\_path`. “doc/claude”.resolve(“../README.md”) ⟶ “doc/README.md” “doc/claude”.resolve(“./PROBLEMS.md”) ⟶ “doc/claude/PROBLEMS.md” “a/b/c”.resolve(“../../x”) ⟶ “a/x” “”.resolve(“foo”) ⟶ “foo”

43. Roadmap

Loft is under active development. Everything documented on the language pages works today. This page describes where the project is headed.

The project goal is **browser games that anyone can play via a shared link**. Native OpenGL is supported for desktop enthusiasts. Server and multiplayer features come after the single-player browser experience works.

43.0.1. Current release — 0.8.4: Awesome Brick Buster

0.8.4 turns Brick Buster from a tech demo into a game someone would actually want to share with a friend, and makes sharing trivial via single-file HTML export.

43.0.1.1. Game polish

- **Hand-designed levels** — five themed layouts dispatched by `level_brick(lv, r, c)`, with a procedural fallback for deeper runs.
- **Cel-shaded sprite atlas** — outlined icons, a round ball with velocity-directional squash, hearts for lives, balloon projectiles, fireball after-images.
- **Roman-numeral level display** and **persistent high score** (`.loft/brickbuster_score.txt`).
- **Chiptune soundtrack** — three early-Capcom-style tracks (Heroic / Determined / Calm) that play once per level with silences between, plus brick/paddle/wall/pickup/life sound effects.
- **Screen shake** on brick breaks and life lost; pause menu; title and game-over screens.

43.0.1.2. Single-file HTML export

Compile any loft program to a self-contained `.html` file that runs in any browser — no install, no server.

```
$ loft --html app.loft
```

The game's WebAssembly, textures, and audio are all embedded. Host the file anywhere, send someone the URL, they play.

43.0.1.3. Infrastructure

- **Audio** — `rodio`-backed playback on desktop; WebAudio bridge in the browser.
- **Store-leak fixes** — per-frame idiomatic struct APIs now work without workarounds (P122 + P117–P131 resolved).
- **CLI hardening** — script-level arguments, file-scope constants, and release-mode coroutine-iterator fixes (P126, P128, P131, P132).
- **Bytecode cache** — `.loftc` files are invalidated on interpreter rebuild via the embedded git commit hash.

43.0.2. Next — 0.8.5: Working Moros editor

The Moros hex RPG scene editor runs end-to-end in the browser: load a map, paint hexes, place walls and items, see a live 3D preview, export to GLB. Web only — multiplayer comes in 1.0.0.

- **Map model** — `Hex`, `Chunk`, `Map` with material/wall/item palettes and spawn/NPC routines, serialised as JSON.
- **2D scene canvas** — hex coordinate math, pan/zoom/hit-test, layered rendering.
- **Editor shell** — toolbar, palettes, Paint / Height / Wall / Item tools, undo/redo, localStorage autosave.

- **3D renderer** — hex surface geometry (flat and slope), wall slabs, stairs, camera orbit, hex picking, GLB export.
- **Live 3D preview** alongside the 2D editor, powered by the loft graphics library compiled to WASM.

43.0.3. 0.9.0 — Fully working loft language

The language itself becomes feature-complete, well-documented, and tooling-friendly. No "appears fixed but unverified" bugs. Anyone can write loft code in their preferred editor with syntax highlighting, decent error messages, and a REPL for experimentation.

43.0.3.1. Language polish

- **Error recovery** — the parser continues after a token-level failure and reports multiple errors in a single pass.
- **REPL** — running `loft` with no arguments starts an interactive session; definitions persist across lines.
- **Developer warnings** — Clippy-inspired lints for common mistakes.
- **AOT library compilation** — automatically compile libraries to native shared libs for faster startup.
- **Stdlib hygiene** — name-clash warnings and a `std::` prefix for shadowed builtins; library enums in `match` arms without qualification.

43.0.3.2. Compilation cache

Bytecode cache (`.loftc`) and the shared constant store are already implemented. Remaining work — mmap-based native cache loading, serialised `Data` definitions, pre-compiled stdlib for WASM — lands in 0.9.0.

43.0.3.3. Developer experience

- TextMate grammar for `.loft` syntax highlighting.
- VS Code extension with snippets and a run task.
- Quick-start `examples/` directory.
- CI covering package tests and native codegen.

43.0.3.4. Packaging and FFI

- **Lock file** (`loft.lock`) for reproducible builds.
- **Generic FFI marshaller** — zero-boilerplate native functions from a `#native` signature; generic `cdylib` loader scans exports into a hash map.

43.0.4. 1.0.0 — Totally sure everything works

The stability contract. Anyone can write, run, and share a program — terminal or browser — and trust that it will keep working.

43.0.4.1. Web IDE

A browser-based IDE running the full loft interpreter as WebAssembly — no installation, no server.

- Editor shell with CodeMirror 6 and a loft grammar.
- Symbol navigation (go-to-definition, find-usages).
- Multi-file projects stored locally (IndexedDB).
- Documentation and examples browser built in.
- Export/import ZIP; PWA offline mode.

43.0.4.2. Scene scripting

The Moros scene editor gains a loft scripting panel with in-browser compile and hot reload. Scene scripts are sandboxed, travel with the scene JSON, and can be shared alongside the map.

43.0.4.3. Multiplayer

- **Server library** — plain HTTP routing, HTTPS with static certs or automatic ACME, WebSockets, JWT / session / API-key auth, CORS, rate limiting, static file serving.
- **Game loop primitives** — WebSocket polling, broadcast, connection registry.
- **Game client library** — `GameEnvelope` protocol, lobby + matchmaking, fixed-timestep loop, client-side prediction and reconciliation, WASM script loading with Ed25519 verification.
- **Moros multiplayer** — DM and players share a live scene hosted on a loft server.

43.0.4.4. Stability contract

Version 1.0 means: any program that works on 1.0 will compile and run identically on all future 1.x releases. The language syntax, type system, standard library, and command-line flags are frozen. Until then, breaking changes are possible between minor versions.

Tagging 1.0 requires every stability gate to be satisfied without shortcuts: zero open high-severity bugs, valgrind-clean debug builds of the full Brick Buster and tuple tests, green CI on Linux, macOS Intel, macOS ARM, and Windows, pre-built binaries on the GitHub release, and the reference + PDF linked from the release page.

43.0.5. 1.1 and beyond

Post-1.0 items include route decorators (`@get` / `@post` / `@ws`), WASM Web Worker pools for browser-side `par()`, iterator protocol (`for msg in ws`), interface factory methods, lazy work-variable initialisation, stack raw-pointer cache, spatial index operations, and additional native codegen optimisations.

43.0.6. Following progress

Development is tracked in the GitHub repository. The full internal roadmap with effort estimates, dependencies, and design pointers is in the repository's `doc/claude/ROADMAP.md`.